C. Armana
armana@math.jussieu.fr

MK1 "Calcul formel" Maple

TP1 : Premiers pas en Maple

Qu’est-ce que Maple ?
C’est un logiciel propriétaire de calcul formel et, dans une moindre mesure, de calcul approché :
- calcul approché ("flottant") : Maple donne et manipule des valeurs approchées de nombres (avec un nombre fini de décimales), comme une "supercalculatrice"
- calcul formel ("exact") : Maple manipule des nombres, des symboles représentant des nombres ou des objets mathématiques plus compliqués (fonctions, équations) de façon abstraite, sans passer par des valeurs approchées.

Le but des TP MK1
Il s’agit d’apprendre à utiliser Maple en illustrant votre programme de mathématiques par des exemples et des exercices "à faire avec Maple". Bien entendu, **cela suppose que vous connaissiez votre cours de maths !** N’hésitez pas à amener votre cours avec vous lors des séances de TP.

Bibliographie
"Maple sugar", de Guy Le Bris (éd. Cassini)

Et surtout, n’oubliez pas de vous (et de me) poser des questions !

Comment se débrouiller avec Maple : l’aide du logiciel

Maple est un logiciel très riche et il est hors de question d’en connaitre toutes les commandes qui seront abordées en TP et leurs syntaxes. L’aide de Maple est très utile pour retrouver ce genre d’informations, et **il est essentiel de savoir l’utiliser.** Deux façons d’y accéder :
* Par le menu Help, Topic search, pour une recherche thématique.
* Pour des informations sur une commande particulière, taper à l’invite ? suivi du nom de la commande :

```
> ?isprime
```

1. Présentation de l’interface

* La feuille blanche est appelée *feuille de calcul*. C’est là que vous donnez à Maple des commandes, qu’il les exécute et vous affiche le résultat. C’est l’analogue de l’écran d’une calculatrice ordinaire (mais on peut y afficher bien plus de choses !)
* La feuille de calcul commence par une *invite* (symbole >) : cela signifie que Maple est pret à recevoir des commandes de l’utilisateur.
* Si on tape alors une commande comme 1+2 qu’on valide en appuyant sur la touche "Entrée" :

```
> 1+2

Warning, premature end of input
```

...on reçoit un message d’erreur.
Règle n°1: toute commande doit etre suivie d’un caractère terminateur, la plupart du temps ; (point-virgule), parfois : (deux-points).

```
> 1+2;
```

$$3$$

Ouf !
Si on utilise deux-points à la place du point-virgule, Maple effectue la commande mais n’affiche pas le résultat. Cela peut etre pratique quand le résultat donné par Maple prend trop de place à afficher, ou si la valeur exacte ne nous intéresse pas.

```
> 80!;
```

715694570462638022948115337231865321655846573423657525771094450582270\
925548014884266894486728081408000000000000000000

```
> 80!:
```

* On peut donner plusieurs commandes à Maple sur la meme ligne :

```
> 2^5;11*7;
```

$$32$$
$$77$$

* Pour enregistrer sa feuille de calcul, on utilise le menu "File", "Save" ou "Save as". Le fichier a une extension .mws.
* Pour ouvrir une feuille enregistrée, on utilise le menu "File", "Open". Si on souhaite travailler à nouveau dessus, il faut alors re-exécuter toutes les commandes de la feuille,soit en les validant une à une avec la touche "Entrée", soit en utilisant le menu Edit, Execute, Worksheet.

2. L’affectation de variables

Il est très pratique de donner des noms à des résultats antérieurs, notamment pour pouvoir les réutiliser par la suite : c’est l’ *affectation.* En voici un exemple :

```
> produit:=6!;
```

$$produit := 720$$

A gauche du signe :=, on entre le nom de la variable et à droite la valeur affectée. On peut vérifier l’affectation par :

```
> produit;
```

$$720$$

et l’utiliser pour faire de nouveaux calculs :

```
> produit/5!;
```

$$6$$

Le nom de variable ne doit pas comporter de signes de ponctuation, d’espace ni de caractères spéciaux (par ex. +,*,#,%,@). On peut utiliser des majuscules et des minuscules (attention ! *Maple différencie les deux !)* . La procédure d’affectation est très générale. Ici, on a donné un nom de variable à un nombre entier (6!) mais on peut nommer également des nombres rationnels, décimaux, complexes, des fonctions, des matrices,...
Pour réinitialiser (désaffecter) la variable *produit* et faire en sorte qu’elle ne contienne plus la valeur 6!, on effectue l’une ou l’autre des commandes suivantes:

```
> produit:='produit';
```

$$produit := produit$$

```
> unassign('produit');
> produit;
```

$$produit$$

Si on veut réinitialiser toutes les variables, on utilise la commande *restart*.

3. L'ordre des commandes

Il est très important de comprendre que **le comportement de Maple dépend de l'ordre chronologique de validation des commandes, et non de l'ordre d'apparition sur la feuille de calcul.**

Dans une feuille de calcul, rien ne vous empeche de modifier une commande entrée précédemment : il suffit pour cela de remonter à la ligne qui vous intéresse (au clavier ou à la souris), de modifier la commande et de valider avec *"Entrée"* . Cependant, attention à cette manipulation, sous peine d'arriver à des choses bizarres... Par exemple :

```
> a:=3;
```

$$a := 3$$

```
> b:=a/2;
```

$$b := \frac{3}{2}$$

Si en remontant, je décide de modifier a:=3 en a:=2, mais que j'oublie de valider avec "Entrée" la ligne suivante, b vaudra toujours 3/2 et ne sera plus égal à a/2 !

Règle n°2 : si vous modifiez une commande précédente dans la feuille, faites re-exécuter les lignes suivantes à Maple avec la touche "Entrée".

Règle n°3 : il vaut mieux commencer une feuille de calcul ou un exercice par la commande :

```
> restart;
```

afin d'etre certain que toutes les variables sont désaffectées.

5. Calculs sur les nombres entiers

Maple fait automatiquement des calculs exacts sur de très grands entiers. Les opérations usuelles sont +,-,*. La puissance est ^.

Dans Maple, les différents objets ont un *type*. On peut demander le type d'un objet par la commande *whattype*. Par exemple, le type d'un entier est *integer*.

```
> 2*4*6*8;
```

$$384$$

```
> 5^3;
```

$$125$$

```
> whattype(15!);
```

integer

6. Calculs sur les nombres réels

Pour calculer des *valeurs approchées* de nombres réels (avec un certain nombre de chiffres significatifs, par défaut 10), on utilise la commande *evalf* (f pour flottant). L'équivalent de notre virgule décimal est ici le point décimal anglo-saxon (.).

```
> evalf(1/3);
```

$$0.3333333333$$

```
> 300/45;
```

$$\frac{20}{3}$$

Pour Maple, le nombre précédent n'est pas un nombre réel, c'est un nombre rationnel (une fraction de deux nombres entiers) : d'ailleurs, il nous a proposé spontanément une simplification. Pour lui faire comprendre qu'on souhaite une valeur approchée de ce nombre réel :

```
> evalf(300/45);
```

$$6.666666667$$

Attention, manipuler un nombre de façon exacte ou par valeur approchée, ce n'est pas du tout la meme chose ! C'est toute la différence entre le calcul formel et le calcul approché. Voici un exemple à méditer, lié aux erreurs d'approximation.

```
> sqrt(3)^2;
```

$$3$$

```
> x:=evalf(sqrt(3));
  x^2;
```

$$x := 1.732050808$$

$$3.000000001$$

Sauf mention contraire, dans les TPs, la règle sera de manipuler les nombres de façon exacte (car l'intitulé de l'UE est "calcul formel !") .

Le type d'un réel est *float* (nombre flottant).

```
> whattype(evalf(300/45));
```

float

Pour choisir le nombre de chiffres significatifs :

```
> evalf(300/45,15);
```

$$6.66666666666667$$

Maple connait certains nombres réels classiques comme **e** :

```
> exp(1);
  evalf(exp(1));
```

$$\mathbf{e}$$

$$2.718281828$$

et π **(attention à la majuscule !):**

```
> Pi;
  evalf(Pi);
```

$$\pi$$

$$3.141592654$$

Voici quelques fonctions classiques que connait Maple. Utilisez l'aide pour en savoir plus :

exp (exponentielle)
ln ou log
sqrt (racine carrée)
sin, cos, tan
abs
trunc, floor, ceil
max(x1,x2,...,xn), min(x1,...,xn)

7. Calculs sur les nombres complexes

Pour définir un nombre complexe, on peut utiliser le nombre imaginaire *i*, que Maple représente par **I (attention à la majuscule !).**

```
> z:=3+4*I;
```

$$z := 3 + 4\,I$$

Pour obtenir ses parties réelles et imaginaires :

```
> Re(z);Im(z);
```

$$3$$

$$4$$

Pour forcer Maple à écrire le nombre sous forme cartésienne (partie réelle + *i* * partie imaginaire), on utilise *evalc* (c pour complexe) :

```
> z*(1+sqrt(2)*I);
```

$$z\,(1+\sqrt{2}\,I)$$

```
> evalc(%);
```

$$3-4\sqrt{2}+(4+3\sqrt{2})\,I$$

(la commande % sert à rappeler le résultat précédent - par ordre chronologique).
Pour obtenir le nombre complexe conjugué, le module et l'argument, on utilise les commandes *conjugate, abs, argument*. Le module est un réel positif. L'argument (donné par Maple) est dans]-π,π].

```
> conjugate(z);abs(z);argument(z);
```

$$3-4\,I$$

$$5$$

$$\arctan\left(\frac{4}{3}\right)$$

Pour définir un nombre complexe sous forme trigonométrique $r\,\mathbf{e}^{(I\,\theta)}$, où r est le module et θ un argument, on utilise la commande *polar* (le premier argument est r, le deuxième θ) :

```
> polar(3, Pi/6);
```

$$\operatorname{polar}\left(3,\frac{\pi}{6}\right)$$

```
> evalc(%);
```

$$\frac{3\sqrt{3}}{2}+\frac{3}{2}I$$

Pour passer de l'écriture cartésienne $a+b\,I$ à l'écriture polaire, on utilise encore *polar* mais la syntaxe est différente :

```
> polar(1+I);
```

$$\operatorname{polar}\left(\sqrt{2},\frac{\pi}{4}\right)$$

<u>8. Les fonctions simplify et assume</u>

Maple ne signifie pas forcément les formules algébriques, comme le montre l'exemple :

```
> 4^(1/2)+4;
```

$$\sqrt{4}+4$$

Dans ces cas, ou plus généralement lorsque le résultat n'a pas la forme souhaitée, on peut appliquer certaines fonctions de Maple. La fonction *simplify* est l'une d'entre elles. Elle est très riche, aussi nous ne regarderons que des exemples.

```
> simplify(%);
```

$$6$$

```
> (sin(x))^4-(cos(x))^4;
```

$$\sin(x)^4-\cos(x)^4$$

```
> simplify(%);
```

$$1-2\cos(x)^2$$

Un exemple important :

```
> y:=sqrt(x^2);
```

$$y:=\sqrt{x^2}$$

```
> simplify(y);
```

$$\operatorname{csgn}(x)\,x$$

Quelle est la fonction *csgn* ? Le résultat est-il correct ?
Par défaut, Maple ne sait rien de la variable non affectée x et la considère comme un nombre complexe. Si on suppose (*assume* en anglais) que x est positif, on peut encore simplifier l'expression :

```
> simplify(y,assume=positive);
```

$$x$$

La supposition est temporaire (le temps que la commande soit effectuée). Si on souhaite qu'elle soit permanente, c'est-à-dire jusqu'à la fin de la session ou jusqu'à ce que la variable soit réinitialisée, on utilise *assume* avec une syntaxe différente :

```
> assume(x,positive);
> simplify(y);
```

$$x\sim$$

Le ~ (tilde) rappelle qu'une hypothèse a été faite sur la variable x.

Université Paris 7
Licence 1ère année

Premier semestre 2006-2007
MK1 - Maple

Feuille d'exercices n°1

Exercice 1. *Une formule de Ramanujan.* Soit :

$$k = (\sqrt{2}-1)^2(2-\sqrt{3})(\sqrt{7}-\sqrt{6})^2(8-3\sqrt{7})(\sqrt{10}-3)^2(\sqrt{15}-\sqrt{14})(4-\sqrt{15})^2(6-\sqrt{35}).$$

Calculer $A = \frac{-2}{\sqrt{210}}\ln(\frac{k}{4})$ avec 30 chiffres significatifs. Que pensez-vous du résultat ?

Exercice 2. Soit le nombre complexe $z = \left(\frac{1+i\sqrt{3}}{1-i}\right)^{20}$. Calculer son module et son argument. Donner une valeur approchée de son argument.

Exercice 3. Soit le nombre complexe $z = \frac{1}{2} + i\frac{\sqrt{3}}{2}$. En utilisant Maple, démontrer que les points du plan d'affixes z, $z-1$ et z^2 sont alignés.

Exercice 4. Soient a,b,c les trois racines du polynôme en z à coefficients complexes : $z^3-(6+3i)z^2+(9+12i)z-9(2+3i)$. Calculer ces racines à l'aide de la commande *solve*. Montrer que les points du plan d'affixes respectives a,b,c forment un triangle équilateral.

Exercice 5. On se place dans le plan complexe rapporté à un repère orthonormé direct $(0,i,j)$. Soit M_0 le point d'affixe $z_0 = 1+i\sqrt{3}$. Pour $n \geq 1$, soit M_n le point d'affixe $z_n = a^n z_0$ où $a = i/2$. En utilisant la commande *seq* (consulter l'aide), construire la séquence des dix premiers termes de la suite $(z_n)_{n\in\mathbb{N}^*}$ mis sous forme cartésienne. Construire la séquence des modules des dix premiers termes de la suite.

Exercice 6. On rappelle la *formule de Moivre* :

$$(\cos(x)+i\sin(x))^n = \cos(nx)+i\sin(nx).$$

En utilisant cette formule, donner les formules exprimant $\cos(5x)$ et $\sin(5x)$ en fonction de $\cos(x)$ et $\sin(x)$.

Correction du TP1

Exercice 1

```
> restart;
> k:=(sqrt(2)-1)^2*(2-sqrt(3))*(sqrt(7)-sqrt(6))^2*(8-3*sqrt(7))*
  (sqrt(10)-3)^2*(sqrt(15)-sqrt(14))*(4-sqrt(15))^2*(6-sqrt(35));
```

$k :=$

$$(\sqrt{2}-1)^2 (2-\sqrt{3}) (\sqrt{7}-\sqrt{6})^2 (8-3\sqrt{7}) (\sqrt{10}-3)^2 (\sqrt{15}-\sqrt{14}) (4-\sqrt{15})^2 (6-\sqrt{35})$$

```
> A:=(-2/sqrt(210))*ln(k/4);
```

$A := -\frac{1}{105}\sqrt{210}\,\ln($

$$(\sqrt{2}-1)^2 (2-\sqrt{3}) (\sqrt{7}-\sqrt{6})^2 (8-3\sqrt{7}) (\sqrt{10}-3)^2 (\sqrt{15}-\sqrt{14}) (4-\sqrt{15})^2 (6-\sqrt{35}) / 4)$$

```
> evalf(A,30);
```

$$3.14159265358979323847198212950$$

```
> evalf(A-Pi);
```

$$0.$$

Attention, le calcul ci-dessus a été effectué avec 10 chiffres significatifs. C'est un calcul *approché* . Cela ne signifie pas que $A = \pi$. D'ailleurs, si on augmente le nombre de chiffres significatifs :

```
> evalf(Pi-A,30);
```

$$-0.933874622\ 10^{-20}$$

Donc le nombre A est une approximation de π à 10^(-20) près.

Exercice 2

```
> restart;
> z:=((1+I*sqrt(3))/(1-I))^20;
```

$$z := -\frac{(1+\sqrt{3}\,I)^{20}}{1024}$$

```
> abs(z);argument(z);
```

$$1024$$

$$\mathrm{argument}(-(1+\sqrt{3}\,I)^{20})$$

```
> simplify(argument(z));
```

$$\mathrm{argument}(-(1+\sqrt{3}\,I)^{20})$$

Maple ne sait pas simplifier algébriquement l'argument de *z*. Par contre, on peut lui demander une valeur *approchée :*

```
> evalf(argument(z));
```

$$-1.047197549$$

Exercice 3

```
> restart;
  z:=1/2+I*sqrt(3)/2;
```

$$z := \frac{1}{2} + \frac{1}{2} I\sqrt{3}$$

Rappel : A, B, C trois points du plan d'affixes zA, zB et zC. Les points A, B, C sont alignés si et seulement si les vecteurs AB et AC sont colinéaires. Le vecteur AB a pour affixe zB-zA, le vecteur AC a pour affixe zC-zA. Les points sont alignés si et seulement s'il existe un nombre réel k tel que zB-zA=k*(zC-zA), c'est-à-dire si le nombre complexe (zB-zA)/(zC-zA) est réel, c'est-à-dire s'il a une partie imaginaire nulle !

```
> Im((z-(z-1))/(z-z^2));
```

$$0$$

Donc les trois points sont alignés.

Exercice 4

```
> restart;
  S:=solve(z**3-(6+3*I)*z**2+(9+12*I)*z-9*(2+3*I));
```

$$S := 3+\sqrt{3}+\sqrt{3}\,I,\ 3-\sqrt{3}-\sqrt{3}\,I,\ 3\,I$$

Maple nous donne l'ensemble des solutions S sous forme d'une *séquence* :

```
> whattype(S);
```

$$\mathit{exprseq}$$

C'est une collection *ordonnée* d'expressions séparées par des virgules. La commande pour appeler le k ème membre de S est *S[k].*

```
> a:=S[1];b:=S[2];c:=S[3];
```

$$a := 3+\sqrt{3}+\sqrt{3}\,I$$

$$b := 3-\sqrt{3}-\sqrt{3}\,I$$

$$c := 3\,I$$

Pour montrer que le triangle est équilatéral, on vérifie par exemple que les cotés AB, BC et CA ont meme longueur. La distance AB est donnée par le module du nombre complexe b-a (c'est-à-dire la norme du vecteur AB).

```
> abs(a-b);abs(b-c);abs(c-a);
```

$$2\sqrt{6}$$

$$\sqrt{(3-\sqrt{3})^2+(-3-\sqrt{3})^2}$$

$$\sqrt{(3-\sqrt{3})^2+(-3-\sqrt{3})^2}$$

Sont-ils égaux ? On peut encore simplifier ces expressions :

```
> abs(a-b);simplify(abs(b-c));simplify(abs(c-a));
```

$$2\sqrt{6}$$

$$2\sqrt{6}$$

$$2\sqrt{6}$$

Donc le triangle est équilatéral !

Exercice 5

```
> restart;
> z0:=1+I*sqrt(3);
```

$$z0 := 1+\sqrt{3}\,I$$

```
> a:=I/2;
```

$$a := \frac{1}{2} I$$

```
> seq(a^n*z0,n=1..10);
```

$$\frac{1}{2} I (1+\sqrt{3}\,I),\ -\frac{1}{4}-\frac{1}{4} I\sqrt{3},\ \frac{-1}{8} I (1+\sqrt{3}\,I),\ \frac{1}{16}+\frac{1}{16} I\sqrt{3},\ \frac{1}{32} I (1+\sqrt{3}\,I),\ -\frac{1}{64}-\frac{1}{64} I\sqrt{3},$$

$$\frac{-1}{128} I (1+\sqrt{3}\, I), \frac{1}{256}+\frac{1}{256} I\sqrt{3}, \frac{1}{512} I (1+\sqrt{3}\, I), -\frac{1}{1024}-\frac{1}{1024} I\sqrt{3}$$

```
> seq(evalc(a^n*z0),n=1..10);
```

$$\frac{1}{2} I-\frac{\sqrt{3}}{2}, -\frac{1}{4}-\frac{1}{4} I\sqrt{3}, -\frac{1}{8} I+\frac{\sqrt{3}}{8}, \frac{1}{16}+\frac{1}{16} I\sqrt{3}, \frac{1}{32} I-\frac{\sqrt{3}}{32}, -\frac{1}{64}-\frac{1}{64} I\sqrt{3}, -\frac{1}{128} I+\frac{\sqrt{3}}{128},$$
$$\frac{1}{256}+\frac{1}{256} I\sqrt{3}, \frac{1}{512} I-\frac{\sqrt{3}}{512}, -\frac{1}{1024}-\frac{1}{1024} I\sqrt{3}$$

```
> seq(abs(a^n*z0),n=1..10);
```

$$1, \frac{1}{2}, \frac{1}{4}, \frac{1}{8}, \frac{1}{16}, \frac{1}{32}, \frac{1}{64}, \frac{1}{128}, \frac{1}{256}, \frac{1}{512}$$

Exercice 6

```
> restart;
  z:=(cos(x)+I*sin(x))^5;Re(z);Im(z);
```

$$z := (\cos(x)+\sin(x)\, I)^5$$
$$\Re((\cos(x)+\sin(x)\, I)^5)$$
$$\Im((\cos(x)+\sin(x)\, I)^5)$$

```
> evalc(z);
```

$$\cos(x)^5-10\cos(x)^3\sin(x)^2+5\cos(x)\sin(x)^4$$
$$+(5\cos(x)^4\sin(x)-10\cos(x)^2\sin(x)^3+\sin(x)^5)\, I$$

Il suffit alors d’extraire les parties réelles et imaginaires. Pourtant :

```
> Re(evalc(z));
```

$$\Re(\cos(x)^5-10\cos(x)^3\sin(x)^2+5\cos(x)\sin(x)^4$$
$$+(5\cos(x)^4\sin(x)-10\cos(x)^2\sin(x)^3+\sin(x)^5)\, I)$$

La commande qui fonctionne ici est la suivante : on demande d’abord à Maple de calculer formellement la partie réelle de z, puis on demande de faire une évaluation complexe (*evalc*) :

```
> evalc(Re(z));
  evalc(Im(z));
```

$$\cos(x)^5-10\cos(x)^3\sin(x)^2+5\cos(x)\sin(x)^4$$
$$5\cos(x)^4\sin(x)-10\cos(x)^2\sin(x)^3+\sin(x)^5$$

Attention, si on utilise *evalf* (évaluation approchée des nombres réels à 10 chiffres significatifs près) à la place de *evalc*, on n’obtient pas le résultat voulu :

```
> evalf(Re(z));
```

$$\Re((\cos(x)+1.\, I\sin(x))^5)$$

```
>
```

C. Armana
armana@math.jussieu.fr

MK1 "Calcul formel" Maple

TP2 : Séquences, ensembles, listes, fonctions, représentations graphiques

Restart au début de chaque exercice :
Il est très fortement recommandé de mettre une commande :

```
> restart;
```

au début de chaque nouvel exercice afin de réinitialiser les variables.

Interface de Maple :
Pour créer une nouvelle ligne d'invite dans Maple, cliquer sur l'icône [> .
Pour taper du texte non interprété par Maple (par exemple, "Exercice n°1"), créer une ligne d'invite avec [> puis cliquer sur l'icône T.
Pour mettre des commentaires au milieu de commandes Maple, on utilise le symbole # (tout ce qui suit le symbole jusqu'à la fin de la ligne est alors ignoré).

Et surtout, n'oubliez pas de vous (et de me) poser des questions !

1. Les collections d'expressions

La semaine dernière, nous avons utilisé des variables, en leur affectant une donnée simple (un nombre entier, un nombre complexe,...). Il peut être intéressant de *regrouper plusieurs "objets" dans une seule variable*. Par exemple, si on affecte dans une variable l'ensemble des solutions d'une équation, cela permettra de manipuler toutes les solutions à la fois. Il existe différentes façons de collecter des expressions sous Maple.

1.1 Les séquences

Pour Maple, une *séquence* est une collection **ordonnée** d'expressions séparées par des **virgules**. Ces expressions peuvent être du même type (par exemple des nombres entiers) ou des objets de natures différentes (on peut former une séquence composée d'une fonction, d'un nombre entier, et d'un nombre réel approché).

```
> a,b,c ; # exemple de séquence
```

$$a, b, c$$

```
> whattype(%);
```

$$exprseq$$

Le type des séquences est *exprseq*.
Les séquences sont extremement utiles : par exemple, la commande *solve*, qui permet de résoudre des équations, retourne les solutions sous forme de séquence. Il est donc important de savoir les manipuler.

Pour former une séquence, **une première méthode** est donc d'en lister les éléments dans l'ordre en les séparant par des virgules. On peut affecter cette séquence à une variable (c'est-à-dire lui "donner un nom") :

```
> s:=a,b,c;
```

$$s := a, b, c$$

Voici sur des exemples comment rappeler les éléments d'une séquence en utilisant des *crochets* :

```
> s[1]; # premier élément
  s[3]; # troisième élément
  s[4];
```

$$a$$

$$c$$

```
Error, invalid subscript selector
```

Bien sur, il n'existe pas de 4ème élément, c'est pour ça que Maple nous retourne un message d'erreur !

```
> s[2..3];
```

$$b, c$$

La séquence vide (aucun élément) se note NULL :

```
> vide:=NULL;
```

$$vide :=$$

On peut former une séquence en mettant bout à bout (concaténant) plusieurs séquences existantes : il suffit de séparer les séquences par des *virgules* :

```
> t:=1,2,3;
```

$$t := 1, 2, 3$$

```
> u:=t,NULL,s;
```

$$u := 1, 2, 3, a, b, c$$

Remarque : les séquences sont des **structures à lecture seule**Cela signifie qu'on ne peut pas modifier une séquence existante, par exemple en affectant l'un de ses éléments :

```
> u[5];
```

$$b$$

```
> u[5]:=0;
Error, invalid assignment (1, 2, 3, a, b, c)[5] := 0; cannot assign to an expression sequence
```

Par contre, rien ne nous empêche de créer une nouvelle séquence à partir de "morceaux" de la séquence existante.
Une deuxième méthode pour créer une séquence est la fonction *seq,* qui est extremement utile. Elle permet de former seulement les séquences définies à partir de "formules" : voir les exercices et l'aide sur *seq.*

1.2 Les ensembles

Pour Maple, un *ensemble* est une collection **non ordonnée** d'expressions **toutes différentes** . Il se note à l'aide d'une séquence entre**accolades** { , }.
La collection est non ordonnée : l'ordre des éléments que nous donnons à Maple en entrée n'est pas toujours respecté. D'ailleurs, il peut changer d'une fois sur l'autre.

```
> e:={d,b,c,a};
```

$$e := \{b, a, d, c\}$$

Les expressions sont toutes différentes : si Maple voit plusieurs fois apparaitre la même expression, il élimine les doublons :

```
> f:={a,a,a,a,b};
```

$$f := \{b, a\}$$

```
> whattype(e);
```

set

Le type des ensembles est *set* (ensemble, en anglais). Les méthodes de construction de séquences permettent aussi d'obtenir des ensembles : il suffit d'ajouter des accolades. Par exemple :

```
> {seq(i,i=1..15)};
```

$$\{1, 2, 3, 4, 5, 6, 7, 8, 9, 10, 11, 12, 13, 14, 15\}$$

Remarque : comme les séquences, les ensembles sont des **structures à lecture seule.**
On peut effectuer les manipulations usuelles sur les ensembles grâce aux fonctions suivantes :
- la réunion de deux ensembles : *union*
- l'intersection de deux ensembles : *intersect*
- la différence de deux ensembles : *minus*

```
> e minus {a,d};
```

$$\{b, c\}$$

On obtient le nombre d'éléments de l'ensemble par la commande *nops* :

```
> nops(e);
```

$$4$$

et la *séquence* des éléments de l'ensemble par la commande *op* :

```
> op(e);
```

$$a, b, c, d$$

1.3 Les listes

Pour Maple, une *liste* est une collection **ordonnée** d'expressions, notée entre **crochets**. C'est une "séquence entre crochets" et il peut y avoir des doublons. On peut la construire à partir d'une séquence.

```
> L:=[seq(1/(2*i),i=1..20)]; whattype(L);
```

$$L := \left[\frac{1}{2}, \frac{1}{4}, \frac{1}{6}, \frac{1}{8}, \frac{1}{10}, \frac{1}{12}, \frac{1}{14}, \frac{1}{16}, \frac{1}{18}, \frac{1}{20}, \frac{1}{22}, \frac{1}{24}, \frac{1}{26}, \frac{1}{28}, \frac{1}{30}, \frac{1}{32}, \frac{1}{34}, \frac{1}{36}, \frac{1}{38}, \frac{1}{40}\right]$$

list

Pour accéder au k-ème élément, on procède comme pour les séquences :

```
> L[4];L[10..15];
```

$$\frac{1}{8}$$

$$\left[\frac{1}{20}, \frac{1}{22}, \frac{1}{24}, \frac{1}{26}, \frac{1}{28}, \frac{1}{30}\right]$$

Remarque :comme les séquences et les ensembles, les listes sont des **structures à lecture seule.**
Pour obtenir le nombre d'éléments de la liste :

```
> nops(L);
```

$$20$$

Pour obtenir la *séquence* formée des éléments de la liste (= "enlever les crochets autour de la liste"):

```
> op(L);
```

$$\frac{1}{2}, \frac{1}{4}, \frac{1}{6}, \frac{1}{8}, \frac{1}{10}, \frac{1}{12}, \frac{1}{14}, \frac{1}{16}, \frac{1}{18}, \frac{1}{20}, \frac{1}{22}, \frac{1}{24}, \frac{1}{26}, \frac{1}{28}, \frac{1}{30}, \frac{1}{32}, \frac{1}{34}, \frac{1}{36}, \frac{1}{38}, \frac{1}{40}$$

On souhaite fondre deux listes en une.

```
> restart;L1:=[a,b,c];L2:=[d,e,f];
```

$$L1 := [a, b, c]$$

$$L2 := [d, e, f]$$

```
> L1,L2;
  whattype(L1,L2);
```

$$[a, b, c], [d, e, f]$$

exprseq

L'objet que l'on obtient en les séparant par une virgule est une séquence de deux listes. Ce n'est pas le résultat voulu. La façon de fondre ces deux listes en une est de concaténer les séquences des éléments de L1 et L2 et de créer une nouvelle liste :

```
> L:=[op(L1),op(L2)];
```

$$L := [a, b, c, d, e, f]$$

2. Les fonctions

2.1 Attention, *fonction* $\neq$ *expression* !!!

Comme en mathématiques, Maple fait la différence entre les fonctions et les expressions.

```
> x^2+x+1 ; # ceci est une expression en x
```

$$x^2 + x + 1$$

On souhaite maintenant remplacer *x* par une ou plusieurs valeurs. Comment s'y prendre ?

```
> A:=x^2+x+1;
  A(3);
```

$$A := x^2 + x + 1$$

$$\mathrm{x}(3)^2 + \mathrm{x}(3) + 1$$

Maple n'a pas substitué à *x* la valeur 3... La bonne méthode est de définir d'abord une fonction. Maple connait un certain nombre de fonctions de base : *exp, ln, sin, sqrt, ...* Pour définir vos propres fonctions, il y a deux méthodes :

2.2 La flèche ->

Elle est très proche de la notation mathématique. La syntaxe est: nom_fonction : = nom_variable -> expression_en_la_variable

```
> f:=x->x^2+x+1; # ceci est une fonction
  f(3);
  f(10);
```

$$f := x \to x^2 + x + 1$$

$$13$$

$$111$$

```
> f(x); # f(x) est une expression en x
```

$$x^2 + x + 1$$

On peut aussi définir des fonctions de plusieurs variables : nom_fonction : = (séquence_noms_variables) -> expression_en_les_variables

```
> g:=(a,b)->a^b;
  g(2,3);
```

$$g := (a, b) \to a^b$$

$$8$$

(attention à bien mettre la séquence des variables entre parenthèses)
Le résultat rendu par une fonction peut être d'un type quelconque, par exemple une liste :

```
> h:=x->[cos(x),sin(x)];
```

```
h(Pi/6);
```

$$h := x \to [\cos(x), \sin(x)]$$

$$\left[\frac{\sqrt{3}}{2}, \frac{1}{2}\right]$$

2.3 La fonction *unapply*

L'autre méthode pour définir une fonction est la commande *unapply* dont la syntaxe est : unapply(expression, séquence_noms_variables).

```
> f:=unapply(x^2+x+1,x);f(3);
```

$$f := x \to x^2 + x + 1$$

$$13$$

Pour une fonction de plusieurs variables :

```
> g:=unapply([x^2+y^2,x*y],x,y);
```

$$g := (x, y) \to [x^2 + y^2, x\, y]$$

2.3 La composition des fonctions

La composition des applications (notation g o f en mathématiques) se fait à l'aide de @.

```
> restart;f:=exp@ln;
```

$$f := \exp@\ln$$

```
> f(x);
```

$$x$$

3. Les fonctions add et mul

Elles permettent de calculer des *sommes* et des *produits*. Leur syntaxe est la suivante :
add(f(k),k=a..b) calcule la somme des f(k), pour k entier variant de a à b
mul(f(k),k=a..b) calcule le produit des f(k), pour k entier variant de a à b
Bien sur, k, a et b peuvent avoir d'autres noms. Mais k doit etre une variable non affectée.

```
> add(cos(x)^k,k=0..7);
```

$$1 + \cos(x) + \cos(x)^2 + \cos(x)^3 + \cos(x)^4 + \cos(x)^5 + \cos(x)^6 + \cos(x)^7$$

4. La fonction map

La fonction *map* permet, entre autres, d'évaluer une fonction (d'une variable) en plusieurs valeurs qui sont les éléments d'un ensemble ou d'une liste.

```
> restart; f:=x->2*sin(x);
```

$$f := x \to 2\sin(x)$$

```
> E:={a,b,c,d,e}; L:=[a,b,d,c,e];
```

$$E := \{a, b, c, d, e\}$$

$$L := [a, b, d, c, e]$$

```
> map(f,E);
```

$$\{2\sin(a), 2\sin(b), 2\sin(c), 2\sin(d), 2\sin(e)\}$$

```
> map(f,L);
```

$$[2\sin(a), 2\sin(b), 2\sin(d), 2\sin(c), 2\sin(e)]$$

Attention, on ne peut pas appliquer la fonction *map* à une séquence ! Voici un exemple :

```
> S:=a,b,c,d,e; map(f,S);
```

$$S := a, b, c, d, e$$

$$2\sin(a)$$

On voit que la fonction s'est appliquée uniquement au premier élément de la séquence.

5. Les représentations graphiques

Nous allons voir comment tracer le graphique d'une fonction f définie sur **R** (ou un intervalle de **R**) à valeurs dans **R**. On utilise pour cela la fonction *plot* (qui permet de tracer également d'autres types de graphiques). La syntaxe est par exemple : plot (f(x) , x=a..b) où a et b désignent les bornes de l'intervalle (de l'axe des abscisses) sur lequel on trace *f*.

```
> restart;
  plot(sin(x),x=0..4*Pi);
```



On peut par exemple spécifier un intervalle en ordonnée pour le tracé et choisir la couleur de la courbe.

```
> plot(sin(x),x=0..4*Pi,y=0..1.5,color=blue);
```



Pour tracer plusieurs fonctions *f1,...,fn* à la fois, on utilise la syntaxe *[f1(x),...,fn(x)]* :

```
> plot([x,x^2,x^3],x=0..1.5);
```



Référez-vous à l'aide de *plot* pour en savoir plus !

Il existe des fonctions avancées permettant de tracer d'autres types de graphiques. La plupart sont contenues dans la **bibliothèque** (package en anglais) nommée **plots**. Une bibliothèque est un ensemble de commandes et fonctions relatives à un thème (ici, la représentation graphique) qui ne sont pas disponibles par défaut au lancement de Maple. Si vous souhaitez utiliser ces fonctions, vous devez charger la bibliothèque avec la commande *with.*

```
> with(plots);
Warning, the name changecoords has been redefined
```

[*animate*, *animate3d*, *animatecurve*, *arrow*, *changecoords*, *complexplot*, *complexplot3d*, *conformal*, *conformal3d*, *contourplot*, *contourplot3d*, *coordplot*, *coordplot3d*, *cylinderplot*, *densityplot*, *display*, *display3d*, *fieldplot*, *fieldplot3d*, *gradplot*, *gradplot3d*, *graphplot3d*, *implicitplot*, *implicitplot3d*, *inequal*, *interactive*, *interactiveparams*, *listcontplot*, *listcontplot3d*, *listdensityplot*, *listplot*, *listplot3d*, *loglogplot*, *logplot*, *matrixplot*, *multiple*, *odeplot*, *pareto*, *plotcompare*, *pointplot*, *pointplot3d*, *polarplot*, *polygonplot*, *polygonplot3d*, *polyhedra_supported*, *polyhedraplot*, *replot*, *rootlocus*, *semilogplot*, *setoptions*, *setoptions3d*, *spacecurve*, *sparsematrixplot*, *sphereplot*, *surfdata*, *textplot*, *textplot3d*, *tubeplot*]

Maple affiche la liste des nouvelles fonctions disponibles. Par exemple, pour placer des points dans un plan en donnant leurs coordonnées, on peut utiliser la fonction *pointplot*. Pour avoir de l'aide sur la commande *pointplot*, on doit préciser que la fonction est dans la bibliothèque **plots** :

```
> ?plots,pointplot
> pointplot({[2,1],[-3,4],[7,2]});
```



```
>
```

Université Paris 7 | **Premier semestre 2006-2007**
Licence 1ère année | **MK1 - Maple**

Feuille d'exercices n°2

Exercice 1. Regarder l'aide à propos de la commande *seq* et comprendre sa syntaxe. L'utiliser pour construire :

- la séquence des k^2 pour les entiers k variant de 1 à 30 ;
- la séquence des nombres pairs entre 1 et 50 ;
- la séquence suivante : $1, \frac{1}{8}, \frac{1}{27}, \frac{1}{64}, \frac{1}{125}$;
- la séquence des $e^{a\pi/b}$ pour a variant de 1 à 15 et b variant de 1 à 10.

Exercice 2. Nous avons déjà vu la commande *solve* pour résoudre des équations. Trouver les racines complexes du polynôme $x^{20} - 1$. Quel est le type d'objet que retourne Maple ? En utilisant Maple et sans les afficher, déterminer combien il a de racines. A l'aide de la fonction *map*, vérifier que ce sont bien des racines. Montrer que la somme des racines est nulle.

Exercice 3. Construire en une seule commande :

- la liste des 100 premiers nombres premiers (regarder la commande *ithprime*) ;
- la liste des nombres premiers entre 1 et 100.

Exercice 4. Tracer les graphes de $x \mapsto \log(x+1)$ et de $x \mapsto 1,01 \log(x)$. Chercher leur(s) point(s) d'intersection : avec *solve*, que se passe-t-il ? avec *fsolve* ? Tracer sur un même dessin les graphes de ces deux fonctions, en choisissant l'intervalle en abscisse de façon à faire apparaître ce(s) point(s) d'intersection.

Exercice 5. Utiliser *seq* pour tracer sur un même dessin les graphes des fonctions $f_n(x) = x^{n^2/10}$ pour n allant de 1 à 10. Recadrer le dessin en abscisse et en ordonnée pour le rendre plus lisible.

Exercice 6. On reprend l'exercice 2. En utilisant *pointplot*, tracer dans le plan les points correspondant aux racines complexes du polynôme $x^{20} - 1$. Que remarquez-vous ? Comment l'expliquez-vous ?

Correction du TP2

Exercice 1

```
> restart;
> seq(k^2,k=1..30);
```

1, 4, 9, 16, 25, 36, 49, 64, 81, 100, 121, 144, 169, 196, 225, 256, 289, 324, 361, 400, 441, 484, 529, 576, 625, 676, 729, 784, 841, 900

Les nombres pairs sont de la forme 2*i. Attention : si on souhaite ceux qui sont compris entre 1 et 50, l'entier i doit varier de 1 à 25.

```
> seq(2*i,i=1..25);
```

2, 4, 6, 8, 10, 12, 14, 16, 18, 20, 22, 24, 26, 28, 30, 32, 34, 36, 38, 40, 42, 44, 46, 48, 50

Pour utiliser la fonction *seq*, il faut d'abord "deviner la formule" qui se cache derrière la séquence 1, 1/8, 1/27, 1/64, 1/125.

```
> seq(1/k^3,k=1..5);
```

$$1, \frac{1}{8}, \frac{1}{27}, \frac{1}{64}, \frac{1}{125}$$

Pour la séquence des exp(a*Pi/b), a variant de 1 à 15 et b de 1 à 10, la première idée est de faire :

```
> seq(exp(a*Pi/b),a=1..15,b=1..10);
Error, wrong number (or type) of parameters in function seq
```

L'erreur vient du fait que *seq* utilise une fonction à 1 paramètre ; ici, notre problème est à 2 paramètres. On peut s'en sortir à l'aide de *seq* en faisant d'abord varier a puis b.

```
> seq(exp(a*Pi/b),a=1..15);
```

$$\mathbf{e}^{\left(\frac{\pi}{b}\right)}, \mathbf{e}^{\left(\frac{2\pi}{b}\right)}, \mathbf{e}^{\left(\frac{3\pi}{b}\right)}, \mathbf{e}^{\left(\frac{4\pi}{b}\right)}, \mathbf{e}^{\left(\frac{5\pi}{b}\right)}, \mathbf{e}^{\left(\frac{6\pi}{b}\right)}, \mathbf{e}^{\left(\frac{7\pi}{b}\right)}, \mathbf{e}^{\left(\frac{8\pi}{b}\right)}, \mathbf{e}^{\left(\frac{9\pi}{b}\right)}, \mathbf{e}^{\left(\frac{10\pi}{b}\right)}, \mathbf{e}^{\left(\frac{11\pi}{b}\right)}, \mathbf{e}^{\left(\frac{12\pi}{b}\right)}, \mathbf{e}^{\left(\frac{13\pi}{b}\right)}, \mathbf{e}^{\left(\frac{14\pi}{b}\right)}, \mathbf{e}^{\left(\frac{15\pi}{b}\right)}$$

```
> S:=seq(seq(exp(a*Pi/b),a=1..15),b=1..10);
```

$$S := \mathbf{e}^{\pi}, \mathbf{e}^{(2\pi)}, \mathbf{e}^{(3\pi)}, \mathbf{e}^{(4\pi)}, \mathbf{e}^{(5\pi)}, \mathbf{e}^{(6\pi)}, \mathbf{e}^{(7\pi)}, \mathbf{e}^{(8\pi)}, \mathbf{e}^{(9\pi)}, \mathbf{e}^{(10\pi)}, \mathbf{e}^{(11\pi)}, \mathbf{e}^{(12\pi)},$$
$$\mathbf{e}^{(13\pi)}, \mathbf{e}^{(14\pi)}, \mathbf{e}^{(15\pi)}, \mathbf{e}^{\left(\frac{\pi}{2}\right)}, \mathbf{e}^{\pi}, \mathbf{e}^{\left(\frac{3\pi}{2}\right)}, \mathbf{e}^{(2\pi)}, \mathbf{e}^{\left(\frac{5\pi}{2}\right)}, \mathbf{e}^{(3\pi)}, \mathbf{e}^{\left(\frac{7\pi}{2}\right)}, \mathbf{e}^{(4\pi)}, \mathbf{e}^{\left(\frac{9\pi}{2}\right)},$$
$$\mathbf{e}^{(5\pi)}, \mathbf{e}^{\left(\frac{11\pi}{2}\right)}, \mathbf{e}^{(6\pi)}, \mathbf{e}^{\left(\frac{13\pi}{2}\right)}, \mathbf{e}^{(7\pi)}, \mathbf{e}^{\left(\frac{15\pi}{2}\right)}, \mathbf{e}^{\left(\frac{\pi}{3}\right)}, \mathbf{e}^{\left(\frac{2\pi}{3}\right)}, \mathbf{e}^{\pi}, \mathbf{e}^{\left(\frac{4\pi}{3}\right)}, \mathbf{e}^{\left(\frac{5\pi}{3}\right)}, \mathbf{e}^{(2\pi)},$$
$$\mathbf{e}^{\left(\frac{7\pi}{3}\right)}, \mathbf{e}^{\left(\frac{8\pi}{3}\right)}, \mathbf{e}^{(3\pi)}, \mathbf{e}^{\left(\frac{10\pi}{3}\right)}, \mathbf{e}^{\left(\frac{11\pi}{3}\right)}, \mathbf{e}^{(4\pi)}, \mathbf{e}^{\left(\frac{13\pi}{3}\right)}, \mathbf{e}^{\left(\frac{14\pi}{3}\right)}, \mathbf{e}^{(5\pi)}, \mathbf{e}^{\left(\frac{\pi}{4}\right)}, \mathbf{e}^{\left(\frac{\pi}{2}\right)},$$
$$\mathbf{e}^{\left(\frac{3\pi}{4}\right)}, \mathbf{e}^{\pi}, \mathbf{e}^{\left(\frac{5\pi}{4}\right)}, \mathbf{e}^{\left(\frac{3\pi}{2}\right)}, \mathbf{e}^{\left(\frac{7\pi}{4}\right)}, \mathbf{e}^{(2\pi)}, \mathbf{e}^{\left(\frac{9\pi}{4}\right)}, \mathbf{e}^{\left(\frac{5\pi}{2}\right)}, \mathbf{e}^{\left(\frac{11\pi}{4}\right)}, \mathbf{e}^{(3\pi)}, \mathbf{e}^{\left(\frac{13\pi}{4}\right)},$$
$$\mathbf{e}^{\left(\frac{7\pi}{2}\right)}, \mathbf{e}^{\left(\frac{15\pi}{4}\right)}, \mathbf{e}^{\left(\frac{\pi}{5}\right)}, \mathbf{e}^{\left(\frac{2\pi}{5}\right)}, \mathbf{e}^{\left(\frac{3\pi}{5}\right)}, \mathbf{e}^{\left(\frac{4\pi}{5}\right)}, \mathbf{e}^{\pi}, \mathbf{e}^{\left(\frac{6\pi}{5}\right)}, \mathbf{e}^{\left(\frac{7\pi}{5}\right)}, \mathbf{e}^{\left(\frac{8\pi}{5}\right)}, \mathbf{e}^{\left(\frac{9\pi}{5}\right)}, \mathbf{e}^{(2\pi)},$$
$$\mathbf{e}^{\left(\frac{11\pi}{5}\right)}, \mathbf{e}^{\left(\frac{12\pi}{5}\right)}, \mathbf{e}^{\left(\frac{13\pi}{5}\right)}, \mathbf{e}^{\left(\frac{14\pi}{5}\right)}, \mathbf{e}^{(3\pi)}, \mathbf{e}^{\left(\frac{\pi}{6}\right)}, \mathbf{e}^{\left(\frac{\pi}{3}\right)}, \mathbf{e}^{\left(\frac{\pi}{2}\right)}, \mathbf{e}^{\left(\frac{2\pi}{3}\right)}, \mathbf{e}^{\left(\frac{5\pi}{6}\right)}, \mathbf{e}^{\pi}, \mathbf{e}^{\left(\frac{7\pi}{6}\right)},$$
$$\mathbf{e}^{\left(\frac{4\pi}{3}\right)}, \mathbf{e}^{\left(\frac{3\pi}{2}\right)}, \mathbf{e}^{\left(\frac{5\pi}{3}\right)}, \mathbf{e}^{\left(\frac{11\pi}{6}\right)}, \mathbf{e}^{(2\pi)}, \mathbf{e}^{\left(\frac{13\pi}{6}\right)}, \mathbf{e}^{\left(\frac{7\pi}{3}\right)}, \mathbf{e}^{\left(\frac{5\pi}{2}\right)}, \mathbf{e}^{\left(\frac{\pi}{7}\right)}, \mathbf{e}^{\left(\frac{2\pi}{7}\right)}, \mathbf{e}^{\left(\frac{3\pi}{7}\right)},$$
$$\mathbf{e}^{\left(\frac{4\pi}{7}\right)}, \mathbf{e}^{\left(\frac{5\pi}{7}\right)}, \mathbf{e}^{\left(\frac{6\pi}{7}\right)}, \mathbf{e}^{\pi}, \mathbf{e}^{\left(\frac{8\pi}{7}\right)}, \mathbf{e}^{\left(\frac{9\pi}{7}\right)}, \mathbf{e}^{\left(\frac{10\pi}{7}\right)}, \mathbf{e}^{\left(\frac{11\pi}{7}\right)}, \mathbf{e}^{\left(\frac{12\pi}{7}\right)}, \mathbf{e}^{\left(\frac{13\pi}{7}\right)}, \mathbf{e}^{(2\pi)},$$
$$\mathbf{e}^{\left(\frac{15\pi}{7}\right)}, \mathbf{e}^{\left(\frac{\pi}{8}\right)}, \mathbf{e}^{\left(\frac{\pi}{4}\right)}, \mathbf{e}^{\left(\frac{3\pi}{8}\right)}, \mathbf{e}^{\left(\frac{\pi}{2}\right)}, \mathbf{e}^{\left(\frac{5\pi}{8}\right)}, \mathbf{e}^{\left(\frac{3\pi}{4}\right)}, \mathbf{e}^{\left(\frac{7\pi}{8}\right)}, \mathbf{e}^{\pi}, \mathbf{e}^{\left(\frac{9\pi}{8}\right)}, \mathbf{e}^{\left(\frac{5\pi}{4}\right)}, \mathbf{e}^{\left(\frac{11\pi}{8}\right)},$$
$$\mathbf{e}^{\left(\frac{3\pi}{2}\right)}, \mathbf{e}^{\left(\frac{13\pi}{8}\right)}, \mathbf{e}^{\left(\frac{7\pi}{4}\right)}, \mathbf{e}^{\left(\frac{15\pi}{8}\right)}, \mathbf{e}^{\left(\frac{\pi}{9}\right)}, \mathbf{e}^{\left(\frac{2\pi}{9}\right)}, \mathbf{e}^{\left(\frac{\pi}{3}\right)}, \mathbf{e}^{\left(\frac{4\pi}{9}\right)}, \mathbf{e}^{\left(\frac{5\pi}{9}\right)}, \mathbf{e}^{\left(\frac{2\pi}{3}\right)}, \mathbf{e}^{\left(\frac{7\pi}{9}\right)},$$
$$\mathbf{e}^{\left(\frac{8\pi}{9}\right)}, \mathbf{e}^{\pi}, \mathbf{e}^{\left(\frac{10\pi}{9}\right)}, \mathbf{e}^{\left(\frac{11\pi}{9}\right)}, \mathbf{e}^{\left(\frac{4\pi}{3}\right)}, \mathbf{e}^{\left(\frac{13\pi}{9}\right)}, \mathbf{e}^{\left(\frac{14\pi}{9}\right)}, \mathbf{e}^{\left(\frac{5\pi}{3}\right)}, \mathbf{e}^{\left(\frac{\pi}{10}\right)}, \mathbf{e}^{\left(\frac{\pi}{5}\right)}, \mathbf{e}^{\left(\frac{3\pi}{10}\right)},$$
$$\mathbf{e}^{\left(\frac{2\pi}{5}\right)}, \mathbf{e}^{\left(\frac{\pi}{2}\right)}, \mathbf{e}^{\left(\frac{3\pi}{5}\right)}, \mathbf{e}^{\left(\frac{7\pi}{10}\right)}, \mathbf{e}^{\left(\frac{4\pi}{5}\right)}, \mathbf{e}^{\left(\frac{9\pi}{10}\right)}, \mathbf{e}^{\pi}, \mathbf{e}^{\left(\frac{11\pi}{10}\right)}, \mathbf{e}^{\left(\frac{6\pi}{5}\right)}, \mathbf{e}^{\left(\frac{13\pi}{10}\right)}, \mathbf{e}^{\left(\frac{7\pi}{5}\right)},$$
$$\mathbf{e}^{\left(\frac{3\pi}{2}\right)}$$

Remarque : on peut vérifier qu'il y en le bon nombre (150). Pour cela, on utilise la fonction *nops*. Comme elle ne s'applique qu'aux ensembles ou aux listes, il faut convertir S en ensemble ou en liste. En fait, il ne faut pas convertir S en ensemble auquel cas on perdrait les éléments redondants (par ex. exp(Pi)) et l'ensemble contiendrait moins d'éléments que la séquence. On convertir donc S en liste :

```
> L:=[S]:
> nops(L);
```

150

Exercice 2

```
> restart;
> S:=solve(x^20-1,x);
```

$$S := -I, I, -1, 1, -\sqrt{-\sqrt{\frac{\sqrt{5}}{4}-\frac{1}{4}+\frac{1}{4}I\sqrt{2}\sqrt{5+\sqrt{5}}}}, -\sqrt{-\sqrt{-\frac{\sqrt{5}}{4}-\frac{1}{4}+\frac{1}{4}I\sqrt{2}\sqrt{5-\sqrt{5}}}},$$
$$-\sqrt{-\sqrt{-\frac{\sqrt{5}}{4}-\frac{1}{4}-\frac{1}{4}I\sqrt{2}\sqrt{5-\sqrt{5}}}}, -\sqrt{-\sqrt{\frac{\sqrt{5}}{4}-\frac{1}{4}-\frac{1}{4}I\sqrt{2}\sqrt{5+\sqrt{5}}}},$$
$$\sqrt{-\sqrt{\frac{\sqrt{5}}{4}-\frac{1}{4}+\frac{1}{4}I\sqrt{2}\sqrt{5+\sqrt{5}}}}, \sqrt{-\sqrt{-\frac{\sqrt{5}}{4}-\frac{1}{4}+\frac{1}{4}I\sqrt{2}\sqrt{5-\sqrt{5}}}},$$
$$\sqrt{-\sqrt{-\frac{\sqrt{5}}{4}-\frac{1}{4}-\frac{1}{4}I\sqrt{2}\sqrt{5-\sqrt{5}}}}, \sqrt{-\sqrt{\frac{\sqrt{5}}{4}-\frac{1}{4}-\frac{1}{4}I\sqrt{2}\sqrt{5+\sqrt{5}}}},$$

$$-\left(\frac{\sqrt{5}}{4}-\frac{1}{4}+\frac{1}{4}I\sqrt{2}\sqrt{5+\sqrt{5}}\right)^{(1/4)}, -\left(-\frac{\sqrt{5}}{4}-\frac{1}{4}+\frac{1}{4}I\sqrt{2}\sqrt{5-\sqrt{5}}\right)^{(1/4)},$$

$$-\left(-\frac{\sqrt{5}}{4}-\frac{1}{4}-\frac{1}{4}I\sqrt{2}\sqrt{5-\sqrt{5}}\right)^{(1/4)}, -\left(\frac{\sqrt{5}}{4}-\frac{1}{4}-\frac{1}{4}I\sqrt{2}\sqrt{5+\sqrt{5}}\right)^{(1/4)},$$

$$\left(\frac{\sqrt{5}}{4}-\frac{1}{4}+\frac{1}{4}I\sqrt{2}\sqrt{5+\sqrt{5}}\right)^{(1/4)}, \left(-\frac{\sqrt{5}}{4}-\frac{1}{4}+\frac{1}{4}I\sqrt{2}\sqrt{5-\sqrt{5}}\right)^{(1/4)},$$

$$\left(-\frac{\sqrt{5}}{4}-\frac{1}{4}-\frac{1}{4}I\sqrt{2}\sqrt{5-\sqrt{5}}\right)^{(1/4)}, \left(\frac{\sqrt{5}}{4}-\frac{1}{4}-\frac{1}{4}I\sqrt{2}\sqrt{5+\sqrt{5}}\right)^{(1/4)}$$

Une autre syntaxe possible pour la fonction *solve* est :

```
> solve(x^20-1=0,x):
> whattype(S);
```

exprseq

Maple retourne les solutions sous forme de *séquence* .
Pour compter les racines, on procède comme dans la remarque de l'exercice 1.

```
> nops([S]);
```

20

Il y a 20 racines.
Pour vérifier que ce sont des racines, il suffit d'appliquer la fonction :

```
> f:=x->x^20-1;
```

à chacune des solutions, à l'aide de *map*. Attention, la fonction *map* ne s'utilise pas avec des séquences. Cependant, on peut convertir la séquence en liste :

```
> map(f,[S]);
```

$$\left[0, 0, 0, 0, \left(\frac{\sqrt{5}}{4}-\frac{1}{4}+\frac{1}{4}I\sqrt{2}\sqrt{5+\sqrt{5}}\right)^5-1, \left(-\frac{\sqrt{5}}{4}-\frac{1}{4}+\frac{1}{4}I\sqrt{2}\sqrt{5-\sqrt{5}}\right)^5-1,\right.$$

$$\left(-\frac{\sqrt{5}}{4}-\frac{1}{4}-\frac{1}{4}I\sqrt{2}\sqrt{5-\sqrt{5}}\right)^5-1, \left(\frac{\sqrt{5}}{4}-\frac{1}{4}-\frac{1}{4}I\sqrt{2}\sqrt{5+\sqrt{5}}\right)^5-1,$$

$$\left(\frac{\sqrt{5}}{4}-\frac{1}{4}+\frac{1}{4}I\sqrt{2}\sqrt{5+\sqrt{5}}\right)^5-1, \left(-\frac{\sqrt{5}}{4}-\frac{1}{4}+\frac{1}{4}I\sqrt{2}\sqrt{5-\sqrt{5}}\right)^5-1,$$

$$\left(-\frac{\sqrt{5}}{4}-\frac{1}{4}-\frac{1}{4}I\sqrt{2}\sqrt{5-\sqrt{5}}\right)^5-1, \left(\frac{\sqrt{5}}{4}-\frac{1}{4}-\frac{1}{4}I\sqrt{2}\sqrt{5+\sqrt{5}}\right)^5-1,$$

$$\left(\frac{\sqrt{5}}{4}-\frac{1}{4}+\frac{1}{4}I\sqrt{2}\sqrt{5+\sqrt{5}}\right)^5-1, \left(-\frac{\sqrt{5}}{4}-\frac{1}{4}+\frac{1}{4}I\sqrt{2}\sqrt{5-\sqrt{5}}\right)^5-1,$$

$$\left(-\frac{\sqrt{5}}{4}-\frac{1}{4}-\frac{1}{4}I\sqrt{2}\sqrt{5-\sqrt{5}}\right)^5-1, \left(\frac{\sqrt{5}}{4}-\frac{1}{4}-\frac{1}{4}I\sqrt{2}\sqrt{5+\sqrt{5}}\right)^5-1,$$

$$\left(\frac{\sqrt{5}}{4}-\frac{1}{4}+\frac{1}{4}I\sqrt{2}\sqrt{5+\sqrt{5}}\right)^5-1, \left(-\frac{\sqrt{5}}{4}-\frac{1}{4}+\frac{1}{4}I\sqrt{2}\sqrt{5-\sqrt{5}}\right)^5-1,$$

$$\left.\left(-\frac{\sqrt{5}}{4}-\frac{1}{4}-\frac{1}{4}I\sqrt{2}\sqrt{5-\sqrt{5}}\right)^5-1, \left(\frac{\sqrt{5}}{4}-\frac{1}{4}-\frac{1}{4}I\sqrt{2}\sqrt{5+\sqrt{5}}\right)^5-1\right]$$

Visiblement, Maple n'a pas simplifié le résultat, alors on lui demande de le faire :

```
> simplify(%);
```

[0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0]

Donc les solutions sont bien des racines.
S[i] désigne la i-ème racine. Pour vérifier que la somme est nulle, on utilise la fonction *add* pour additionner les S[i], pour i variant de 1 à 20 (le nombre de racines).

```
> add(S[i],i=1..20);
```

0

Exercice 3

```
> restart;
```

D'après l'aide, *ithprime(k)* donne le k-ème nombre premier. Pour obtenir la liste des 100 premiers nombres premiers, on demande :

```
> L1:=[seq(ithprime(k),k=1..100)];
```

L1 := [2, 3, 5, 7, 11, 13, 17, 19, 23, 29, 31, 37, 41, 43, 47, 53, 59, 61, 67, 71, 73, 79, 83, 89, 97, 101, 103, 107, 109, 113, 127, 131, 137, 139, 149, 151, 157, 163, 167, 173, 179, 181, 191, 193, 197, 199, 211, 223, 227, 229, 233, 239, 241, 251, 257, 263, 269, 271, 277, 281, 283, 293, 307, 311, 313, 317, 331, 337, 347, 349, 353, 359, 367, 373, 379, 383, 389, 397, 401, 409, 419, 421, 431, 433, 439, 443, 449, 457, 461, 463, 467, 479, 487, 491, 499, 503, 509, 521, 523, 541]

(les crochets sont là pour obtenir une liste)
On peut vérifier qu'il y en a bien 100.

```
> nops(L1);
```

100

Pour obtenir la liste des nombres premiers entre 1 et 100, on peut partir de la construction précédente et enlever ceux qui sont "en trop" : par exemple, travailler avec des ensembles et faire l'intersection avec l'ensemble des entiers entre 1 et 100.

```
> S1:={seq(ithprime(k),k=1..100)};
```

S1 := {2, 3, 5, 7, 11, 13, 17, 19, 23, 29, 31, 37, 41, 43, 47, 53, 59, 61, 67, 71, 73, 79, 83, 89, 97, 101, 103, 107, 109, 113, 127, 131, 137, 139, 149, 151, 157, 163, 167, 173, 179, 181, 191, 193, 197, 199, 211, 223, 227, 229, 233, 239, 241, 251, 257, 263, 269, 271, 277, 281, 283, 293, 307, 311, 313, 317, 331, 337, 347, 349, 353, 359, 367, 373, 379, 383, 389, 397, 401, 409, 419, 421, 431, 433, 439, 443, 449, 457, 461, 463, 467, 479, 487, 491, 499, 503, 509, 521, 523, 541 }

```
> S1 intersect {seq(i,i=1..100)};
```

{2, 3, 5, 7, 11, 13, 17, 19, 23, 29, 31, 37, 41, 43, 47, 53, 59, 61, 67, 71, 73, 79, 83, 89, 97}

L'objet obtenu est un ensemble. On nous demande une liste. Il faut donc extraire la séquence des éléments de l'ensemble avec *op* et la convertir en liste à l'aide de crochets.

```
> op(S1 intersect {seq(i,i=1..100)});
```

2, 3, 5, 7, 11, 13, 17, 19, 23, 29, 31, 37, 41, 43, 47, 53, 59, 61, 67, 71, 73, 79, 83, 89, 97

```
> [op(S1 intersect {seq(i,i=1..100)})];
```

[2, 3, 5, 7, 11, 13, 17, 19, 23, 29, 31, 37, 41, 43, 47, 53, 59, 61, 67, 71, 73, 79, 83, 89, 97]

La commande en une ligne est donc :

```
> L2:=[op({seq(ithprime(k),k=1..100)} intersect
  {seq(i,i=1..100)})];
```

L2 :=

[2, 3, 5, 7, 11, 13, 17, 19, 23, 29, 31, 37, 41, 43, 47, 53, 59, 61, 67, 71, 73, 79, 83, 89, 97]

Ouf !

<u>Exercice 4</u>

```
> restart;
> f:=x->log(x+1); g:=x->1.01*log(x);
```

```
> plot([f(x),g(x)],x=-10..10);
```



Cherchons les points d'intersection avec la fonction solve.

```
> solve(f(x)=g(x),x);
Warning, computation interrupted
```

Le calcul ne s'arrete pas. Nous sommes obligés de l'interrompre à la main. *Maple ne sait pas résoudre de façon exacte cette équation*.
L'aide indique que *fsolve* résout les équations de façon numérique (*approchée*) : les solutions sont alors des nombres réels flottants, qui sont des *approximations* des solutions exactes.

```
> fsolve(f(x)=g(x),x);
```

29.15368826

Il y a un point d'intersection, d'abscisse approchée 29.15368826. On décide de centrer l'intervalle des abscisses autour de 30.

```
> plot([f(x),g(x)],x=20..40);
```



<u>Exercice 5</u>

```
> restart;
```

Commencons par définir une fonction f des deux variables n et x.

```
> f:=(n,x)->x^(n^2/10);
```

On souhaite tracer les graphes des fonctions f(1,x), ... , f(10,x) (fonctions d'une variable x). Pour cela, on utilise encore et toujours *seq* :

```
> seq(f(n,x),n=1..10);
```

$$x^{(1/10)}, x^{(2/5)}, x^{(9/10)}, x^{(8/5)}, x^{(5/2)}, x^{(18/5)}, x^{\left(\frac{49}{10}\right)}, x^{(32/5)}, x^{\left(\frac{81}{10}\right)}, x^{10}$$

```
> plot([seq(f(n,x),n=1..10)],x);
```



L'intervalle en abscisse que Maple choisit par défaut n'est pas vraiment adapté (on ne distingue presque pas les fonctions). En tatonnant un peu, on obtient un résultat meilleur :

```
> plot([seq(f(n,x),n=1..10)],x=0..3,y=0..2);
```



Exercice 6

```
> restart; S:=solve(x^20-1);
```

$$S := -I, I, -1, 1, -\sqrt{-\sqrt{\frac{\sqrt{5}}{4}-\frac{1}{4}+\frac{1}{4}I\sqrt{2}\sqrt{5+\sqrt{5}}}}, -\sqrt{-\sqrt{-\frac{\sqrt{5}}{4}-\frac{1}{4}+\frac{1}{4}I\sqrt{2}\sqrt{5-\sqrt{5}}}},$$
$$-\sqrt{-\sqrt{-\frac{\sqrt{5}}{4}-\frac{1}{4}-\frac{1}{4}I\sqrt{2}\sqrt{5-\sqrt{5}}}}, -\sqrt{-\sqrt{\frac{\sqrt{5}}{4}-\frac{1}{4}-\frac{1}{4}I\sqrt{2}\sqrt{5+\sqrt{5}}}},$$
$$\sqrt{-\sqrt{\frac{\sqrt{5}}{4}-\frac{1}{4}+\frac{1}{4}I\sqrt{2}\sqrt{5+\sqrt{5}}}}, \sqrt{-\sqrt{-\frac{\sqrt{5}}{4}-\frac{1}{4}+\frac{1}{4}I\sqrt{2}\sqrt{5-\sqrt{5}}}},$$
$$\sqrt{-\sqrt{-\frac{\sqrt{5}}{4}-\frac{1}{4}-\frac{1}{4}I\sqrt{2}\sqrt{5-\sqrt{5}}}}, \sqrt{-\sqrt{\frac{\sqrt{5}}{4}-\frac{1}{4}-\frac{1}{4}I\sqrt{2}\sqrt{5+\sqrt{5}}}},$$
$$-\left(\frac{\sqrt{5}}{4}-\frac{1}{4}+\frac{1}{4}I\sqrt{2}\sqrt{5+\sqrt{5}}\right)^{(1/4)}, -\left(-\frac{\sqrt{5}}{4}-\frac{1}{4}+\frac{1}{4}I\sqrt{2}\sqrt{5-\sqrt{5}}\right)^{(1/4)},$$
$$-\left(-\frac{\sqrt{5}}{4}-\frac{1}{4}-\frac{1}{4}I\sqrt{2}\sqrt{5-\sqrt{5}}\right)^{(1/4)}, -\left(\frac{\sqrt{5}}{4}-\frac{1}{4}-\frac{1}{4}I\sqrt{2}\sqrt{5+\sqrt{5}}\right)^{(1/4)},$$
$$\left(\frac{\sqrt{5}}{4}-\frac{1}{4}+\frac{1}{4}I\sqrt{2}\sqrt{5+\sqrt{5}}\right)^{(1/4)}, \left(-\frac{\sqrt{5}}{4}-\frac{1}{4}+\frac{1}{4}I\sqrt{2}\sqrt{5-\sqrt{5}}\right)^{(1/4)},$$
$$\left(-\frac{\sqrt{5}}{4}-\frac{1}{4}-\frac{1}{4}I\sqrt{2}\sqrt{5-\sqrt{5}}\right)^{(1/4)}, \left(\frac{\sqrt{5}}{4}-\frac{1}{4}-\frac{1}{4}I\sqrt{2}\sqrt{5+\sqrt{5}}\right)^{(1/4)}$$

Chacune des racines est un nombre complexe, qui a une partie imaginaire et une partie réelle. Pour chaque racine z, on calcule les coordonnées [Re(z),Im(z)] du point du plan correspondant à z, et on forme la séquence de ces coordonnées :

```
> points:=seq([Re(S[i]),Im(S[i])],i=1..nops([S]));
```

$$points := [0, -1], [0, 1], [-1, 0], [1, 0],$$
$$\left[-\Re\left(\sqrt{-\sqrt{\frac{\sqrt{5}}{4}-\frac{1}{4}+\frac{1}{4}I\sqrt{2}\sqrt{5+\sqrt{5}}}}\right), -\Im\left(\sqrt{-\sqrt{\frac{\sqrt{5}}{4}-\frac{1}{4}+\frac{1}{4}I\sqrt{2}\sqrt{5+\sqrt{5}}}}\right)\right],$$
$$\left[-\Re\left(\sqrt{-\sqrt{-\frac{\sqrt{5}}{4}-\frac{1}{4}+\frac{1}{4}I\sqrt{2}\sqrt{5-\sqrt{5}}}}\right), -\Im\left(\sqrt{-\sqrt{-\frac{\sqrt{5}}{4}-\frac{1}{4}+\frac{1}{4}I\sqrt{2}\sqrt{5-\sqrt{5}}}}\right)\right],$$
$$\left[-\Re\left(\sqrt{-\sqrt{-\frac{\sqrt{5}}{4}-\frac{1}{4}-\frac{1}{4}I\sqrt{2}\sqrt{5-\sqrt{5}}}}\right), -\Im\left(\sqrt{-\sqrt{-\frac{\sqrt{5}}{4}-\frac{1}{4}-\frac{1}{4}I\sqrt{2}\sqrt{5-\sqrt{5}}}}\right)\right],$$
$$\left[-\Re\left(\sqrt{-\sqrt{\frac{\sqrt{5}}{4}-\frac{1}{4}-\frac{1}{4}I\sqrt{2}\sqrt{5+\sqrt{5}}}}\right), -\Im\left(\sqrt{-\sqrt{\frac{\sqrt{5}}{4}-\frac{1}{4}-\frac{1}{4}I\sqrt{2}\sqrt{5+\sqrt{5}}}}\right)\right],$$
$$\left[\Re\left(\sqrt{-\sqrt{\frac{\sqrt{5}}{4}-\frac{1}{4}+\frac{1}{4}I\sqrt{2}\sqrt{5+\sqrt{5}}}}\right), \Im\left(\sqrt{-\sqrt{\frac{\sqrt{5}}{4}-\frac{1}{4}+\frac{1}{4}I\sqrt{2}\sqrt{5+\sqrt{5}}}}\right)\right],$$
$$\left[\Re\left(\sqrt{-\sqrt{-\frac{\sqrt{5}}{4}-\frac{1}{4}+\frac{1}{4}I\sqrt{2}\sqrt{5-\sqrt{5}}}}\right), \Im\left(\sqrt{-\sqrt{-\frac{\sqrt{5}}{4}-\frac{1}{4}+\frac{1}{4}I\sqrt{2}\sqrt{5-\sqrt{5}}}}\right)\right],$$

$$\left[\Re\left(\sqrt{-\sqrt{-\frac{\sqrt{5}}{4}-\frac{1}{4}-\frac{1}{4}I\sqrt{2}\sqrt{5-\sqrt{5}}}}\right), \Im\left(\sqrt{-\sqrt{-\frac{\sqrt{5}}{4}-\frac{1}{4}-\frac{1}{4}I\sqrt{2}\sqrt{5-\sqrt{5}}}}\right)\right],$$

$$\left[\Re\left(\sqrt{-\sqrt{\frac{\sqrt{5}}{4}-\frac{1}{4}-\frac{1}{4}I\sqrt{2}\sqrt{5+\sqrt{5}}}}\right), \Im\left(\sqrt{-\sqrt{\frac{\sqrt{5}}{4}-\frac{1}{4}-\frac{1}{4}I\sqrt{2}\sqrt{5+\sqrt{5}}}}\right)\right],$$

$$\left[-\Re\left(\left(\frac{\sqrt{5}}{4}-\frac{1}{4}+\frac{1}{4}I\sqrt{2}\sqrt{5+\sqrt{5}}\right)^{(1/4)}\right), -\Im\left(\left(\frac{\sqrt{5}}{4}-\frac{1}{4}+\frac{1}{4}I\sqrt{2}\sqrt{5+\sqrt{5}}\right)^{(1/4)}\right)\right],$$

$$\left[-\Re\left(\left(-\frac{\sqrt{5}}{4}-\frac{1}{4}+\frac{1}{4}I\sqrt{2}\sqrt{5-\sqrt{5}}\right)^{(1/4)}\right), -\Im\left(\left(-\frac{\sqrt{5}}{4}-\frac{1}{4}+\frac{1}{4}I\sqrt{2}\sqrt{5-\sqrt{5}}\right)^{(1/4)}\right)\right],$$

$$\left[-\Re\left(\left(-\frac{\sqrt{5}}{4}-\frac{1}{4}-\frac{1}{4}I\sqrt{2}\sqrt{5-\sqrt{5}}\right)^{(1/4)}\right), -\Im\left(\left(-\frac{\sqrt{5}}{4}-\frac{1}{4}-\frac{1}{4}I\sqrt{2}\sqrt{5-\sqrt{5}}\right)^{(1/4)}\right)\right],$$

$$\left[-\Re\left(\left(\frac{\sqrt{5}}{4}-\frac{1}{4}-\frac{1}{4}I\sqrt{2}\sqrt{5+\sqrt{5}}\right)^{(1/4)}\right), -\Im\left(\left(\frac{\sqrt{5}}{4}-\frac{1}{4}-\frac{1}{4}I\sqrt{2}\sqrt{5+\sqrt{5}}\right)^{(1/4)}\right)\right],$$

$$\left[\Re\left(\left(\frac{\sqrt{5}}{4}-\frac{1}{4}+\frac{1}{4}I\sqrt{2}\sqrt{5+\sqrt{5}}\right)^{(1/4)}\right), \Im\left(\left(\frac{\sqrt{5}}{4}-\frac{1}{4}+\frac{1}{4}I\sqrt{2}\sqrt{5+\sqrt{5}}\right)^{(1/4)}\right)\right],$$

$$\left[\Re\left(\left(-\frac{\sqrt{5}}{4}-\frac{1}{4}+\frac{1}{4}I\sqrt{2}\sqrt{5-\sqrt{5}}\right)^{(1/4)}\right), \Im\left(\left(-\frac{\sqrt{5}}{4}-\frac{1}{4}+\frac{1}{4}I\sqrt{2}\sqrt{5-\sqrt{5}}\right)^{(1/4)}\right)\right],$$

$$\left[\Re\left(\left(-\frac{\sqrt{5}}{4}-\frac{1}{4}-\frac{1}{4}I\sqrt{2}\sqrt{5-\sqrt{5}}\right)^{(1/4)}\right), \Im\left(\left(-\frac{\sqrt{5}}{4}-\frac{1}{4}-\frac{1}{4}I\sqrt{2}\sqrt{5-\sqrt{5}}\right)^{(1/4)}\right)\right],$$

$$\left[\Re\left(\left(\frac{\sqrt{5}}{4}-\frac{1}{4}-\frac{1}{4}I\sqrt{2}\sqrt{5+\sqrt{5}}\right)^{(1/4)}\right), \Im\left(\left(\frac{\sqrt{5}}{4}-\frac{1}{4}-\frac{1}{4}I\sqrt{2}\sqrt{5+\sqrt{5}}\right)^{(1/4)}\right)\right]$$

(nops([S]) compte le nombre d'éléments de la séquence S ; en fait, on sait que c'est 20 d'après l'exercice 2).
Si on souhaite exprimer vraiment les parties réelles et imaginaires, on peut passer un coup de *evalc* (mais le résultat n'est pas plus beau).

```
> seq([evalc(Re(S[i])),evalc(Im(S[i]))],i=1..nops([S])) :
```

On charge la bibliothèque **plots** dans laquelle se trouve *pointplot*.

```
> with(plots):
Warning, the name changecoords has been redefined
```

Enfin, on regarde l'aide sur *pointplot* et on trace l'ensemble des points :

```
> pointplot({points});
```




```
>
```

On remarque que les points se trouvent sur le cercle de centre 0 et de rayon 1. C'est normal : les racines nèmes de l'unité (ici, vingtièmes) sont des nombres complexes de module 1. On remarque aussi que les points sont les sommets d'un polygone régulier à 20 cotés (pouvez-vous le démontrer mathématiquement ?)

C. Armana
armana@math.jussieu.fr

MK1 "Calcul formel" Maple

TP3 : Résolution d'équations, dérivation, intégration

Version de démonstration de Maple :
Vous pouvez installer une version de démonstration de Maple (gratuite, mais limitée) sur votre ordinateur personnel, quelque soit son système d'exploitation (Windows, Linux, MacOS,...) : il suffit de la télécharger sur le ftp de Maplesoft, la société qui édite Maple :
ftp://ftp.maplesoft.com/pub/maple/demo
La version de Maple proposée est V Release 4. En TP, vous utilisez la version 9. Il existe de (petites) différences entre les deux.

Commandes pour les caractères spéciaux
Voici les commandes pour obtenir les caractères qui ne sont pas présents sur les claviers Mac.
pour ~ : Alt n
pour { : Alt (pour [: Alt Shift (
pour } : Alt) pour] : Alt Shift)

Et surtout, n'oubliez pas de vous (et de me) poser des questions !

1. Résoudre des équations

1.1 Résoudre une équation à une inconnue

Pour résoudre une équation à une inconnue, on utilise la commande *solve* dont la syntaxe est *solve(equation, variable).*
L'équation est définie par une égalité =, à bien différencier des affectations, qui se font par : =.

Commençons par les **équations polynomiales** . Maple résout complètement sur **C** les équations polynomiales de degré <=3 et quelques équations de degré supérieur.

```
> restart;
  solve(x=1+1/x,x);
```

$$\frac{1}{2}+\frac{\sqrt{5}}{2}, \frac{1}{2}-\frac{\sqrt{5}}{2}$$

Ici, Maple donne les solutions exactes et explicites de l'équation. Le type de résultat renvoyé par Maple est une *séquence* . On peut demander les valeurs approchées des solutions par :

```
> evalf(%);
```

$$1.618033988, -0.6180339880$$

Malheureusement, en degré supérieur ou égal à 4, cela ne se passe pas toujours aussi bien.

Essayons de résoudre l'équation : x^7-x^6+x^2-1=0.

```
> S:=solve(x^7-x^6+x^2-1=0,x);
```

$$S := 1, \mathrm{RootOf}(\_Z^6+\_Z+1, index=1), \mathrm{RootOf}(\_Z^6+\_Z+1, index=2), \mathrm{RootOf}(\_Z^6+\_Z+1, index=3), \mathrm{RootOf}(\_Z^6+\_Z+1, index=4), \mathrm{RootOf}(\_Z^6+\_Z+1, index=5), \mathrm{RootOf}(\_Z^6+\_Z+1, index=6)$$

Parfois, lorsque Maple ne peut pas trouver de solution explicite, ou lorsque celle-ci est trop compliquée pour être utilisable, les solutions sont exprimées à l'aide de la fonction *RootOf* ("racine de"), qui "représente" toutes les racines à la fois et aucune en particulier. Factorisons le polynôme de départ.

```
> factor(x^7-x^6+x^2-1);
```

$$(x-1)(x^6+x+1)$$

Une racine évidente est 1, les autres racines sont celles de x^6+x+1 (au nombre de 6), que Maple refuse d'expliciter. Cependant, ce n'est pas complètement un échec car :
* certaines fonctions de Maple sont capables de travailler avec *RootOf,* et on peut continuer alors à "faire des calculs" :

```
> alias(alpha=RootOf(x^6+x+1));
```

$$\alpha$$

```
> alpha^8;
  simplify(alpha^8);
```

$$\alpha^8$$

$$-\alpha^2(1+\alpha)$$

* si l'équation ne comporte aucun paramètre, on peut obtenir des valeurs numériques approchées des racines :

```
> evalf(S);
```

$$1., 0.9454023333+0.6118366938\,I, -0.1547351445+1.038380754\,I, -0.7906671888+0.3005069203\,I, -0.7906671888-0.3005069203\,I, -0.1547351445-1.038380754\,I, 0.9454023333-0.6118366938\,I$$

La fonction *allvalues* permet parfois de "déterminer" toutes les racines désignées par le *RootOf :* quand c'est possible, Maple donne des expressions exactes.

```
> T:=solve(-x**4+5*x**3+x**2-1,x);
```

$$T := \mathrm{RootOf}(\_Z^4-5\,\_Z^3-\_Z^2+1, index=1), \mathrm{RootOf}(\_Z^4-5\,\_Z^3-\_Z^2+1, index=2), \mathrm{RootOf}(\_Z^4-5\,\_Z^3-\_Z^2+1, index=3), \mathrm{RootOf}(\_Z^4-5\,\_Z^3-\_Z^2+1, index=4)$$

```
> allvalues([T]):
  allvalues([T])[1];
```

$$\frac{5}{4}+\frac{\sqrt{3}\sqrt{\frac{83\,(2980+12\sqrt{60693})^{(1/3)}+2\,(2980+12\sqrt{60693})^{(2/3)}+104}{(2980+12\sqrt{60693})^{(1/3)}}}}{12}-\Bigg(\Bigg(\Bigg(498$$

$$(2980+12\sqrt{60693})^{(1/3)}$$

$$\sqrt{\frac{83\,(2980+12\sqrt{60693})^{(1/3)}+2\,(2980+12\sqrt{60693})^{(2/3)}+104}{(2980+12\sqrt{60693})^{(1/3)}}}-6$$

$$\sqrt{\frac{83\,(2980+12\sqrt{60693})^{(1/3)}+2\,(2980+12\sqrt{60693})^{(2/3)}+104}{(2980+12\sqrt{60693})^{(1/3)}}}$$

$$(2980+12\sqrt{60693})^{(2/3)}$$

$$-312\sqrt{\frac{83\,(2980+12\sqrt{60693})^{(1/3)}+2\,(2980+12\sqrt{60693})^{(2/3)}+104}{(2980+12\sqrt{60693})^{(1/3)}}}$$

$$\left.+2610\sqrt{3}\,(2980+12\sqrt{60693})^{(1/3)}\right) \Bigg/ \left((2980+12\sqrt{60693})^{(1/3)}\right.$$

$$\left.\left.\sqrt{\frac{83\,(2980+12\sqrt{60693})^{(1/3)}+2\,(2980+12\sqrt{60693})^{(2/3)}+104}{(2980+12\sqrt{60693})^{(1/3)}}}\right)\right)^{(1/2)} / 12$$

Maple peut aussi résoudre quelques **équations non polynomiales** simples.

```
> solve(exp(x)=1+x,x);
```

$$0$$

Pour les équations plus compliquées, Maple peut trouver une partie des solutions, aucune des solutions, ou meme des solutions qui sont fausses ! Voir la feuille d'exercices.

1.2 Résoudre un système d'équations

Pour résoudre un système d'équations, on utilise la commande *solve* avec la syntaxe :
solve({equations},{variables})
Maple retourne les solutions sous la forme {variable1=expression1, variable2=expression2,...}.

Exemple : le système {x+y+z=4, x+y-z=1, x-y-z=-3}.

```
> solve({x+y+z=4,x+y-z=1,x-y-z=-3},{x,y,z});
```

$$\{x=\frac{1}{2}, z=\frac{3}{2}, y=2\}$$

Ici, Maple trouve une solution unique.

Dans certains cas, on a une famille de solutions, qui s'expriment en fonction de certaines des variables.
Exemple : le système { x-3y=2,3x-9y=6}.

```
> solve({x-3*y=2,3*x-9*y=6});
```

$$\{x=3\,y+2, y=y\}$$

Maple a choisi de garder l'inconnue y comme paramètre, et a pu exprimer toutes les solutions en fonction de ce paramètre.

1.3 Résolution approchée

Lorsqu'on ne peut résoudre de manière exacte une équation (sans paramètre), on peut essayer la fonction *fsolve* pour obtenir des solutions *approchées* .

```
> fsolve(tan(sin(x))=1,x);
```

$$0.9033391108$$

On peut spécifier un intervalle sur lequel on cherche les solutions :

```
> fsolve(x^5-7*x^4+3*x^2+15,x,0..10);
```

$$1.386962979, 6.931051639$$

Par défaut, Maple cherche des solutions approchées réelles. On peut aussi lui demander des solutions approchées complexes :

```
> fsolve(x^5-7*x^4+3*x^2+15,x,complex);
```

$$-1.242272278, -0.03787117016-1.120100886\,I, -0.03787117016+1.120100886\,I,$$
$$1.386962979, 6.931051639$$

2. Dérivation et intégration

2.1 Dériver une expression

On dérive une *expression* à l'aide de la commande *diff*, par la syntaxe
diff(expression_en_variable,variable).

```
> diff(1/(1+x^2),x);
```

$$-\frac{2\,x}{(1+x^2)^2}$$

Maple retourne alors une *expression* en la variable.

2.2 Dériver une fonction

On dérive une *fonction* (préalablement définie à l'aide de -> ou de unapply) par l'opérateur *D.*

```
> restart; f:=x->cos(x)+5*x^2-3;
```

$$f := x \to \cos(x)+5\,x^2-3$$

```
> g:=D(f);
```

$$g := x \to -\sin(x)+10\,x$$

Maple retourne alors une *fonction*.

```
> g(Pi);
```

$$10\,\pi$$

2.3 Calculer une primitive

On utilise la commande *int*, dont la syntaxe est *int(expression_en_variable,variable)*. Le résultat est sans constante d'intégration. C'est une expression en la variable.

```
> int(tan(x),x);
```

$$-\ln(\cos(x))$$

Parfois, Maple ne sait pas calculer exactement une primitive.

```
> int(exp(x)*cos(x)^n,x);
```

$$\int \mathbf{e}^x \cos(x)^n\,dx$$

Vérifier qu'ici, si l'on donne une valeur précise à n, Maple sait mener le calcul.
Parfois, Maple exprime les primitives en fonction de fonctions dites spéciales :

```
> int(sin(x)/x,x);
```

$$\mathrm{Si}(x)$$

C'est la fonction sinus intégral (pour en savoir plus, consulter l'aide sur Si). Maple la connait, il sait par exemple en tracer le graphe :

```
> plot(Si);
```



2.4 Calculer une intégrale

Pour calculer l'intégrale de a à b de f, on utilise la commande*int* avec la syntaxe *int(expression_en_variable, variable=a..b).*

```
> restart; int(exp(x)*cos(x)^3,x=0..Pi/2);
```

$$\frac{3}{10}\mathbf{e}^{\left(\frac{\pi}{2}\right)}-\frac{2}{5}$$

Les bornes ne sont pas nécessairemment des constantes. Elles peuvent etre des indéterminées :

```
> int(x^2*cos(x),x=0..N*Pi);
```

$$N^2\,\pi^2\sin(N\,\pi)-2\sin(N\,\pi)+2\,N\,\pi\cos(N\,\pi)$$

Elles peuvent etre aussi +l'infini (noté *infinity*) ou -l'infini (noté - *infinity*).

```
> int(exp(-x^2)*cos(x),x=0..infinity);
```

$$\frac{1}{2}\sqrt{\pi}\,\mathbf{e}^{(-1/4)}$$

(mais vous n'aurez pas de telles intégrales dans votre cours de mathématiques du premier semestre)

Université Paris 7 — **Premier semestre 2005-2006**
Licence 1ère année — **MK1 - Maple**

Feuille d'exercices n°3

Exercice 1. Résoudre l'équation $x^2 + 2bx + c = 0$ d'inconnue x et de paramètres b et c. Donner la liste des solutions. En utilisant la commande *subs*, trouver les solutions pour $b = 1$ et $c = 2$. Que constatez-vous ? Que remarquez-vous sur les solutions générales de l'équation ?

Exercice 2. Résoudre l'équation $x^3 - 5x^2 - 2 = 0$. Donner une valeur approchée de la première solution donnée par Maple.

Exercice 3. Cet exercice a pour but de vous faire sentir les limites de Maple pour la résolution d'équations non polynomiales. Résolvez chaque équation et réfléchissez sur les résultats donnés par Maple.

1) Equation $\tan(x) = \sqrt{3}$.
2) Equation $\cos(x) = x$.
3) Equation $\sqrt{x^2} = x$.

Exercice 4. Chercher l'ensemble des nombres complexes z tels que $|z| = |z - 1|$.

Exercice 5.

1) Calculer les dérivées des fonctions suivantes : $x \mapsto \tan x$ et $x \mapsto \ln(x^3 + ax^2 + 1)$, où a est un paramètre réel.
2) Soient $f : x \mapsto 1 + x^5$ et $g : x \mapsto ln(x)$. Calculer la dérivée de $g \circ f$.
3) Soit f comme précédemment. Construire la séquence des $f'(i)$, pour i de 1 à 20.

Exercice 6. Calculer les primitives des fonctions suivantes : $x \mapsto \frac{\sin(x)\tan(x)+\cos(x)}{\sin(x)-\cos(x)^2}$ et $x \mapsto \ln(x^2 + 1)$.

Exercice 7. Calculer $\int_0^1 e^{-t^2} dt$. Donner une valeur approchée de l'intégrale.

Correction du TP3

Exercice 1

```
> restart; S:=solve(x^2+2*b*x+c=0,x);
```

$$S := -b+\sqrt{b^2-c}, -b-\sqrt{b^2-c}$$

La *liste* des solutions est :

```
> L:=[S];
```

$$L := [-b+\sqrt{b^2-c}, -b-\sqrt{b^2-c}]$$

Pour b=1 et c=2, les solutions sont :

```
> subs(b=1,c=2,L);
```

$$[-1+I, -1-I]$$

```
> subs(b=1,c=2,S);
Error, wrong number (or type) of parameters in function subs
```

Attention, la fonction *subs* permet de substituer des valeurs dans des *ensembles* ou des *listes* d'équations, pas dans des séquences : c'est pour cela qu'on a transformé la séquence S des solutions retournée par *solve* en une liste L.

On constate que pour b=1 et c=2, les solutions sont des nombres complexes (avec une partie imaginaire non nulle). Dans ce cas, le discriminant b^2-c est un nombre négatif. Pour Maple, sa "racine carrée" est un nombre complexe. Conclusion : Maple utilise le signe "racine carrée" meme pour les nombres négatifs ! En fait, il utilise meme ce symbole pour les nombres complexes. Si on se donne un nombre complexe, il a deux "racines carrées" possibles : Maple choisit celle des deux (qu'on note z) telle que :
Re(z)>=0 ou (Re(z)=0 et Im(z)>=0).

Exercice 2

```
> restart; S:=solve(x^3-5*x^2-2=0,x);
```

$$S := \frac{(152+3\sqrt{831})^{(1/3)}}{3}+\frac{25}{3\,(152+3\sqrt{831})^{(1/3)}}+\frac{5}{3}, -\frac{(152+3\sqrt{831})^{(1/3)}}{6}-\frac{25}{6\,(152+3\sqrt{831})^{(1/3)}}+\frac{5}{3}+\frac{1}{2}I\sqrt{3}\left(\frac{(152+3\sqrt{831})^{(1/3)}}{3}-\frac{25}{3\,(152+3\sqrt{831})^{(1/3)}}\right),$$
$$-\frac{(152+3\sqrt{831})^{(1/3)}}{6}-\frac{25}{6\,(152+3\sqrt{831})^{(1/3)}}+\frac{5}{3}-\frac{1}{2}I\sqrt{3}\left(\frac{(152+3\sqrt{831})^{(1/3)}}{3}-\frac{25}{3\,(152+3\sqrt{831})^{(1/3)}}\right)$$

Une valeur approchée de la première solution est :

```
> evalf(S[1]);
```

$$5.077574223$$

Exercice 3

```
> restart; solve(tan(x)=sqrt(3),x);
```

$$\frac{\pi}{3}$$

Maple nous donne une solution. Pourtant, la fonction tangente est 2*Pi périodique. Représentons graphiquement les deux fonctions :

```
> plot([tan(x),sqrt(3)],x, y=-5..5,discont=true);
```



On voit que l'équation a en réalité une infinité de solutions ! Ce sont les translatés de Pi/3 par 2*Pi. Vérifions-le par exemple pour Pi/3 + 2*Pi :

```
> tan(Pi/3 + 2*Pi);
```

$$\sqrt{3}$$

Dans ce cas, Maple ne nous a pas donné toutes les solutions.

```
> restart; s:=solve(cos(x)=x,x);
```

$$s := \mathrm{RootOf}(-\_Z+\cos(\_Z))$$

Ici, Maple refuse d'expliciter la solution.

```
> plot([cos(x),x],x);
```



On peut cependant demander une valeur approchée :

```
> evalf(s);
```

$$0.7390851332$$

```
> restart; solve(sqrt(x^2)=x,x);
```

$$x$$

Cette réponse signifie que tout x est solution, ce qui est faux si x est négatif : par exemple

```
> subs(x=-1,sqrt(x^2)=x);
```

$$1 = -1$$

Malheureusement, ici, Maple nous a donné des solutions qui n'en sont pas !

Exercice 4

L'approche naive est de demander à Maple de résoudre directement |z| = |z-1|.

```
> restart;
  eq:=abs(z)=abs(z-1);
  solve(eq,z);
```

$$eq := |z| = |z-1|$$

$$\frac{1}{2}$$

Manifestement, il manque des solutions, par exemple z=1/2+i :

```
> simplify(subs(z=1/2+I,eq));
```

$$\frac{\sqrt{5}}{2} = \frac{\sqrt{5}}{2}$$

En fait, si on regarde l'aide de Maple à propos de *solve*, on y lit que si *abs* est utilisé dans une commande *solve*, Maple suppose que l'argument de la fonction *abs* est à valeur réelle. Ce qui n'est pas le cas ici, puisqu'on s'intéresse aux solutions complexes z de l'équation. On va tenter une deuxième approche.

```
> z:=x+I*y;
  eq;
  solve(eq,{x,y});
```

$$z := x + y\,I$$

$$|x + y\,I| = |x + y\,I - 1|$$

$$\{x = -y\,I + \frac{1}{2}, y = y\}$$

Cette fois, Maple nous donne une famille de solutions. Malheureusement, d'après son expression, x peut etre un nombre complexe. Or, on veut que x et y soient les parties réelles et imaginaires de z, c'est-à-dire des nombres réels. Le problème vient du fait que Maple ne sait rien à propos de x et y, et suppose par défaut que ce sont des complexes ! Qu'à cela ne tienne, faisons supposer à Maple que ce sont des réels.

```
> assume(x,real);assume(y,real);
  solve(eq,{x,y});
```

$$\{y\sim = y\sim, x\sim = \frac{1}{2}\}$$

Cette fois, c'est bon. L'ensemble des solutions est {z=1/2+I*y, pour y réel quelconque}. Géométriquement, c'est la droite verticale d'abscisse 1/2.

Exercice 5

```
> restart; diff(tan(x),x); diff(ln(x^3+a*x^2+1),x);
```

$$1 + \tan(x)^2$$

$$\frac{3\,x^2 + 2\,a\,x}{x^3 + a\,x^2 + 1}$$

```
> restart; f:=x->1+x^5; g:=x->ln(x);
```

$$f := x \rightarrow 1 + x^5$$

$$g := x \rightarrow \ln(x)$$

La fonction composée gof s'obtient par g@f.

```
> (g@f)(x);
```

$$\ln(1 + x^5)$$

La dérivée de gof peut s'obtenir de deux façons, par :

```
> diff((g@f)(x),x);
```

$$\frac{5\,x^4}{1 + x^5}$$

ou par :

```
> D(g@f);
  D(g@f)(x);
```

$$\left(\left(x \rightarrow \frac{1}{x}\right)@f\right)(x \rightarrow 5\,x^4)$$

$$\frac{5\,x^4}{1 + x^5}$$

Pour la séquence de f'(i), i=1 à 20 : la commande :

```
> diff(f(x),i);
```

$$0$$

... ne fournit pas f'(i) : en fait, elle demande à Maple de dériver l'expression f(x) par rapport à la variable i. Comme f(x) ne dépend pas de i, cette dérivée est nulle. Pour y remédier, il faut définir la *fonction* f', par exemple grace à l'opérateur *D*.

```
> fprime:=D(f);
  fprime(i);
```

$$fprime := x \rightarrow 5\,x^4$$

$$5\,i^4$$

On peut alors construire la séquence à l'aide de *seq* :

```
> seq(fprime(i),i=1..20);
```

5, 80, 405, 1280, 3125, 6480, 12005, 20480, 32805, 50000, 73205, 103680, 142805, 192080, 253125, 327680, 417605, 524880, 651605, 800000

Exercice 6

```
> int((sin(x)*tan(x)+cos(x))/(sin(x)-cos(x)^2),x);
```

$$-\frac{1}{2}\ln(\sin(x) - 1) - \frac{1}{2}\ln(1 + \sin(x)) + \frac{1}{2}\ln(\sin(x) - 1 + \sin(x)^2)$$
$$- \frac{1}{5}\sqrt{5}\,\mathrm{arctanh}\left(\frac{1}{5}(1 + 2\sin(x))\sqrt{5}\right)$$

```
> int(ln(x^2+1),x);
```

$$x\ln(x^2 + 1) - 2\,x + 2\arctan(x)$$

Exercice 7

```
> restart; n:=int(exp(-t^2),t=0..1);
```

$$n := \frac{1}{2}\,\mathrm{erf}(1)\sqrt{\pi}$$

Maple nous donne un résultat exprimé à l'aide de la fonction spéciale *erf* (regardez l'aide sur *erf* pour avoir plus d'informations sur cette fonction). On peut lui demander une valeur *approchée* de l'intégrale :

```
> evalf(n);
```

$$0.7468241330$$

C. Armana
armana@math.jussieu.fr

MK1 "Calcul formel" Maple

TP4 : Limites, continuité

Commandes pour les caractères spéciaux
Voici les commandes pour obtenir les caractères qui ne sont pas présents sur les claviers Mac.
pour { : Alt (pour [: Alt Shift (
pour } : Alt) pour] : Alt Shift)

Controle des connaissances - changement de planning
Avant l'examen, il y aura plusieurs évaluations :
- un controle C la semaine du 17 octobre (durée : 30 min)
- un partiel P : le 4 novembre pour le groupe 1D5, le 8 novembre pour le groupe 1A5 (durée : 1h)
- un devoir à la maison DM : le sujet est donné la semaine du 21 novembre, à rendre la semaine du 5 décembre

La note finale tiendra compte de l'examen et du controle continu par la formule :
note finale = (4E + 2P + C + DM)/8
Rappel : l'année n'est pas validée lorsqu'il y a une absence injustifiée à un examen.

Et surtout, n'oubliez pas de vous (et de me) poser des questions !

1. Limites

Pour calculer des limites d'expressions en une variable, on utilise la commande *limit* avec la syntaxe *limit(expression_en_x,x=a).* La valeur de *x* pour laquelle on cherche la limite de l'expression est *a*.

```
> restart; f:=x->(1-cos(x))/x^2 ;
```

$$f := x \rightarrow \frac{1-\cos(x)}{x^2}$$

La fonction f n'est pas définie en 0, mais a une limite (finie) en 0 que Maple peut calculer :

```
> f(0);
  limit(f(x),x=0);
Error, (in f) numeric exception: division by zero
```

$$\frac{1}{2}$$

a peut valoir + l'infini et -l'infini (infinity et -infinity) :

```
> limit(ln(x),x=infinity);
```

$$\infty$$

```
> limit(1/x,x=0);
```

$$undefined$$

Ici, Maple répond 'undefined' car la fonction x->1/x n'a pas de limite en 0. Par contre, elle a une limite à gauche et une limite à droite :

```
> limit(1/x,x=0,left); limit(1/x,x=0,right);
```

$$-\infty$$

$$\infty$$

Pour définir une suite dont on connaît le terme général, on procède comme pour une fonction "ordinaire" : par exemple, pour la suite u_n définie par u_n=cos(1/n) :

```
> u:=n->cos(1/n);
```

$$u := n \rightarrow \cos\left(\frac{1}{n}\right)$$

On peut alors utiliser la commande *limit* pour calculer la limite :

```
> limit(u(n),n=infinity);
```

$$1$$

2. Continuité d'une fonction

Maple dispose de commandes pour étudier la continuité et la discontinuité des fonctions de **R** dans **R**.
La commande *iscont* permet de tester la continuité d'une expression en *x* sur un intervalle donné. Le résultat donné par Maple est un booléen : *true* (vrai) si la fonction est continue sur l'intervalle, *false* (faux) si elle ne l'est pas. La syntaxe est *iscont(expression_en_x,x=a..b)* .

```
> restart; f:=x->1/(x-1);
  iscont(f(x),x=0..2);
```

$$f := x \rightarrow \frac{1}{x-1}$$

$$false$$

Ici, f n'est pas continue sur l'intervalle 0..2 car elle n'est pas continue en 1. Attention, par défaut, Maple étudie la continuité sur l'intervalle *ouvert*]a,b[. Par exemple, :

```
> iscont(f(x),x=0..1);
```

$$true$$

car *f* n'est pas continue au point 1, mais l'est sur l'intervalle ouvert en 1]0,1[. Il est possible de lui demander de travailler sur l'intervalle *fermé* [a,b] :

```
> iscont(f(x),x=0..1,'closed');
```

$$false$$

La commande *discont* donne l'ensemble des points de discontinuité. Sa syntaxe est *discont(expression_en_x,x).*

```
> discont(f(x),x);
```

$$\{1\}$$

```
> discont(1/sin(x),x);
```

$$\{\pi\_Z1\sim\}$$

(Quel est l'ensemble des points de discontinuité ? Essayez de comprendre la dernière réponse de Maple)

Université Paris 7 — **Premier semestre 2005-2006**
Licence 1ère année — **MK1 - Maple**

Feuille d'exercices n°4

Exercice 1. Etude de fonction à l'aide de Maple

Soit f la fonction définie pour tout réel $x > 0$ par :

$$f(x) = x^{\frac{x}{1-x}}.$$

1) Donnez le domaine de définition de f. Vérifiez que f est continue sur son domaine de définition Df.
2) Etudiez les limites de f aux bornes des intervalles qui composent Df. Le graphe de f admet-il une asymptote ? Si oui, quelle est-elle ?
3) Calculez la dérivée de f. Etudiez son signe (on pourra arranger l'expression de la dérivée à l'aide de la commande *normal*, puis utiliser une fonction auxiliaire g pour l'étude du signe). Qu'en déduisez-vous pour f ?
4) Tracez le graphe de f.
5) Expliquez (d'un point de vue mathématique) comment prolonger f par continuité aux points 0 et 1.
6) Etudiez la dérivabilité de ce prolongement aux points 0 et 1. Quelle interprétation géométrique pouvez-vous faire ? Tracez le graphe correspondant.

Correction du TP n°4

Exercice 1

1)

```
> restart; f:=x-> x^(x/(1-x));
```

$$f := x \to x^{\left(\frac{x}{1-x}\right)}$$

```
> discont(f(x),x);
```

$$\{0, 1\}$$

La fonction f, définie pour x>0, n'est pas continue en 0 et en 1. Son domaine de définition est]0,1[U]1, +oo[.

```
> iscont(f(x),x=0..1);
  iscont(f(x),x=1..infinity);
```

true

true

Donc f est continue sur son domaine de définition. Remarque : par défaut, Maple considère des intervalles ouverts ; si on ferme un des intervalles en 1, la réponse n'est plus la même car f n'est pas continue en 1 :

```
> iscont(f(x),x=0..1,'closed');
```

false

2) On étudie la limite de f en 0, 1 et +oo.

```
> limit(f(x),x=0,right);
```

$$1$$

En 1, il faut distinguer les limites à gauche et à droite :

```
> limit(f(x),x=1,left);
  limit(f(x),x=1,right);
```

$$\mathbf{e}^{(-1)}$$

$$\mathbf{e}^{(-1)}$$

On remarque que les deux limites à gauche et à droite sont égales.

```
> limit(f(x),x=infinity);
```

$$0$$

Donc la courbe représentant la fonction f admet une asymptote au voisinage de + l'infini, et cette asymptote est l'axe des abscisses (y=0).

3) La dérivée de f est :

```
> diff(f(x),x);
```

$$x^{\left(\frac{x}{1-x}\right)}\left(\left(\frac{1}{1-x}+\frac{x}{(1-x)^2}\right)\ln(x)+\frac{1}{1-x}\right)$$

```
> normal(%);
```

$$\frac{x^{\left(-\frac{x}{-1+x}\right)}(\ln(x)+1-x)}{(-1+x)^2}$$

Le dénominateur est positif, ainsi que le terme $x^{\left(-\frac{x}{x-1}\right)}$. La dérivée est donc du même signe que la fonction $\ln(x)+1-x$.

```
> g:=x->ln(x)+1-x;
```

$$g := x \to \ln(x)+1-x$$

```
> solve(g(x)=0,x);
```

$$1$$

```
> solve(g(x)<0,x);
```

$$\text{RealRange}(\text{Open}(0), \text{Open}(1)), \text{RealRange}(\text{Open}(1), \infty)$$

La fonction g s'annule en 1. Elle est strictement négative sur l'intervalle]0,1[U]1, +oo[.

```
> solve(g(x)>0,x);
```

Elle n'est jamais strictement positive sur *Df*. Sur *Df*, elle est négative ou nulle. Cela signifie que f est décroissante sur *Df*.

4) Le graphe de f est :

```
> plot(f(x),x=0..10);
```



5) On a vu que la limite de f au point 0 est 1, donc on peut prolonger f de façon continue en 0 par f(0)=1.

On a vu que les limites de f à gauche et à droite en 1 sont égales et valent $e^{(-1)}$.On peut donc prolonger f de façon continue en 1 par $f(1)=e^{(-1)}$.

6) Cherchons si f, prolongée par continuité, est dérivable en 0. Pour cela, on regarde la limite du taux d'accroissement en 0 :

```
> f(0):=1; f(1):=exp(-1);
```

$$f(0) := 1$$

$$f(1) := \mathbf{e}^{(-1)}$$

(cela permet de définir f aux points 0 et 1)

```
> limit( (f(x)-f(0)) / (x-0) , x=0, right);
```

$$-\infty$$

La limite est infinie, donc le prolongement de f n'est pas dérivable en 0. Géométriquement, cela signifie que la tangente au graphe de f en le point (0,1) est la droite Oy.

```
> limit ( (f(x)-f(1)) / (x-1), x=1, left);
  limit ( (f(x)-f(1)) / (x-1), x=1, right);
```

$$-\frac{1}{2}\mathbf{e}^{(-1)}$$

$$-\frac{1}{2}\mathbf{e}^{(-1)}$$

Les limites du taux d'accroissement en 1 à gauche et à droite sont finies et égales, donc le prolongement de f est dérivable en 1, et f '(1) = $-\frac{1}{2\,e}$. Géométriquement, cela signifie que la droite tangente au point (1, $\frac{1}{e}$) a pour pente $\frac{1}{e}$. L'équation de cette tangente est :

```
> tangente:=x->f(1)-1/(2*exp(1))*(x-1);
```

$$tangente := x \rightarrow \mathrm{f}(1) - \frac{1}{2}\,\frac{x-1}{\mathbf{e}}$$

On trace sur un meme graphique la fonction et sa tangente :

```
> plot([f(x),tangente(x)],x=0..10,color=[red,blue]);
```



C. Armana
armana@math.jussieu.fr

MK1 "Calcul formel" Maple

TP5 : Programmation

Commandes pour les caractères spéciaux
Voici les commandes pour obtenir les caractères qui ne sont pas présents sur les claviers Mac.
pour { : Alt (pour [: Alt Shift (
pour } : Alt)   pour] : Alt Shift)

But du TP5
Maple est un langage de calcul formel, mais c'est aussi un langage de programmation. Nous allons voir comment créer des programmes avec Maple. Pour cela, nous allons utiliser des structures communes à la plupart des langages de programmation : procédures, tests*(if)*, boucles (*for*, *while*), fonctions récursives.

Et surtout, n'oubliez pas de vous (et de me) poser des questions !

1. Les booléens

On appelle *expression booléenne* une expression dont l'évaluation conduit ou bien à la valeur *true* (vrai) ou bien à la valeur *false* (faux). L'évaluation des expressions booléennes se fait par la commande *evalb*.
Les expressions booléennes sont très utiles en programmation car elles permettent d'effectuer des tests qui détermineront la suite des instructions à effectuer. Voici différentes méthodes pour fabriquer des expressions booléennes..

1.1 Les opérateurs de comparaison

On peut former une expression booléenne en comparant deux expressions de même type à l'aide d'un des opérateurs suivants :
= (égale)
<> (est différent de)
< (est strictement inférieur à)
> (est strictement supérieur à)
<= (est inférieur ou égal à)
>= (est supérieur ou égal à)
Dans l'exemple suivant, on définit une expression booléenne avant de l'évaluer.

```
> restart;
  bool:=1<2;
```

$$bool := 1 < 2$$

```
> evalb(bool);
```

$$true$$

```
> evalb(Pi=3.14);
```

$$false$$

Parfois, quand Maple ne dispose pas d'informations suffisantes, il ne peut pas évaluer l'expression booléenne ; par exemple :

```
> evalb(x>=y);
```

$$y - x \leq 0$$

Attention, la fonction *evalb* ne sait pas faire des calculs algébriques (contrairement à *simplify*), comme le montre l'exemple suivant :

```
> evalb( x^2-y^2 = (x-y) * (x+y) );
```

$$false$$

1.2 Les opérateurs logiques

Ils permettent de modifier des expressions booléennes :
not (non)
and (et)
or (ou inclusif)
(remarque : ils ne nécessitent pas d'utiliser *evalb*)

```
> not(1<2);
```

$$false$$

```
> 1<2 and 4<3;
```

$$false$$

1.3 Les fonctions booléennes

Les fonctions booléennes sont des commandes de Maple qui renvoient *true* ou *false*. Par exemple, nous avons vu la fonction *iscont*, qui permet de tester la continuité d'une fonction sur un intervalle :

```
> iscont(tan(x),x=0..1);
```

$$true$$

Il y a aussi la fonction *type*, qui permet de tester si une expression est d'un type donné. Par exemple, la commande suivante permet de tester si Pi est un entier (*integer*) :

```
> type(Pi,integer);
```

$$false$$

(pour en savoir plus sur la commande *type* et les différents types possibles, consultez l'aide)

2. Les procédures

Une *procédure* est un petit programme qui a un nom, des entrées et une sortie. Au coeur du programme, il y a une suite d'instructions à accomplir. Une procédure utilise des variables qui lui sont propres (variables locales). Par exemple, la procédure suivante, qui se nomme différence, prend en entrées deux nombres x et y et renvoie leur différence.

```
> différence:=proc(x,y)
  RETURN(x-y);
  end;
```

$$\textit{différence} := \mathbf{proc}(x, y)\ \mathrm{RETURN}(x - y)\ \mathbf{end\ proc}$$

```
> différence(4,5);
```

$$-1$$

(pour aller à la ligne lorsque vous tapez une procédure, utilisez les touches Maj Entrée)
* *proc* signifie que l'on est en train de définir une procédure (attention, il ne faut pas mettre de point-virgule après *proc*).
* *RETURN* (en majuscules !) affiche le résultat de la procédure.

* *end* signifie que la définition de la procédure est terminée.

Voici une procédure qui se nomme sommeproduit, qui prend en entrées trois nombres *a*,*b* et *c*, et qui renvoie la liste formée de leur somme et leur produit. Au cours de la procédure, on stocke les calculs intermédiaires dans des variables locales, qui sont auparavant déclarées par la ligne *local*.

```
> sommeproduit:=proc(a,b,c)
  local s,p;
  s:=a+b+c;
  p:=a*b*c;
  RETURN([s,p]);
  end:
> sommeproduit(2,3,4);
```

[9, 24]

Recommandation

En programmation, il est important de bien réfléchir à ce que l'on veut faire avant de passer sur la machine. Avec un papier et un crayon, réfléchissez d'abord au programme : ses entrées, ses instructions, ses sorties. Ensuite, implémentez-le sur Maple. Cela permet de mieux distinguer les erreurs qui viennent de la conception du programme de celles dues à une mauvais syntaxe des commandes.

3. Le test (if)

La structure *if* permet de tester si une condition est vérifiée. Elle peut être employée dans une procédure. Elle commence par un *if* et se termine par un *fi*.Voici un exemple avec une procédure qui calcule le minimum de deux nombres :

```
> minimum:=proc(x,y)
  if x<y then RETURN(x) else RETURN(y) fi ;
  end:
> minimum(7,-2);
```

-2

Les syntaxes possibles sont les suivantes. A chaque fois, les conditions sont des expressions booléennes et les instructions sont des commandes Maple.

* Pour une exécution conditionnelle :

if condition *then* instructions *fi* ;

Maple évalue d'abord l'expression booléenne condition. Si le résultat est *true*, alors les instructions sont effectuées. Sinon, Maple passe à la suite (après le *fi*).

* Pour un choix binaire :

if condition *then* instructions_1 *else* instructions_2 *fi*;

Maple évalue d'abord l'expression booléenne condition_1. Si le résultat est *true*, alors les instructions_1 sont effectuées. Sinon, Maple effectue les instructions_2.

* Pour un choix multiple :

if condition_1 *then* instructions_1
elif condition_2 *then* instructions_2
...
elif condition_n *then* instructions_n
else instructions_(n+1)
fi

Maple évalue l'expression booléenne condition_1. Si le résultat est *true*, alors les instructions_1 sont effectuées et Maple passe à la suite (après le *fi*). Si c'est *false*, Maple évalue l'expression booléenne condition_2 et procède de même. Si aucune des expressions booléennes condition_1,..,condition_n n'est vraie, Maple effectue les instructions_(n+1).

(rappels pour les non-anglophones : if=si, then=alors, else=sinon, elif est une contraction de else if)

4. Les boucles (for, while)

Ce sont des structures itératives qui permettent de répéter un groupe de commandes un certai nombre de fois. Elles peuvent être employées dans des procédures.

4.1 For

Lorsqu'on sait à l'avance le nombre de répétitions, on utilise *for*, dont la syntaxe est :

for i *from* début *to* fin *do* instructions *od* ;

Cela signifie: pour i allant de début à fin, exécuter les instructions (i est une variable choisie par l'utilisateur). En général, début et fin sont des nombres entiers. Par exemple, voici une boucle *for* qui, pour i variant de 1 à 10, affiche i! .

```
> for i from 1 to 10 do
    i!
  od;
```

1
2
6
24
120
720
5040
40320
362880
3628800

Par défaut, le pas est 1. On peut spécifier un autre pas d'incrémentation à l'aide de *by*. Par exemple, si on décide d'aller de 2 en 2 :

```
> for i from 1 to 10 by 2 do
  i!
  od;
```

1
6
120
5040
362880

4.2 While

On utilise la boucle *while* quand on doit déterminer "en cours de route" le nombre de répétitions. Sa syntaxe est :

while condition *do* instructions *od ;*

La condition est une expression booléenne et les instructions sont des commandes Maple. Maple évalue l'expression booléenne condition. Tant qu'elle est vraie, il exécute les instructions. Quand elle est fausse, il passe à la suite (après le *od*).
(rappel pour les non-anglophones : while = tant que)

Par exemple, on souhaite calculer le plus petit entier *n* tel que la somme des entiers de 1 à *n* soit supérieure ou égale à 1000.

```
> somme:=0: a:=0:
  while somme<1000 do
      a:=a+1;
      somme:=somme+a;
  od:
  a;
```

$$45$$

5. La récursivité

Maple accepte des fonctions et des procédure *récursives* , c'est-à-dire qui s'appellent elles-memes
Soyez particulièrement soigneux lorsque vous utilisez la récursivité, c'est une source d'erreurs importante. Il faut notamment prévoir un "cas d'arret" (un cas particulier sans appel récursif) et etre sur qu'il se réalisera ; faute de quoi la récurrence ne s'arrete jamais !
Par exemple, la factorielle est définie sur les entiers naturels par la récurrence : n! = n * (n-1) ! avec 0!=1. On peut programmer la factorielle dans une procédure récursive :

```
> factorielle:=proc(n)
  if n=0 then RETURN(1)   # cas d'arret
  else RETURN(n*factorielle(n-1))  # on appelle la procédure avec
  l'entrée n-1
  fi;
  end:
> factorielle(0); factorielle(10);
```

$$1$$

$$3628800$$

```
> factorielle(-1);
Error, (in factorielle) too many levels of recursion
```

Voilà le message d'erreur qu'on obtient quand aucun cas d'arret n'a été prévu !

On aurait également pu définir la factorielle dans une fonction :

```
> f:=n->n*f(n-1);    # relation de récurrence
```

$$f := n \to n\, \mathrm{f}(n-1)$$

```
> f(0):=1;     # initialisation
```

$$\mathrm{f}(0) := 1$$

```
> f(10);
```

$$3628800$$

Un autre exemple : la suite récurrente définie par :

si $n \leq 0$, $u_n = 3$

si $0 < n$, $u_n = \frac{4\, u_{n-1} + 1}{5}$

```
> u:=proc(n)
  if n<=0 then RETURN(3) else RETURN((4*u(n-1)+1)/5) fi;
  end:
> u(-3); u(0); u(5); u(15);
```

$$3$$

$$3$$

$$\frac{5173}{3125}$$

$$\frac{32665061773}{30517578125}$$

Université Paris 7 **Premier semestre 2005-2006**
Licence 1ère année **MK1 - Maple**

Feuille d'exercices n°5

Exercice 1. Ecrire une procédure qui prend en entrée une fonction et qui trace sa dérivée en sortie.

Exercice 2. Afficher la séquence des 20 premiers nombres premiers (*ithprime*) :

1) Avec la commande *seq*.
2) Avec une boucle *for*.

Exercice 3. Ecrire une procédure qui, étant donné les coordonnées x, y d'un point du plan, calcule la distance entre les points (x, y) et $(3, 4)\cdot$ Calculer la distance de $(7, -5)$ à $(3, 4)$.

Exercice 4. Ecrire une procédure permettant de savoir si un triangle est équilatéral, étant données les longueurs de chacun des côtés. Même question pour un triangle rectangle.

Exercice 5. Nombres de Mersenne
Le n-ème nombre de Mersenne est $M_n = 2^n - 1$, pour $n \geq 0$. Ces nombres ont la propriété suivante : si M_n est un nombre premier, alors n est un nombre premier. Cependant, la réciproque n'est pas vraie.

1) Calculer le plus petit nombre premier p tel que M_p ne soit pas premier (*isprime*, *nextprime*).
2) Donner M_p et sa factorisation en nombres premiers (*ifactor*).

Exercice 6. Suite de Fibonacci
La suite de Fibonacci est la suite récurrente d'ordre deux définie par :

$$\begin{cases} u_0 = 0 \\ u_1 = 1 \\ u_{n+2} = u_{n+1} + u_n \text{ pour } n \geq 2 \end{cases}$$

1) Définir la suite de Fibonacci.
2) Calculer u_{10}, u_{25}.
3) Calculer u_{50}. Que remarquez-vous ? Pouvez-vous expliquez cela ?

Correction du TP n°5

Exercice 1

Une façon de faire :

```
> restart;
  grapheder:=proc(f)
  RETURN(plot(diff(f(x),x),x));
  end:
```

Attention à bien donner en entrée une fonction ! (et non pas une expression en x). Par exemple, la commande suivante ne fonctionne pas :

```
> grapheder(x^2);
Warning, unable to evaluate the function to numeric values in the region; see
the plotting command's help page to ensure the calling sequence is correct

Plotting error, empty plot
```

Mais celle-ci fonctionne :

```
> grapheder(x->x^2);
```




```
> grapheder(x->cos(x));
```



Exercice 2

ithprime(k) donne le k-ième nombre premier.

Avec la commande *seq* :

```
> restart;
  seq(ithprime(k),k=1..20);
```

$$2, 3, 5, 7, 11, 13, 17, 19, 23, 29, 31, 37, 41, 43, 47, 53, 59, 61, 67, 71$$

Avec une boucle *for* :

```
> s:=NULL:  # on initialise la séquence s en y mettant la
  séquence vide
  for k from 1 to 20 do
    s:=s,ithprime(k);  #à chaque étape, on met ithprime(k) à droit
  de la séquence
  od:
  s;
```

$$2, 3, 5, 7, 11, 13, 17, 19, 23, 29, 31, 37, 41, 43, 47, 53, 59, 61, 67, 71$$

Exercice 3

La distance entre les points (x,y) et (3,4) est donnée par : $\sqrt{(x-3)^2+(y-4)^2}$.

```
> restart;
  distance:=proc(x,y)
  RETURN(sqrt((x-3)^2+(y-4)^2));
  end:
```

```
> distance(7,-5);
```

$$\sqrt{97}$$

Exercice 4

Pour le triangle équilatéral : en entrée, on donne les longueurs a,b, et c des trois cotés.
Le triangle est équilatéral si et seulement si a=b=c.

```
> restart;
  testéquilatéral:=proc(a,b,c)
  if a=b and b=c then RETURN(`Le triangle est équilatéral`)
  else RETURN(`Le triangle n'est pas équilatéral`) fi;
  end:
```

Remarque : pour définir une chaine de caractères, on a utilisé le symbole ` (à distinguer du guillemet simple ')

```
> testéquilatéral(2,2,3);
```

Le triangle n'est pas équilatéral

```
> testéquilatéral(8,8,8);
```

Le triangle est équilatéral

Pour le triangle rectangle : en entrée, on donne encore les longueurs a,b, et c des trois cotés. Une façon de faire : le triangle est équilatéral si et seulement s'il vérifie $a^2+b^2=c^2$ ou bien $a^2+c^2=b^2$ ou bien $b^2+c^2=a^2$ (c'est la réciproque du théorème de Pythagore).

```
> testrectangle:=proc(a,b,c)
  local aa,bb,cc;
  aa:=a^2; bb:=b^2; cc:=c^2;
  if aa+bb=cc or aa+cc=bb or bb+cc=aa then RETURN(`Le triangle
  est rectangle`)
  else RETURN(`Le triangle n'est pas rectangle`) fi;
  end:
```

```
> testrectangle(3,4,5);
```

Le triangle est rectangle

```
> testrectangle(5,3,4);
```

Le triangle est rectangle

```
> testrectangle(1,2,3);
```

Le triangle n'est pas rectangle

Exercice 5: nombres de Mersenne

On commence par définir une fonction M qui à n associe le nombre de Mersenne M(n).

```
> restart;
  M:=n->2^n-1;
```

$$M := n \rightarrow 2^n - 1$$

1) *isprime(k)* dit si le nombre k est un nombre premier (répond *true* ou *false*).

nextprime(k) donne le nombre premier qui suit *k*.

Une façon de calculer le plus petit nombre premier p tel que M(p) ne soit pas premier est la suivante :

- on démarre à p=2 (le premier nombre premier)
- tant que M(p) est premier, on remplace p par le nombre premier qui suit p

La boucle s'arrete la premiere fois que M(p) n'est pas premier. On demande à la fin d'afficher p.

```
> p:=2:
  while isprime(M(p)) do p:=nextprime(p) od:
  p;
```

11

La réponse est p=11.

2) M(11) est le plus petit nombre de Mersenne non premier. Il vaut :

```
> M(11);
```

2047

La commande *ifactor* donne sa factorisation en produit de nombres premiers :

```
> ifactor(M(11));
```

(23) (89)

Exercice 6: suite de Fibonacci

1) On peut définir la suite en utilisant la récursivité. C'est une récurrence d'ordre 2, il faut donc donner deux conditions d'initialisation pour que la suite soit bien définie.

```
> restart;
  u:=proc(n)
  if n=0 then RETURN(0)
  elif n=1 then RETURN(1)
  else RETURN(u(n-1)+u(n-2));
  fi;
  end:
> u(0); u(1); u(2); u(3);
```

0

1

1

2

2)

```
> u(10);
```

55

```
> u(25);
```

75025

3)

```
> # u(50);
```

Le calcul de u(50) est excessivement long...

L'explication est la suivante : quand Maple calcule u(50), il calcule séparément u(49) et u(48). Pour calculer u(49), Maple calcule à nouveau u(48) et u(47)... et ainsi de suite. Comme Maple ne stocke pas les calculs intermédiaires, il passe son temps à recalculer des données qu'il a déjà calculées avant... cela lui prend beaucoup de temps.

Une façon de remédier à cela : modifier la procédure en mettant "option remember" sur la première ligne (remember = souviens-toi !).

```
> restart;
  u:=proc(n) option remember;
  if n=0 then RETURN(0)
  elif n=1 then RETURN(1)
  else RETURN(u(n-1)+u(n-2));
  fi;
  end:
```

Maple crée alors une table interne dans laquelle il stocke les valeurs qu'il a déjà calculées. A chaque appel de la procédure, Maple cherche d'abord si le résultat se trouve dans la table (si elle ne s'y trouve pas, il fait le calcul et le stocke).

```
> u(50);
```

12586269025

Le calcul de u(50) est alors immédiat.

C. Armana
armana@math.jussieu.fr

MK1 "Calcul formel" Maple

TP6 : Développements limités

But du TP6
Aujourd'hui nous allons utiliser Maple pour calculer des développements limités de fonctions et utiliser ces développements pour déterminer des asymptotes. Le terme "développement limité" est une spécialité française. Les anglo-saxons (en particulier Maple) préfèrent parler de développement de Taylor ("Taylor series expansion") ou de développement en série ("series expansion").

Et surtout, n'oubliez pas de vous (et de me) poser des questions !

1. Les développements de Taylor

Les développements de Taylor sont des cas particuliers de développements limités.
Rappel mathématique : Soit f une fonction sur un intervalle I de **R**. Soit $a \in I$. Rappeler la condition sur f pour qu'elle admette un développement de Taylor à l'ordre n et donner ce développement (formule de Taylor-Young).

Maple calcule les développements de Taylor avec la commande *taylor* (qui est un cas particulier de la commande *series*).

```
> restart; ?taylor
> taylor(exp(-x),x,4);
```

$$1-x+\frac{1}{2}x^2-\frac{1}{6}x^3+O(x^4)$$

La commande ci-dessus calcule le développement de Taylor de $\mathbf{e}^{(-x)}$ en $x=0$, avec un "reste" en $O(x^4)$. Attention, il s'agit d'un O, et non pas d'un o.

Rappel mathématique : o et O

$O(x^n)$ est une notation pour une fonction de x nulle pour $x=0$ et telle que $\frac{O(x^n)}{x^n}$ soit bornée au voisinage de 0 (par valeurs différentes).

$o(x^n)$ est une notation pour une fonction de x nulle pour $x=0$ et telle que $\frac{o(x^n)}{x^n}$ tende vers 0 quand x tend vers 0 (par valeurs différentes).
Un $O(x^n)$ est donc un $o(x^{(n-1)})$.

Dans l'exemple ci-dessus, on a donc obtenu un développement à l'ordre 3 au sens de votre cours de mathématiques. Plus généralement, si vous voulez obtenir un développement à l'ordre k (au sens de votre cours de maths), il faut *en général* demander à Maple taylor(f(x), x, $k+1$).

Cependant, attention ! La commande taylor(f(x), x, k) ne rend pas forcément une réponse en $O(x^k)$. En fait, Maple fera ses calculs intermédiaires à l'ordre Maple k mais rien ne garantit que le résultat final sera en $O(x^k)$. Regardez l'exemple :

```
> taylor(tan(x),x,6);
```

$$x+\frac{1}{3}x^3+\frac{2}{15}x^5+O(x^6)$$

```
> taylor(tan(x)/x,x,6);
```

$$1+\frac{1}{3}x^2+\frac{2}{15}x^4+O(x^5)$$

On peut demander des développements ailleurs qu'en 0 :

```
> taylor(exp(-x),x=1,6);
```

$$\mathbf{e}^{(-1)}-\mathbf{e}^{(-1)}(x-1)+\frac{1}{2}\mathbf{e}^{(-1)}(x-1)^2-\frac{1}{6}\mathbf{e}^{(-1)}(x-1)^3+\frac{1}{24}\mathbf{e}^{(-1)}(x-1)^4-\frac{1}{120}\mathbf{e}^{(-1)}(x-1)^5+O((x-1)^6)$$

Si on souhaite réutiliser la "partie principale" (= partie polynomiale) fournie par le développement, on procède avec la commande *convert(,polynom)* :

```
> dev:=taylor(exp(-x),x,4);
```

$$dev := 1-x+\frac{1}{2}x^2-\frac{1}{6}x^3+O(x^4)$$

```
> P:=convert(dev,polynom);
```

$$P := 1-x+\frac{1}{2}x^2-\frac{1}{6}x^3$$

On peut maintenant tracer $\mathbf{e}^{(-x)}$ et le polynome P qui l'approche, au voisinage de 0 :

```
> plot([exp(-x),P],x=-5..5,y=-5..5,color=[red,blue]);
```



2. Les développements limités

Pour calculer des développements plus généraux que les développements de Taylor (développements généralisés, asymptotiques), on dispose de la commande *series*. Sa syntaxe est identique à celle de *taylor*.

Tout développement limité n'est pas forcément un développement de Taylor. Prenons la fonction définie par $f(x) = x^4 \cos\left(\frac{1}{x^6}\right)$ pour $x \neq 0$ et $f(0) = 0$. Elle a un développement limité en 0 à l'ordre 3 qui est : $f(x) = o(x^3)$ (pourquoi ?). Pourtant, f n'est pas dérivable deux fois en 0, comme nous allons le voir.

```
> f:=x->x^4*cos(1/x^6);
  f(0):=0;
```

$$f := x \to x^4 \cos\left(\frac{1}{x^6}\right)$$

$$f(0) := 0$$

Sa dérivée g, pour $x \neq 0$, est :

```
> g:=D(f): g(x);
```

$$4\,x^3 \cos\left(\frac{1}{x^6}\right) + \frac{6 \sin\left(\frac{1}{x^6}\right)}{x^3}$$

Calculons g(0).

```
> limit((f(x)-f(0))/(x-0),x=0);
```

$$0$$

Donc g(0)=0. La fonction g n'est pas continue en 0 car :

```
> limit(g(x),x=0);
```

$$\textit{undefined}$$

Donc g n'est pas dérivable en 0. Donc f n'est pas dérivable deux fois en 0.

Essayons d'obtenir le développement limité de f par Maple (on lui demande à l'ordre 4 pour obtenir un ordre "mathématique" 3).

```
> taylor(f(x),x,4);
Error, (in series/trig) unable to compute series

> series(f(x),x,4);
Error, (in series/trig) unable to compute series
```

Maple ne peut pas calculer ce développement. Rassurez-vous, en pratique, les fonctions utilisées sont très souvent suffisamment dérivables et vous pourrez utiliser Maple.

Développement limité généralisé

Dans un développement limité, la partie principale (avant le o ou le O) est polynomiale. On peut définir des développements plus généraux. Dans un développement limité généralisé, on autorise des puissances négatives de x dans la partie principale. Un développement limité généralisé (en $x = 0$) est donc de la forme : $f(x) = \frac{1}{x^p}\,(a_0 + a_1\,x + ... + a_n\,x^n + o(x^n))$ où n et p sont des entiers positifs.

On peut calculer de tels développements dans Maple à l'aide de *series* (parfois *taylor* fonctionne également). Par exemple, pour obtenir le développement généralisé de $\frac{1}{\tan(x)}$ à l'ordre (mathématique) 3 en 0 :

```
> ?series
> series(1/tan(x),x=0,4);
```

$$x^{-1} - \frac{1}{3}\,x - \frac{1}{45}\,x^3 + O(x^4)$$

3. Application : détermination d'asymptotes

Soit f une fonction et C sa courbe représentative dans le plan. On suppose que f est définie au voisinage de $+\infty$ ou $-\infty$. Les développements limités généralisés permettent parfois de déterminer des asymptotes à l'infini de C.

Asymptote horizontale ou oblique

Expliquez le résultat suivant : si on a le développement limité généralisé en $+\infty$ ou $-\infty$ $f(x) = ax + b + \frac{a_p}{x^p} + o\left(\frac{1}{x^p}\right)$ avec $a_p \neq 0$ et p entier naturel non nul, alors la droite d'équation $y = a\,x + b$ est asymptote à C en $+\infty$ ou $-\infty$.

De plus, le signe de $\frac{a_p}{x^p}$ permet de déterminer la position de C par rapport à la droite. Le cas particulier $a = 0$ correspond à une asymptote horizontale.

Voici un exemple :

```
> f:=x->x*arctan(x/(x-1));
```

$$f := x \to x \arctan\left(\frac{x}{x-1}\right)$$

```
> taylor(f(x),x=infinity,3);
  taylor(f(x),x=-infinity,3);
```

$$\frac{\pi\,x}{4} + \frac{1}{2} + \frac{1}{4\,x} + O\left(\frac{1}{x^2}\right)$$

$$\frac{\pi\,x}{4} + \frac{1}{2} + \frac{1}{4\,x} + O\left(\frac{1}{x^2}\right)$$

Donc la droite :

```
> d:=Pi*x/4+1/2;
```

$$d := \frac{\pi\,x}{4} + \frac{1}{2}$$

est asymptote à C en $+\infty$ et en $-\infty$. De plus, C est au-dessus de la droite au voisinage de $+\infty$ et en-dessous de la droite au voisinage de $-\infty$. On le vérifie sur un graphique :

```
> plot([f(x),d],x=-5..5);
```



Courbe asymptote

Si on a le développement limité (généralisé ou asymptotique) en $+\infty$ ou $-\infty$

$\mathrm{f}(x) = \mathrm{g}(x) + \dfrac{a_p}{x^p} + \mathrm{o}\left(\dfrac{1}{x^p}\right)$ avec $a_p \neq 0$ et p entier naturel non nul, alors la courbe représentative *C*' de *g* est asymptote à *C* en $+\infty$ ou $-\infty$. La position de *C* par rapport à *C*' est déterminée par le signe de $\dfrac{a_p}{x^p}$.

Si *g* est un polynôme de degré 1, on a une asymptote oblique (droite) ; si *g* est un polynôme de degré 2, on a une asymptote parabolique.

```
> f:=x->x^3*sin(1/x);
```

$$f := x \rightarrow x^3 \sin\left(\frac{1}{x}\right)$$

```
> series(f(x),x=infinity);
```

$$x^2 - \frac{1}{6} + \frac{1}{120\, x^2} + \mathrm{O}\left(\frac{1}{x^3}\right)$$

Donc la parabole d'équation $y = x^2 - \dfrac{1}{6}$ est asymptote à la courbe *C* de *f* en $+\infty$.

```
> plot([f(x),x^2-1/6],x=0..1,y=-1..1);
```



Autre exemple (développement asymptotique) :

```
> g:=x->ln(x^2+1)-1/x;
```

$$g := x \rightarrow \ln(x^2 + 1) - \frac{1}{x}$$

```
> series(g(x),x=infinity);
```

$$2\ln(x) - \frac{1}{x} + \frac{1}{x^2} - \frac{1}{2\, x^4} + \mathrm{O}\left(\frac{1}{x^6}\right)$$

La courbe d'équation $y = 2\ln(x)$ est asymptote à *C* en $+\infty$ et *C* est située sous son asymptote au voisinage de $+\infty$.

```
> plot([g(x),2*ln(x)],x=0..15,y=-5..10,color=[red,blue]);
```



Université Paris 7 | **Premier semestre 2005-2006**
Licence 1ère année | **MK1 - Maple**

Feuille d'exercices n°6

Exercice 1. Calculer les développements limités suivants :

1) $(\ln(1+x))^2$ en 0 à l'ordre mathématique 6
2) $\sqrt{x(\sin(x)+\sinh(x)-2x)}$ en 0 à l'ordre mathématique 9
3) $\dfrac{\sin(x)}{1+x}$ en $\pi/2$ à l'ordre mathématique 3
4) $1+2x+3x^5+x^6$ en 0 à l'ordre mathématique 5 ; même question à l'ordre mathématique 6. Pouvez-vous expliquer ces résultats ?
5) $(1+x)^a$ en 0 à l'ordre mathématique 5 pour $a \in \mathbb{R}$

Exercice 2. Déterminer un équivalent en 0 de $g(x) = \sin(\ln(1+x)) - \ln(1+\sin(x))$.

Exercice 3. Sur un même graphique, tracer la courbe représentative de $x \mapsto \sin(x)$ et les parties principales polynomiales des développements de Taylor de $\sin(x)$ en 0 à l'ordre Maple k, pour k pair variant de 2 à 12 (utiliser une boucle *for*)· Recadrer le graphique pour qu'il soit joli.

Exercice 4. Pour chacune des fonctions suivantes, en utilisant les développements limités, déterminer les asymptotes en $+\infty$ et $-\infty$ de la courbe représentative $\mathcal{C}$ de f. Préciser les positions relatives de $\mathcal{C}$ et de ses asymptotes. Vérifier vos réponses en traçant f et ses asymptotes sur un même graphique et en choisissant des intervalles d'affichage appropriés.

1) $x \mapsto e^{1/x}\sqrt{x^2+2x}$
2) $x \mapsto \dfrac{(x^2-x-6)^2-(x+3)}{x^2-4x+3}$
3) $x \mapsto \sqrt{1+x+x^2+x^4}$

Correction du TP n°6

Exercice 1

Attention, on demande des ordres au sens de votre cours de mathématiques. Il faut donc faire attention aux ordres que vous demandez à Maple et penser à la règle "Un $O(x^n)$ est un $o(x^{(n-1)})$".

Remarque : $x^n\,\varepsilon(x)$ (avec $\varepsilon(x)$ -> 0) est un $o(x^n)$.

```
> restart;
```

1)

```
> taylor((ln(1+x))^2,x,7);
```

$$x^2-x^3+\frac{11}{12}x^4-\frac{5}{6}x^5+\frac{137}{180}x^6+O(x^7)$$

Donc $\ln(1+x)^2=x^2-x^3+\frac{11\,x^4}{12}-\frac{5\,x^5}{6}+\frac{137\,x^6}{180}+o(x^6)$.

2)

```
> taylor(sqrt(x*(sin(x)+sinh(x)-2*x)),x,10);
```

$$\frac{\sqrt{15}}{30}x^3+O(x^7)$$

Demander un calcul à l'ordre Maple 10 ne suffit pas pour obtenir un $o(x^9)$, donc on pousse plus loin :

```
> taylor(sqrt(x*(sin(x)+sinh(x)-2*x)),x,11);
```

$$\frac{\sqrt{15}}{30}x^3+\frac{\sqrt{15}}{181440}x^7+O(x^{11})$$

Cela nous donne le résultat, puisque qu'un $O(x^{11})$ est un $o(x^9)$.

3)

```
> taylor(sin(x)/(1+x),x=Pi/2,4);
```

$$\frac{1}{1+\frac{\pi}{2}}-\frac{2}{(2+\pi)\left(1+\frac{\pi}{2}\right)}\left(x-\frac{\pi}{2}\right)+\frac{-\frac{1}{2}+\frac{4}{(2+\pi)^2}}{1+\frac{\pi}{2}}\left(x-\frac{\pi}{2}\right)^2+\frac{-4+4\,\pi+\pi^2}{(2+\pi)^3\left(1+\frac{\pi}{2}\right)}\left(x-\frac{\pi}{2}\right)^3+O\left(\left(x-\frac{\pi}{2}\right)^4\right)$$

4)

```
> taylor(1+2*x+3*x^5+x^6,x,6);
```

$$1+2\,x+3\,x^5+O(x^6)$$

Cela nous dit que $x^6=o(x^5)$.

```
> taylor(1+2*x+3*x^5+x^6,x,7);
```

$$1+2\,x+3\,x^5+x^6$$

On constate que le O a disparu.C'est normal : un polynôme a un développement limité exact (sans reste en o ni O), pourvu qu'on le calcule à un ordre suffisamment grand. Si on pousse plus loin, on obtient le même résultat :

```
> taylor(1+2*x+3*x^5+x^6,x,10);
```

$$1+2\,x+3\,x^5+x^6$$

5) Ici, pas la peine de fixer une valeur pour a.

```
> taylor((1+x)^a,x,6);
```

$$1+a\,x+\frac{a\,(a-1)}{2}x^2+\frac{a\,(a-1)\,(a-2)}{6}x^3+\frac{a\,(a-1)\,(a-2)\,(a-3)}{24}x^4+\frac{a\,(a-1)\,(a-2)\,(a-3)\,(a-4)}{120}x^5+O(x^6)$$

Exercice 2

Rappel mathématique : On dit que f est équivalente à g en 0 si $g-f=o(f)$.

Calculons un développement limité de g en 0.

```
> restart;
  g:=x->sin(ln(1+x))-ln(1+sin(x));
```

$$g:=x\to\sin(\ln(1+x))-\ln(1+\sin(x))$$

```
> taylor(g(x),x,5);
```

$$\frac{1}{12}x^4+O(x^5)$$

On a donc $g(x)=\frac{x^4}{12}+o(x^4)$.

```
> taylor(g(x)-x^4/12,x,5);
```

$$O(x^5)$$

Donc si on pose $f(x)=\frac{x^4}{12}$, on a $g(x)-f(x)=O(x^5)=o(x^4)=o\left(\frac{x^4}{12}\right)=o(f(x))$. Donc $g(x)$ est équivalent en 0 à $\frac{x^4}{12}$.

Exercice 3

```
> restart;
```

A l'aide de *seq*, on crée la séquence des parties principales des développements de Taylor d'ordre $2k$ pour k variant de 1 à 6.

```
> S:=seq(convert(taylor(sin(x),x,2*k),polynom),k=1..6);
```

$$S:=x,x-\frac{1}{6}x^3,x-\frac{1}{6}x^3+\frac{1}{120}x^5,x-\frac{1}{6}x^3+\frac{1}{120}x^5-\frac{1}{5040}x^7,$$
$$x-\frac{1}{6}x^3+\frac{1}{120}x^5-\frac{1}{5040}x^7+\frac{1}{362880}x^9,$$
$$x-\frac{1}{6}x^3+\frac{1}{120}x^5-\frac{1}{5040}x^7+\frac{1}{362880}x^9-\frac{1}{39916800}x^{11}$$

On les fait afficher sur un même graphique, avec $\sin(x)$, et on recadre en ordonnées pour mieux y voir :

```
> plot([sin(x),S],x,y=-2..2,numpoints=1000);
```



Remarquez que les parties principales approchent de mieux en mieux la fonction autour de 0, au fur et à mesure que l'ordre augmente (elles "collent" à la fonction sur un intervalle de plus en plus grand autour de 0). Remarquez aussi que la partie principale du DL à l'ordre mathématique 1 fournit la tangente à la courbe en $x=0$ (son équation est $y=x$).

Bonus : une version animée !

```
> with(plots):
```

Avec *seq*, on construit la séquence des graphiques représentants le sinus et la partie principale à l'ordre *k*. Les *plot* sont des objets comme les autres et on peut les stocker dans une séquence ! Attention cependant à bien mettre deux-points (:) à la fin de la commande (essayez avec point-virgule...)

```
> S:=seq(
  plot([sin(x),convert(taylor(sin(x),x,2*k),polynom)],x,y=-5..5)
  ,k=1..10 ):
```

Ensuite, on utilise *display* (librairie *plots*), qui permet d'afficher plusieurs graphiques en même temps. L'option *insequence* = *true* permet de les afficher au fur et à mesure, et donc de faire une animation.

```
> ?display
> display(S,insequence=true);
```

Essayez la commande ci-dessus. Pour démarrer l'animation, cliquez sur le graphique puis sur l'icône de lecture (flèche) qui apparaît en haut. La vitesse peut être réglée par le menu spécial qui apparaît en haut de l'écran.

Exercice 4

1)

```
> restart;
  f:=x->exp(1/x)*sqrt(x^2+2*x);
```

$$f := x \to \mathbf{e}^{\left(\frac{1}{x}\right)}\sqrt{x^2+2\,x}$$

```
> taylor(f(x),x=infinity,3);
  taylor(f(x),x=-infinity,3);
```

$$x+2+\frac{1}{x}+\mathrm{O}\left(\frac{1}{x^2}\right)$$

$$-x-2-\frac{1}{x}+\mathrm{O}\left(\frac{1}{x^2}\right)$$

Donc la droite d'équation $y=x+2$ est asymptote à la courbe C en $+\infty$. Au voisinage de $+\infty$, la courbe est située au-dessus de la droite. La droite d'équation $y=-x-2$ est asymptote à la courbe C en $-\infty$. Au voisinage de $-\infty$, la courbe est située au-dessus de la droite.

```
> plot([f(x),x+2,-x-2],x,y=-10..10,color=[red,blue,green]);
```



2)

```
> g:=x->((x^2-x-6)^2-(x+3))/(x^2-4*x+3);
```

$$g := x \to \frac{(x^2-x-6)^2-x-3}{x^2-4\,x+3}$$

```
> taylor(g(x),x=infinity,2);
```

$$x^2+2\,x-6-\frac{19}{x}+\mathrm{O}\left(\frac{1}{x^2}\right)$$

```
> taylor(g(x),x=-infinity,2);
```

$$x^2+2\,x-6-\frac{19}{x}+\mathrm{O}\left(\frac{1}{x^2}\right)$$

Donc la parabole d'équation $y=x^2+2\,x-6$ est asymptote à C en $+\infty$ et $-\infty$. De plus, au voisinage de $+\infty$, la courbe est au-dessous de la parabole ; au voisinage de $-\infty$, la courbe est au-dessus de la parabole.

```
> plot([g(x),x^2+2*x-6],x,y=-20..50,discont=true);
```



3)

```
> h:=x->sqrt(1+x+x^2+x^4);
```

$$h := x \to \sqrt{1+x+x^2+x^4}$$

```
> taylor(h(x),x=infinity,2);
  taylor(h(x),x=-infinity,2);
```

$$x^2+\frac{1}{2}+\frac{1}{2\,x}+\mathrm{O}\left(\frac{1}{x^2}\right)$$

$$x^2+\frac{1}{2}+\frac{1}{2\,x}+\mathrm{O}\left(\frac{1}{x^2}\right)$$

Donc la parabole d'équation $y=x^2+\dfrac{1}{2}$ est asymptote à la courbe C en $+\infty$ et $-\infty$. Au voisinage de $+\infty$, la courbe est située au-dessus de la parabole. Au voisinage de $-\infty$, la courbe est située en-dessous de la parabole.

```
> plot([h(x),x^2+1/2],x=-3..0,y=0..10,color=[red,blue]);
```




```
> plot([h(x),x^2+1/2],x=-0..3,y=0..10,color=[red,blue]);
```



C. Armana
armana@math.jussieu.fr

MK1 "Calcul formel" Maple

TP7 : Courbes paramétrées, courbes polaires

But du TP7
Nous avons déjà vu comment tracer des courbes représentatives de fonctions dans Maple.
Aujourd'hui, nous allons explorer les possibilités de Maple de tracer d'autres types de courbes : courbes paramétrées, courbes polaires.

Et surtout, n'oubliez pas de vous (et de me) poser des questions !

1. Les courbes paramétrées

Pour tracer une courbe paramétrée donnée par $M(t) = (x(t), y(t))$, on utilise (encore !) la commande *plot*. Par exemple, pour $x(t) = \cos(t)$ et $y(t) = \sin(t)$:

```
> restart;
> ?plot[parametric]
> plot( [sin(t),cos(t),t=0..2*Pi] );
```



Notez bien la position des crochets dans la syntaxe !

Exemple de plan d'étude d'une courbe paramétrée :

1) Intervalle de définition en t
Périodicité éventuelle
Symétries éventuelles de la courbe
-> déterminer un intervalle d'étude minimal
2) Limites de x et y aux bornes des intervalles d'étude
3) Etude des branches infinies : recherche d'asymptotes, de branches paraboliques
4) Etude des variations de x et y et des points singuliers (x'(t)=y'(t)=0)
5) Tracé
6) Détermination des points doubles éventuels ($M(u) = M(v)$ avec $u \neq v$).

2. Les courbes polaires

Une courbe polaire est donnée par une équation de la forme $\rho = r(\theta)$. Elle s'obtient en plaçant dan le plan les points de coordonnées ($r(\theta)\cos(\theta), r(\theta)\sin(\theta)$). Une courbe polaire est donc une courbe paramétrée. Pour tracer une courbe polaire dans Maple, on utilise la commande *plot* avec l'option $coords = polar$.

```
> ?plot[polar]
```

$\rho = \sin(\theta)\cos(\theta)$:

```
> plot([sin(t)*cos(t),t,t=0..2*Pi],coords=polar);
```



Pouvez-vous devinez l'allure des courbes polaires suivantes avant de les tracer ?

$\rho = 1$:

```
> plot([1,t,t=0..2*Pi],coords=polar);
```

$\rho = \theta$:

```
> plot([t,t,t=0..10],coords=polar);
```

$\rho = \cos(\theta)$:

```
> plot([cos(t),t,t=0..Pi],coords=polar);
```

La feuille d'érable de Maple : tracer la courbe polaire suivante: $\rho = \dfrac{2 - \sin(7\theta) - \cos\left(\dfrac{30\theta}{2}\right)}{100 + \left(\theta - \dfrac{\pi}{2}\right)^8}$

pour θ dans $\left[-\dfrac{\pi}{2}, \dfrac{3\pi}{2}\right]$ et en fixant *numpoints* à500.

Feuille d'exercices n°7

Exercice 1. Etudier les courbes paramétrées suivantes : déterminer à la main l'intervalle d'étude et les symétries ; étudier les limites, branches infinies, variations et points doubles avec Maple (si possible). Vérifier vos résultats en traçant la courbe et ses asymptotes sur un même graphique (charger la librairie *plots* et utiliser sa commande *display*).

1. $\begin{cases} x(t) = \sin 2t \\ y(t) = \cos 2t \end{cases}$
2. $\begin{cases} x(t) = \cos t \\ y(t) = (1 + \cos t) \sin t \end{cases}$
3. $\begin{cases} x(t) = t/\ln t \\ y(t) = t^2/(t-1) \end{cases}$
4. $\begin{cases} x(t) = \cos(t) \\ y(t) = \cos(3t/5) \end{cases}$
5. $\begin{cases} x(t) = t^2 + 2/t \\ y(t) = t + 1/t \end{cases}$
6. $\begin{cases} x(t) = 2t + t^2 \\ y(t) = 2t - 1/t^2 \end{cases}$

Exercice 2. Tracer la courbe paramétrée définie par $x(t) = t^3 - 4t$, $y(t) = 2t^2 - 3$ et calculer l'angle formé par les tangentes au point double.

Exercice 3. Tracer la courbe paramétrée $\mathcal{C}$ définie par $x(t) = t^2$, $y(t) = t^3$. Déterminer le lieu des points du plan d'où l'on peut mener (au moins) deux tangentes à $\mathcal{C}$ orthogonales et le tracer. On appelle ce lieu la courbe orthoptique de $\mathcal{C}$.

Exercice 4. On considère l'astroïde définie par $x(t) = a\cos^3 t$, $y(t) = a\sin^3 t$. Tracer la courbe sur sa période minimale pour $a = 1$. Animer la courbe en faisant varier a entre dans un intervalle au choix (librairie *plots*, commande *animate*).

Exercice 5. Etude de la courbe polaire définie par $\rho = \sin(3\theta/2)$.

1) Calculer à la main la période de la courbe et la tracer sur cette période.
2) Déterminer à la main les symétries de la courbe. Qu'en déduisez-vous sur l'intervalle d'étude ? Tracer la courbe sur l'intervalle minimal.

Correction du TP n°7

Exercice 1

1) La période est π donc on prend comme intervalle d'étude $[0, \pi]$.

$x(-t)=-x(t)$ et $y(-t)=y(t)$ donc symétrie par rapport à Oy. On étudie sur $\left[0, \frac{\pi}{2}\right]$ et on complète la courbe par symétrie.

$x\left(t+\frac{\pi}{2}\right)=-x(t)$ et $y\left(t+\frac{\pi}{2}\right)=-y(t)$ donc symétrie centrale par rapport à O. On étudie sur $\left[0, \frac{\pi}{4}\right]$ et on complète par symétrie.

```
> restart;
  x:=t->sin(2*t):
  y:=t->cos(2*t):
> plot([x(t),y(t),t=0..Pi/4]);
```




```
> plot([x(t),y(t),t=0..Pi]);
```



La courbe obtenue est le cercle de rayon 1. C'est la même que dans le premier exemple du cours, sauf qu'elle est parcourue deux fois plus rapidement.

2)

```
> restart;
  x:=t->cos(t):
  y:=t->(1+cos(t))*sin(t):
```

La période est 2π. L'intervalle d'étude est $[0, 2\pi]$.

$x(-t)=x(t)$ et $y(-t)=-y(t)$ donc la courbe est symétrique par rapport à Ox. On étudie sur $[0, \pi]$ et on complète par symétrie.

```
> xx:=D(x);yy:=D(y);
```

$$xx := t \to -\sin(t)$$

$$yy := t \to -\sin(t)^2 + (1+\cos(t))\cos(t)$$

```
> solve(xx(t)=0,t);
  solve(yy(t)=0,t);
```

$$0$$

$$\pi, \frac{\pi}{3}, -\frac{\pi}{3}$$

(Maple ne donne pas toutes les solutions)

```
> solve(xx(t)>=0,t);
  solve(yy(t)>=0,t);
```

Apparemment, Maple n'est pas à l'aise avec les inéquations faisant intervenir des sin et cos. On étudie alors le signe à la main.

```
> x(0);y(0); # point à tangente verticale
```

$$1$$

$$0$$

```
> xx(Pi);yy(Pi);
```

$$0$$

$$0$$

```
> x(Pi);y(Pi); # donc ce point est singulier (stationnaire)
```

$$-1$$

$$0$$

```
> plot([x(t),y(t),t=0..Pi]);
```




```
> plot([x(t),y(t),t=0..2*Pi]);
```



3)

```
> restart;
> x:=t->t/ln(t):
  y:=t->t^2/(t-1):
```

Domaine de définition pour t : $]0,1[\cup]1,+\infty[$. Pas de périodicité ni de symétrie.

Limites :

```
> limit(x(t),t=0);
  limit(y(t),t=0);
```

$$0$$

$$0$$

```
> limit(x(t),t=1);
  limit(y(t),t=1);
```

$$\mathit{undefined}$$
$$\mathit{undefined}$$

```
> limit(x(t),t=1,left);limit(x(t),t=1,right);
  limit(y(t),t=1,left);limit(y(t),t=1,right);
```

$$-\infty$$
$$\infty$$
$$-\infty$$
$$\infty$$

```
> limit(x(t),t=infinity);
  limit(y(t),t=infinity);
```

$$\infty$$
$$\infty$$

Variations :

```
> xx:=D(x);yy:=D(y);
```

$$xx := t \to \frac{1}{\ln(t)} - \frac{1}{\ln(t)^2}$$
$$yy := t \to \frac{2\,t}{t-1} - \frac{t^2}{(t-1)^2}$$

```
> solve(xx(t)=0,t);
  solve(yy(t)=0,t);
```

$$\mathbf{e}$$
$$0,\ 2$$

```
> solve(xx(t)>0,t);
  solve(yy(t)>0,t);
```

$$\mathrm{RealRange}(\mathrm{Open}(\mathbf{e}),\ \infty)$$
$$\mathrm{RealRange}(-\infty,\ \mathrm{Open}(0)),\ \mathrm{RealRange}(\mathrm{Open}(2),\ \infty)$$

Branche infinie en 1

```
> limit(y(t)/x(t),t=1);
```

$$1$$

Donc on étudie la limite de y(t)-x(t) :

```
> limit(y(t)-x(t),t=1);
```

$$\frac{1}{2}$$

Donc la droite d'équation $y = x + \frac{1}{2}$ est asymptote à la courbe quand t -> 1.

Remarque : on pouvait aussi l'obtenir par des développements généralisés :

```
> series(x(t),t=1,3);
  series(y(t),t=1,3);
```

$$(t-1)^{-1} + \frac{3}{2} + \frac{5}{12}(t-1) - \frac{1}{24}(t-1)^2 + \mathrm{O}((t-1)^3)$$
$$(t-1)^{-1} + 2 + t - 1$$

Donc

```
> series(x(t)-y(t)+1/2,t=1,3);
```

$$-\frac{7}{12}(t-1) - \frac{1}{24}(t-1)^2 + \mathrm{O}((t-1)^3)$$

ce qui montre que la limite de x(t)-y(t)+1/2 quand t->1 est nulle.

Branche infinie en $+\infty$:

```
> limit(y(t)/x(t),t=infinity);
```

$$\infty$$

Donc la courbe admet une branche parabolique de direction asymptotique Oy quand t tend vers ∞.

Tracé de la courbe et de son asymptote :

```
> with(plots):
> ?plots[display]
> A:=plot([x(t),y(t),t=-10..10],-7..7,-10..10): #attention à bien
  mettre : en fin de ligne
  B:=plot(x+1/2,x=-7..7,color=blue): # même remarque
  display([B,A]);
```



4) x a pour période $2\,\pi$ et y a pour période $\frac{10\,\pi}{3}$. Donc (x, y) a pour période $10\,\pi$.

$\mathrm{x}(-t) = \mathrm{x}(t)$ et $\mathrm{y}(-t) = \mathrm{y}(t)$ donc sur une période, la courbe est décrite <u>deux fois</u>. On se ramène à l'intervalle d'étude $[0, 5\,\pi]$.

$\mathrm{x}(5\,\pi - t) = -\mathrm{x}(t)$ et $\mathrm{y}(5\,\pi - t) = -\mathrm{y}(t)$ donc symétrie centrale par rapport à O. On étudie sur $\left[0, \frac{5\,\pi}{2}\right]$ et on complète par symétrie.

```
> restart;
  x:=t->cos(t):
  y:=t->cos(3*t/5):
```

Variations :

```
> xx:=D(x); yy:=D(y);
```

$$xx := t \to -\sin(t)$$
$$yy := t \to -\frac{3}{5}\sin\left(\frac{3}{5}\,t\right)$$

```
> solve(xx(t)<0,t);
> solve(xx(t)=0,t);
  solve(yy(t)=0,t);
```

$$0$$
$$0$$

Apparemment, Maple n'est pas à l'aise avec les inéquations et équations faisant intervenir des sin et cos. On étudie alors le signe à la main.

```
> x(0);y(0);
```

$$1$$
$$1$$

```
> x(Pi);y(Pi);evalf(%);
```

$$-1$$
$$-\cos\left(\frac{2\pi}{5}\right)$$
$$-0.3090169938$$

```
> x(2*Pi);y(2*Pi);evalf(%);
```

$$1$$
$$-\cos\left(\frac{\pi}{5}\right)$$
$$-0.8090169943$$

```
> x(5*Pi/2);y(5*Pi/2);
```

$$0$$
$$0$$

```
> plot([x(t),y(t),t=0..5*Pi/2]);
```




```
> plot([x(t),y(t),t=0..5*Pi]);
```



Points doubles

```
> S:=solve({x(u)=x(v),y(u)=y(v)},{u,v});
```

$$S := \{u=u, v=u\}, \{u = 5\arccos\left(\frac{1}{2}\operatorname{RootOf}(\_Z^4+\_Z^3-4\_Z^2-4\_Z+1, label=\_L1)\right), v = 5\arccos\left(-\frac{1}{2}+\frac{1}{2}\operatorname{RootOf}(\_Z^4+\_Z^3-4\_Z^2-4\_Z+1, label=\_L1)^3 - 2\operatorname{RootOf}(\_Z^4+\_Z^3-4\_Z^2-4\_Z+1, label=\_L1)\right)\}, \{$$

$$u = 5\arccos\left(\frac{1}{2}\operatorname{RootOf}(\_Z^4-\_Z^3-4\_Z^2+4\_Z+1, label=\_L2)\right), v = 5\arccos\left(\frac{1}{2}+\frac{1}{2}\operatorname{RootOf}(\_Z^4-\_Z^3-4\_Z^2+4\_Z+1, label=\_L2)^3 - 2\operatorname{RootOf}(\_Z^4-\_Z^3-4\_Z^2+4\_Z+1, label=\_L2)\right)\}$$

On élimine la première solution, qui correspond à $u = v$.

```
> T:=S[2..nops([S])];
```

$$T := \{u = 5\arccos\left(\frac{1}{2}\operatorname{RootOf}(\_Z^4+\_Z^3-4\_Z^2-4\_Z+1, label=\_L1)\right), v = 5\arccos\left(-\frac{1}{2}+\frac{1}{2}\operatorname{RootOf}(\_Z^4+\_Z^3-4\_Z^2-4\_Z+1, label=\_L1)^3 - 2\operatorname{RootOf}(\_Z^4+\_Z^3-4\_Z^2-4\_Z+1, label=\_L1)\right)\}, \{$$

$$u = 5\arccos\left(\frac{1}{2}\operatorname{RootOf}(\_Z^4-\_Z^3-4\_Z^2+4\_Z+1, label=\_L2)\right), v = 5\arccos\left(\frac{1}{2}+\frac{1}{2}\operatorname{RootOf}(\_Z^4-\_Z^3-4\_Z^2+4\_Z+1, label=\_L2)^3 - 2\operatorname{RootOf}(\_Z^4-\_Z^3-4\_Z^2+4\_Z+1, label=\_L2)\right)\}$$

Coordonnées (en valeurs approchées) des points doubles :

```
> for i from 1 to nops([T]) do
  evalf( subs(evalf(T[i]),[x(u),y(u)]) );
  od;
```

$$[0.4999999986, -0.3090169934]$$
$$[-0.5000000007, -0.8090169939]$$

Maple en donne deux. On trouve les deux autres par symétrie centrale.
Remarque : un calcul à la main vous donne les coordonnées exactes de ces points doubles.

5)

```
> restart;
> x:=t->t^2+2/t:
  y:=t->t+1/t:
```

Domaine d'étude : $]-\infty,0[\cup]0,+\infty[$.

Limites :

```
> limit(x(t),t=-infinity);
  limit(y(t),t=-infinity);
```

$$\infty$$
$$-\infty$$

```
> limit(x(t),t=infinity);
  limit(y(t),t=infinity);
```

$$\infty$$
$$\infty$$

```
> limit(x(t),t=0);
  limit(y(t),t=0);
```

$$undefined$$
$$undefined$$

```
> limit(x(t),t=0,left);limit(x(t),t=0,right);
  limit(y(t),t=0,left);limit(x(t),t=0,right);
```

$$-\infty$$
$$\infty$$
$$-\infty$$
$$\infty$$

Branches infinies en $-\infty$ et $+\infty$

```
> limit(y(t)/x(t),t=-infinity);
  limit(y(t)/x(t),t=infinity);
```

$$0$$
$$0$$

Branche parabolique de direction asymptotique Ox quand t->$-\infty$ et t->$+\infty$. On peut obtenir un résultat plus précis en faisant des développements généralisés :

```
> taylor(x(t),t=-infinity);
  taylor(y(t),t=-infinity);
```

$$t^2+\frac{2}{t}$$
$$t+\frac{1}{t}$$

```
> taylor(x(t)-y(t)^2+2,t=-infinity);
  taylor(x(t)-y(t)^2+2,t=infinity);
```

$$\frac{2}{t}-\frac{1}{t^2}$$
$$\frac{2}{t}-\frac{1}{t^2}$$

Ceci montre que la parabole d'équation $x=y^2-2$ est asymptote à la courbe quand t-> $+\infty$ et t->$-\infty$.

Branche infinie en 0

```
> limit(y(t)/x(t),t=0);
```

$$\frac{1}{2}$$

```
> limit(y(t)-1/2*x(t),t=0);
```

$$0$$

Donc la droite d'équation $y=\frac{x}{2}$ est asymptote à la courbe quand t->1.

Variations

```
> xx:=D(x);yy:=D(y);
```

$$xx := t \to 2\,t-\frac{2}{t^2}$$
$$yy := t \to 1-\frac{1}{t^2}$$

```
> solve(xx(t)>0,t);
  solve(yy(t)>0,t);
```

$$\mathrm{RealRange}(\mathrm{Open}(1), \infty)$$
$$\mathrm{RealRange}(-\infty, \mathrm{Open}(-1)), \mathrm{RealRange}(\mathrm{Open}(1), \infty)$$

```
> solve(xx(t)=0,t);
  solve(yy(t)=0,t);
```

$$1, -\frac{1}{2}+\frac{1}{2}I\sqrt{3}, -\frac{1}{2}-\frac{1}{2}I\sqrt{3}$$
$$1, -1$$

```
> x(1);y(1);
```

$$3$$
$$2$$

Point singulier : (3,2)

```
> taylor(x(t)-x(1),t=1);
  taylor(y(t)-y(1),t=1);
```

$$3\,(t-1)^2-2\,(t-1)^3+2\,(t-1)^4-2\,(t-1)^5+\mathrm{O}((t-1)^6)$$
$$(t-1)^2-(t-1)^3+(t-1)^4-(t-1)^5+\mathrm{O}((t-1)^6)$$

Donc c'est un point de rebroussement (de deuxième espèce).

Tracé : pour tracer la parabole définie par $x=y^2-2$, on utilise la commande *implicitplot* de la librairie *plots*, en fixant *numpoints* à 3000 pour avoir un dessin "lisse".

```
> with(plots):
  A:=implicitplot(x=y^2-2,x=-10..10,y=-10..10,color=blue,numpoint
  s=3000,color=blue):
  B:=plot(x/2,x=-10..10,y=-10..10,color=green):
  C:=plot([x(t),y(t),t=-100..100],-10..10,-10..10,numpoints=3000,
  color=red):
  display(A,B,C);
```



6)

```
> restart;
  x:=t->2*t+t^2:
  y:=t->2*t-1/t^2:
```

Domaine de définition :] $-\infty$,0]U]0,$+\infty$[

Limites :

```
> limit(x(t),t=-infinity);
  limit(y(t),t=-infinity);
```

$$\infty$$
$$-\infty$$

```
> limit(x(t),t=infinity);
  limit(y(t),t=infinity);
```

$$\infty$$
$$\infty$$

```
> limit(x(t),t=0);
  limit(y(t),t=0);
```

$$0$$
$$-\infty$$

Donc la courbe a une asymptote verticale d'équation $x=0$ quand t->0.

Branche infinie en $+\infty$ et en $-\infty$

```
> limit(y(t)/x(t),t=infinity);
  limit(y(t)/x(t),t=-infinity);
```

$$0$$
$$0$$

Donc branche parabolique de direction asymptotique Ox. Plus précisément :

```
> taylor(x(t),t=infinity);
  taylor(y(t),t=infinity);
```

$$2\,t+t^2$$
$$2\,t-\frac{1}{t^2}$$

```
> taylor(x(t)-y(t)-y(t)^2/4,t=infinity);
  taylor(x(t)-y(t)-y(t)^2/4,t=-infinity);
```

$$\frac{1}{t}+\frac{1}{t^2}-\frac{1}{4\,t^4}$$
$$\frac{1}{t}+\frac{1}{t^2}-\frac{1}{4\,t^4}$$

Ceci montre que la parabole d'équation $x=y+\frac{y^2}{4}$ est asymptote.

Variations :

```
> xx:=D(x);
  yy:=D(y);
```

$$xx:=t\to 2+2\,t$$
$$yy:=t\to 2+\frac{2}{t^3}$$

```
> solve(xx(t)>0,t);
  solve(yy(t)>0,t);
```

RealRange(Open(-1), ∞)

RealRange(−∞, Open(-1)), RealRange(Open(0), ∞)

```
> solve(xx(t)=0,t);
  solve(yy(t)=0,t);
```

$$-1$$
$$-1,\ \frac{1}{2}+\frac{1}{2}\,I\sqrt{3},\ \frac{1}{2}-\frac{1}{2}\,I\sqrt{3}$$

Point singulier :

```
> x(-1);y(-1);
```

$$-1$$
$$-3$$

```
> taylor(x(t)-x(-1),t=-1);
  taylor(y(t)-y(-1),t=-1);
```

$$(t+1)^2$$
$$-3\,(t+1)^2-4\,(t+1)^3-5\,(t+1)^4-6\,(t+1)^5+\mathrm{O}((t+1)^6)$$

C'est un point de rebroussement (de première espèce).

Point double :

```
> solve({x(u)=x(v),y(u)=y(v),u<>v},{u,v});
```

$$\{u=-\mathrm{RootOf}(\_Z^2+2\,\_Z-1,\,label=\_L1)-2,\ v=\mathrm{RootOf}(\_Z^2+2\,\_Z-1,\,label=\_L1)\}$$

```
> S:=allvalues(%);
```

$$S:=\{u=-1-\sqrt{2},\ v=\sqrt{2}-1\},\ \{u=\sqrt{2}-1,\ v=-1-\sqrt{2}\}$$

```
> subs(S[1],[x(u),y(u)]);
  map(simplify,%);
```

$$\left[-2-2\sqrt{2}+(-1-\sqrt{2})^2,\ -2-2\sqrt{2}-\frac{1}{(-1-\sqrt{2})^2}\right]$$
$$[1,\ -5]$$

Le point double a pour coordonnées (-1,5).

Tracé : pour tracer la parabole, on utilise à nouveau *implicitplot*.

```
> with(plots):
> A:=plot([x(t),y(t),t=-10..10],-5..10,-15..10,numpoints=2000,col
  or=red):
  B:=implicitplot(x=y+y^2/4,x=-5..10,y=-15..10,numpoints=2000,col
  or=blue):
> display(A,B);
```



Exercice 2

```
> restart;
> x:=t->t^3-4*t;
  y:=t->2*t^2-3;
```

$$x:=t\to t^3-4\,t$$
$$y:=t\to 2\,t^2-3$$

```
> plot([x(t),y(t),t=-5..5],-10..10,-5..10);
```



Point double :

```
> solve({x(u)=x(v),y(u)=y(v),u<>v});
```

$$\{v=2, u=-2\}, \{v=-2, u=2\}$$

Les valeurs $t=2$ et $t=-2$ du paramètre donnent le meme point M(t)=(x(t),y(t)). Calculons ses coordoonnées :

Coordonnées du point double P :

```
> x(-2);y(-2);
  x(2);y(2);
```

$$0$$
$$5$$
$$0$$
$$5$$

Vecteur directeur de la première tangente en P :

```
> xx:=D(x);yy:=D(y);
```

$$xx := t \to 3\,t^2-4$$
$$yy := t \to 4\,t$$

```
> xx(-2);yy(2);
```

$$8$$
$$8$$

Vecteur directeur de la deuxième tangente en P :

```
> xx(2);yy(-2);
```

$$8$$
$$-8$$

Les vecteurs sont orthogonaux donc l'angle cherché est $\frac{\pi}{2}$.

Exercice 3

```
> restart;
  x:=t->t^2:
  y:=t->t^3:
> xx:=D(x);yy:=D(y);
```

$$xx := t \to 2\,t$$
$$yy := t \to 3\,t^2$$

Le vecteur tangent à la courbe au point M(t)=(x(t),y(t)) a pour coordonnées : (x'(t),y'(t)).
Equation de la tangente à la courbe en M(t) : (x-x(t)) y'(t) - (y-y(t)) x'(t) = 0.

```
> tangente:=(x-x(t))*yy(t)-(y-y(t))*xx(t)=0;
```

$$tangente := 3\,(x-t^2)\,t^2-2\,(y-t^3)\,t=0$$

Soit P=(x,y) un point du plan, étant sur la tangente en M(u) et la tangente en M(v), et tel que les tangentes soient orthogonales. Les deux premières conditions se traduisent par :

```
> subs(t=u,tangente);
```

$$3\,(x-u^2)\,u^2-2\,(y-u^3)\,u=0$$

```
> subs(t=v,tangente);
```

$$3\,(x-v^2)\,v^2-2\,(y-v^3)\,v=0$$

La troisième par :

```
> xx(u)*xx(v)+yy(u)*yy(v)=0;
```

$$4\,u\,v+9\,u^2\,v^2=0$$

On résoud le système :

```
> solve({subs(t=u,tangente),subs(t=v,tangente),xx(u)*xx(v)+yy(u)*
  yy(v)=0,u<>v});
```

$$\{u=u, v=0, y=\frac{3}{2}\,u\,x-\frac{1}{2}\,u^3, x=x\}, \{v=v, u=0, y=\frac{3}{2}\,v\,x-\frac{1}{2}\,v^3, x=x\},$$
$$\{v=v, y=-\frac{2\,(9\,v^2-4)}{81\,v}, u=-\frac{4}{9\,v}, x=\frac{81\,v^4-36\,v^2+16}{243\,v^2}\}$$

Gardons les solutions avec $u \neq 0$ et $v \neq 0$ (le point (0,0) est singulier et il n'y a pas de droite tangente en ce point).

```
> S:=%[3];
```

$$S := \{v=v, y=-\frac{2\,(9\,v^2-4)}{81\,v}, u=-\frac{4}{9\,v}, x=\frac{81\,v^4-36\,v^2+16}{243\,v^2}\}$$

On obtient une courbe paramétrée $x(v)$, $y(v)$ que l'on peut tracer :

```
> with(plots):
  A:=plot([x(t),y(t),t=-10..10],0..2,-2..2,color=red,numpoints=10
  00):
  B:=plot([1/243*(81*v^4-36*v^2+16)/v^2,-2/81*(9*v^2-4)/v,v=0..5]
  ,color=blue):
  display(A,B);
```



Un calcul à la main montre que cette courbe est en fait une parabole.

Exercice 4

```
> restart;
  x:=t->a*cos(t)^3;
  y:=t->a*sin(t)^3;
```

$$x := t \to a\cos(t)^3$$
$$y := t \to a\sin(t)^3$$

Sa période est $2\,\pi$.

```
> plot( subs(a=1,[x(t),y(t),t=0..2*Pi]) );
```




```
> with(plots):
> ?plots[animate]
> animate( plot, [[x(t),y(t),t=0..2*Pi]], a=-10..10 ) :
```

Exercice 5

```
> restart;
```

1) $\sin\left(\frac{3\,\theta}{2}\right)$ est de période $\frac{4\,\pi}{3}$ et $(\cos(\theta),\sin(\theta))$ est de période $2\,\pi$. Donc la période minimale de la courbe est $4\,\pi$.

```
> plot([sin(3*t/2),t,t=0..4*Pi],coords=polar);
```



2) Soit $x(\theta)=\sin\left(\frac{3\,\theta}{2}\right)\cos(\theta)$ et $y(\theta)=\sin\left(\frac{3\,\theta}{2}\right)\sin(\theta)$.
$x(-\theta)=-x(\theta)$ et $y(-\theta)=y(\theta)$ donc la courbe est symétrique par rapport à l'axe Oy.
$x(\theta+2\,\pi)=-x(\theta)$ et $y(\theta+2\,\pi)=-y(\theta)$ donc la courbe est symétrique pa rapport à O
On en déduit qu'il suffit d'étudier la courbe pour θ dans l'intervalle $[0,\pi]$ et de compléter par symétries.

```
> plot([sin(3*t/2),t,t=0..Pi],coords=polar);
```



C. Armana
armana@math.jussieu.fr

MK1 "Calcul formel" Maple

TP8 : Algèbre linéaire

Rappel : les documents de TP, sujets et corrections de devoir sont disponibles sur :
http://www.math.jussieu.fr/~armana/mk1

But du TP8 :

Nous allons utiliser Maple pour faire de l'algèbre linéaire dans R^n : vecteurs, bases, matrices, systèmes linéaires,...

Et surtout, n'oubliez pas de vous (et de me) poser des questions !

La plupart des commandes de Maple qui permettent de faire de l'algèbre linéaire se trouvent dans la librairie *linalg* :

```
> restart: with(linalg);
Warning, the protected names norm and trace have been redefined and
unprotected
```

[*BlockDiagonal*, *GramSchmidt*, *JordanBlock*, *LUdecomp*, *QRdecomp*, *Wronskian*, *addcol*, *addrow*, *adj*, *adjoint*, *angle*, *augment*, *backsub*, *band*, *basis*, *bezout*, *blockmatrix*, *charmat*, *charpoly*, *cholesky*, *col*, *coldim*, *colspace*, *colspan*, *companion*, *concat*, *cond*, *copyinto*, *crossprod*, *curl*, *definite*, *delcols*, *delrows*, *det*, *diag*, *diverge*, *dotprod*, *eigenvals*, *eigenvalues*, *eigenvectors*, *eigenvects*, *entermatrix*, *equal*, *exponential*, *extend*, *ffgausselim*, *fibonacci*, *forwardsub*, *frobenius*, *gausselim*, *gaussjord*, *geneqns*, *genmatrix*, *grad*, *hadamard*, *hermite*, *hessian*, *hilbert*, *htranspose*, *ihermite*, *indexfunc*, *innerprod*, *intbasis*, *inverse*, *ismith*, *issimilar*, *iszero*, *jacobian*, *jordan*, *kernel*, *laplacian*, *leastsqrs*, *linsolve*, *matadd*, *matrix*, *minor*, *minpoly*, *mulcol*, *mulrow*, *multiply*, *norm*, *normalize*, *nullspace*, *orthog*, *permanent*, *pivot*, *potential*, *randmatrix*, *randvector*, *rank*, *ratform*, *row*, *rowdim*, *rowspace*, *rowspan*, *rref*, *scalarmul*, *singularvals*, *smith*, *stackmatrix*, *submatrix*, *subvector*, *sumbasis*, *swapcol*, *swaprow*, *sylvester*, *toeplitz*, *trace*, *transpose*, *vandermonde*, *vecpotent*, *vectdim*, *vector*, *wronskian*]

Pensez à recharger la librairie après chaque *restart* !

1. Les vecteurs

1.1 Construction d'un vecteur

La commande *vector* permet de créer un vecteur. Voici quelques exemples :

```
> vector([1,4,6]);
  whattype(%);
```

$$[1, 4, 6]$$

array

```
> vector(5,i->i^2);
```

$$[1, 4, 9, 16, 25]$$

```
> randvector(4);
```

$$[-85, -55, -37, -35]$$

Afficher le contenu d'un vecteur :

```
> v:=vector([7,8,9,10]);
```

$$v := [7, 8, 9, 10]$$

```
> v;
```

$$v$$

Pour afficher à nouveau le vecteur contenu dans la variable *v*, on utilise la commande *evalm* :

```
> evalm(v);
```

$$[7, 8, 9, 10]$$

Extraire un coefficient d'un vecteur :

```
> v[2]; # deuxième coefficient
```

$$8$$

Modifier le contenu d'un vecteur

Par affectation des coefficients :

```
> v[1]:=0;
  evalm(v);
```

$$v_1 := 0$$

$$[0, 8, 9, 10]$$

Copier un vecteur

De subtiles règles d'évaluation dans Maple amènent le problème suivant lorsqu'on souhaite copier un vecteur dans une autre variable :

```
> w:=v;
  v[1]:=2;
  evalm(v);evalm(w);
```

$$w := v$$

$$v_1 := 2$$

$$[2, 8, 9, 10]$$

$$[2, 8, 9, 10]$$

Toute modification de *v* entraine une modification de sa copie *w* ! Pour y remédier, on utilise la commande *copy* :

```
> w:=copy(v);
  v[1]:=3;
  evalm(v);evalm(w);
```

$$w := [2, 8, 9, 10]$$

$$v_1 := 3$$

$$[3, 8, 9, 10]$$

$$[2, 8, 9, 10]$$

1.2 Quelques opérations sur les vecteurs

Somme de deux vecteurs :

```
> u:=vector([-7,4,7,3]);
```

$$u := [-7, 4, 7, 3]$$

```
> matadd(u,v);
```

$$[0, 12, 16, 13]$$

ou bien :

```
> evalm(u+v);
```

$$[0, 12, 16, 13]$$

Multiplication d'un vecteur par un scalaire (réel) :

```
> scalarmul(v,Pi);
```

$$[7\pi, 8\pi, 9\pi, 10\pi]$$

ou bien :

```
> evalm(Pi*v);
```

$$[7\pi, 8\pi, 9\pi, 10\pi]$$

Taille d'un vecteur (nombre de composantes) :

```
> vectdim(u);
```

$$4$$

Tester une égalité de deux vecteurs :

```
> equal(u,v);
```

false

Produit scalaire de deux vecteurs :

```
> dotprod(vector([1,0]),vector([0,1]));
```

$$0$$

Produit vectoriel de deux vecteurs :

```
> crossprod(vector([1,1,0]),vector([1,0,1]));
```

$$[1, -1, -1]$$

1.3 Opérations avancées

Trouver une base du sous-espace vectoriel engendré par un *ensemble* de vecteurs :

```
> u:=vector([1,0,0]);
  v:=vector([1,1,0]);
  w:=vector([1,0,1]);
  t:=vector([0,0,2]);
```

$$u := [1, 0, 0]$$
$$v := [1, 1, 0]$$
$$w := [1, 0, 1]$$
$$t := [0, 0, 2]$$

```
> basis( { u,v,w,t } );
```

$$\{v, u, t\}$$

2. Les matrices

2.1 Construction d'une matrice

La commande *matrix* permet de créer une matrice. Voici quelques exemples :

```
> matrix([[1,2,3],[4,5,6],[7,8,9]]);
  whattype(%);
```

$$\begin{bmatrix} 1 & 2 & 3 \\ 4 & 5 & 6 \\ 7 & 8 & 9 \end{bmatrix}$$

array

```
> matrix(3,3,[1,2,3,4,5,6,7,8,9]);
```

$$\begin{bmatrix} 1 & 2 & 3 \\ 4 & 5 & 6 \\ 7 & 8 & 9 \end{bmatrix}$$

```
> matrix(5,3,(i,j)->i+j);
```

$$\begin{bmatrix} 2 & 3 & 4 \\ 3 & 4 & 5 \\ 4 & 5 & 6 \\ 5 & 6 & 7 \\ 6 & 7 & 8 \end{bmatrix}$$

```
> A:=randmatrix(2,3);
```

$$A := \begin{bmatrix} -35 & 97 & 50 \\ 79 & 56 & 49 \end{bmatrix}$$

Il existe beaucoup d'autres fonctions permettant de créer des matrices. Consultez les commandes fournies dans *linalg* pour en savoir plus : par exemple *diag, band, blockmatrix augment*,...

<u>Afficher le contenu d'une matrice</u>

```
> A;
```

$$A$$

Pour afficher à nouveau la matrice contenue dans la variable *A*, on utilise la commande *evalm* :

```
> evalm(A);
```

$$\begin{bmatrix} -35 & 97 & 50 \\ 79 & 56 & 49 \end{bmatrix}$$

<u>Extraire un coefficient d'une matrice</u>

```
> A[2,3]; # deuxième ligne, troisième colonne
```

$$49$$

<u>Modifier d'une matrice</u>

Par affectation des coefficients :

```
> A[1,1]:=Pi;
  evalm(A);
```

$$A_{1,1} := \pi$$

$$\begin{bmatrix} \pi & 97 & 50 \\ 79 & 56 & 49 \end{bmatrix}$$

<u>Copier une matrice</u>

De subtiles règles d'évaluation dans Maple amènent le problème suivant lorsqu'on souhaite copier une matrice dans une autre variable :

```
> B:=A;
  A[1,1]:=1;
```

```
evalm(A);evalm(B);
```

$$B := A$$

$$A_{1,\,1} := 1$$

$$\begin{bmatrix} 1 & 97 & 50 \\ 79 & 56 & 49 \end{bmatrix}$$

$$\begin{bmatrix} 1 & 97 & 50 \\ 79 & 56 & 49 \end{bmatrix}$$

Toute modification de *A* entraine une modification de sa copie *B* ! Pour y remédier, on utilise la commande *copy* :

```
> B:=copy(A);
  A[1,1]:=999;
  evalm(A);evalm(B);
```

$$B := \begin{bmatrix} 1 & 97 & 50 \\ 79 & 56 & 49 \end{bmatrix}$$

$$A_{1,\,1} := 999$$

$$\begin{bmatrix} 999 & 97 & 50 \\ 79 & 56 & 49 \end{bmatrix}$$

$$\begin{bmatrix} 1 & 97 & 50 \\ 79 & 56 & 49 \end{bmatrix}$$

2.2 Quelques opérations sur les matrices

Somme de deux matrices :

```
> B:=matrix(2,3,[1$6]);
```

$$B := \begin{bmatrix} 1 & 1 & 1 \\ 1 & 1 & 1 \end{bmatrix}$$

```
> matadd(A,B);
```

$$\begin{bmatrix} 1000 & 98 & 51 \\ 80 & 57 & 50 \end{bmatrix}$$

ou bien :

```
> evalm(A+B);
```

$$\begin{bmatrix} 1000 & 98 & 51 \\ 80 & 57 & 50 \end{bmatrix}$$

Multiplication d'une matrice par un scalaire (réel) :

```
> scalarmul(A,x);
```

$$\begin{bmatrix} 999\,x & 97\,x & 50\,x \\ 79\,x & 56\,x & 49\,x \end{bmatrix}$$

ou bien :

```
> evalm(x*A);
```

$$\begin{bmatrix} 999\,x & 97\,x & 50\,x \\ 79\,x & 56\,x & 49\,x \end{bmatrix}$$

Multiplication de deux matrices (produit matriciel) :

```
> M:=matrix(2,2,[a1,b1,c1,d1]);
  N:=matrix(2,2,[a2,b2,c2,d2]);
```

$$M := \begin{bmatrix} a1 & b1 \\ c1 & d1 \end{bmatrix}$$

$$N := \begin{bmatrix} a2 & b2 \\ c2 & d2 \end{bmatrix}$$

```
> multiply(M,N);
```

$$\begin{bmatrix} a1\ a2 + b1\ c2 & a1\ b2 + b1\ d2 \\ c1\ a2 + d1\ c2 & c1\ b2 + d1\ d2 \end{bmatrix}$$

ou bien :

```
> evalm(M&*N);
```

$$\begin{bmatrix} a1\ a2 + b1\ c2 & a1\ b2 + b1\ d2 \\ c1\ a2 + d1\ c2 & c1\ b2 + d1\ d2 \end{bmatrix}$$

<u>Notez bien</u> le **&***, qui dénote la multiplication des matrices ! à différencier de *, qui ne fonctionne pas ici :

```
> evalm(M*N);
Error, (in evalm/evaluate) use the &* operator for matrix/vector
multiplication
```

Multiplication d'une matrice par un vecteur :

```
> multiply(A,vector([1,0,0]));
```

$$[999,\ 79]$$

ou bien avec &* :

```
> evalm(A&*vector([1,0,0]));
```

$$[999,\ 79]$$

Taille d'une matrice : nombre de lignes, nombres de colonnes

```
> rowdim(A) ; coldim(A);
```

$$2$$

$$3$$

Tester une égalité de deux matrices :

```
> equal(A,B);
```

false

2.3 Opérations avancées

<u>Trace d'une matrice</u> (somme des coefficients de la première diagonale) :

```
> trace(A);
```

$$5$$

<u>Transposée d'une matrice</u> :

```
> transpose(A);
```

$$\begin{bmatrix} 999 & 79 \\ 97 & 56 \\ 50 & 49 \end{bmatrix}$$

<u>Calcul de l'inverse d'une matrice</u> (carrée inversible) :

```
> A:=matrix(3,3,[7/10,3/5,1/10,-3/5,11/5,1/5,-3/10,3/5,11/10]):
  inverse(A);
```

$$\begin{bmatrix} \frac{23}{20} & \frac{-3}{10} & \frac{-1}{20} \\ \frac{3}{10} & \frac{2}{5} & \frac{-1}{10} \\ \frac{3}{20} & \frac{-3}{10} & \frac{19}{20} \end{bmatrix}$$

<u>Calcul des valeurs propres d'une matrice carrée</u> :

```
> eigenvals(A);
```

$$2, 1, 1$$

<u>Calcul des vecteurs propres d'une matrice carrée</u> :

```
> eigenvects(A);
```

$$[2, 1, \{[1, 2, 1]\}], [1, 2, \{[0, 1, -6], [1, 0, 3]\}]$$

Que signifie la réponse de Maple ? Au besoin, consultez l'aide de Maple sur *linalg[eigenvects]*.

<u>Trouver une base du noyau</u> de l'application linéaire associée à une matrice :

```
> kernel(A);
```

$$\{\ \}$$

```
> B:=matrix(2,2,[1,2,1/2,1]);
```

$$B := \begin{bmatrix} 1 & 2 \\ \frac{1}{2} & 1 \end{bmatrix}$$

```
> kernel(B);
```

$$\{[-2, 1]\}$$

Que signifient les réponses de Maple ?

<u>Trouver une base de l'image</u> de l'application linéaire associée à une matrice :
(= une base du sous-espace vectoriel engendré par les vecteurs colonnes de la matrice)

```
> colspace(B);
```

$$\left\{\left[1, \frac{1}{2}\right]\right\}$$

<u>Rang de la matrice</u> (= dimension de l'image)

```
> rank(B);
```

$$1$$

<u>Déterminant d'une matrice carrée</u> (on rappelle qu'une matrice carrée est inversible si et seulement si son déterminant est non nul) :

```
> det(A);
```

$$2$$

3. Les systèmes linéaires

3.1 Systèmes linéaires donnés par des équations

On utilise *solve*. Par exemple, pour le système :

$$\begin{cases} x + y + z = 3 \\ z + 2\,y - z = -1 \\ -x + y + 2\,z = 0 \end{cases}$$

```
> solve( {x+y+z=3 , x+2*y-z=-1 , -x+y+2*z=0} , {x,y,z} );
```

$$\left\{x = \frac{16}{7}, z = \frac{11}{7}, y = \frac{-6}{7}\right\}$$

3.2 Systèmes linéaires donnés par une matrice

On utilise *linsolve*. Par exemple, le système précédent correspond à A.X=B, où A est la matrice :

```
> A:=matrix( [ [1,1,1], [1,2,-1], [-1,1,2] ] );
```

$$A := \begin{bmatrix} 1 & 1 & 1 \\ 1 & 2 & -1 \\ -1 & 1 & 2 \end{bmatrix}$$

et B le vecteur colonne :

```
> B:=vector([3,-1,0]);
```

$$B := [3, -1, 0]$$

On résout le système par :

```
> linsolve(A,B);
```

$$\left[\frac{16}{7}, \frac{-6}{7}, \frac{11}{7}\right]$$

3.3 Le pivot de Gauss

Maple peut effectuer le pivot de Gauss sur une matrice avec la commande *gausselim*. Le résultat est une matrice échelonnée mais non réduite. En utilisant cette commande avec *augment*, on peut résoudre des systèmes linéaires.

```
> C:=augment(A,B);
```

$$C := \begin{bmatrix} 1 & 1 & 1 & 3 \\ 1 & 2 & -1 & -1 \\ -1 & 1 & 2 & 0 \end{bmatrix}$$

augment crée une nouvelle matrice en mettant le vecteur colonne B à droite de A (A | B) .

```
> gausselim(C);
```

$$\begin{bmatrix} 1 & 1 & 1 & 3 \\ 0 & 1 & -2 & -4 \\ 0 & 0 & 7 & 11 \end{bmatrix}$$

Enfin, la commande *backsub*, appliquée à la matrice échelonnée, permet de trouver le(s) solution(s) :

```
> backsub(%);
```

$$\left[\frac{16}{7}, \frac{-6}{7}, \frac{11}{7}\right]$$

Université Paris 7 | **Premier semestre 2005-2006**
Licence 1ère année | **MK1 - Maple**

Feuille d'exercices n°8

Exercice 1. Prendre deux vecteurs u et v de taille 3 aléatoirement dans Maple. Calculer $w = u \wedge v$ et les produits scalaires (u, w) et (v, w). Que constatez-vous ?

Exercice 2. Calculer l'angle formé entre les vecteurs $\begin{pmatrix} 2 \\ 1 \end{pmatrix}$ et $\begin{pmatrix} 2\sqrt{3} - 1 \\ \sqrt{3} + 2 \end{pmatrix}$.

Exercice 3. Prendre deux matrices carrées A et B aléatoirement dans Maple. Vérifier que : $A + B = B + A$, $(AB)^{-1} = B^{-1}A^{-1}$ et (très probablement) $AB \neq BA$.

Exercice 4. Expliquer pourquoi les commandes suivantes fonctionnent ou pas.

```
[> u :=vector([1,0]) ; v :=vector([3,5,2]) ; matadd(u,v) ;
[> A :=randmatrix(2,2) ; B :=randmatrix(3,3) ; matadd(A,B) ;
[> multiply(A,v) ;
[> C :=matrix([[1,2,3],[4,5,6]]) ; multiply(C,B) ; multiply(B,C) ;
```

Exercice 5. Ecrire une procédure qui prend en entrée une matrice M et renvoie la matrice M à laquelle on a ajouté 1 à tous les coefficients de la première colonne.

Exercice 6. Soit $M = \begin{pmatrix} 0 & 1 & 0 & 0 \\ 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & 1 \\ 1 & 0 & 0 & 0 \end{pmatrix}$. Calculer M^5 et le déterminant de M. Sans utiliser Maple, en déduire que $M^4 = I$, $M^{-1} = M^3$ et deviner ce que vaut M^{125}. Vérifier ensuite les résultats avec Maple.

Exercice 7.

1) Rappeler la définition d'un famille libre (= linéairement indépendante) de vecteurs. Pour chacune des familles suivantes, déterminer si elles sont libres en écrivant le système (linéaire) correspondant, et en le faisant résoudre par la commande *solve* de Maple.
$$\begin{pmatrix} 3 \\ 4 \\ -2 \end{pmatrix}, \begin{pmatrix} 5 \\ 1 \\ 7 \end{pmatrix} \;;\; \begin{pmatrix} 1 \\ -1 \\ 0 \end{pmatrix}, \begin{pmatrix} 2 \\ 0 \\ 5 \end{pmatrix}, \begin{pmatrix} 8 \\ -2 \\ 15 \end{pmatrix}$$

2) Rappeler la définition d'une famille génératrice de vecteurs d'un sous-espace vectoriel. Pour chacune des familles suivantes, déterminer si elles sont génératrices de $\mathbb{R}^3$ en écrivant le système (linéaire) correspondant, et en le faisant résoudre par la commande *solve* de Maple.
$$\begin{pmatrix} 1 \\ -1 \\ 0 \end{pmatrix}, \begin{pmatrix} 1 \\ 0 \\ 0 \end{pmatrix}, \begin{pmatrix} 0 \\ 1 \\ 1 \end{pmatrix} \;;\; \begin{pmatrix} 2 \\ 4 \\ 6 \end{pmatrix}, \begin{pmatrix} 1 \\ 3 \\ 5 \end{pmatrix}, \begin{pmatrix} 1 \\ 1 \\ 1 \end{pmatrix}, \begin{pmatrix} 4 \\ 5 \\ 7 \end{pmatrix} \;;\; \begin{pmatrix} 1 \\ -1 \\ 0 \end{pmatrix}, \begin{pmatrix} 1 \\ 0 \\ 0 \end{pmatrix}, \begin{pmatrix} 0 \\ -1 \\ 0 \end{pmatrix}$$

Exercice 8. Donner une base du sous-espace vectoriel engendré par les familles de la question 2 de l'exercice 7,

Exercice 9. On considère les vecteurs $v_1 = [2\ 4\ 5\ 6]$, $v_2 = [1\ 2\ 5\ 3]$, $v_3 = [3\ 1\ -1\ 0]$, $v_4 = [4\ 3\ 4\ 3]$. Donner une base du sous-espace vectoriel engendré par v_1, v_2, v_3, v_4. Le vecteur $w = [4\ 3\ -1\ 3]$ est-il dans ce sous-espace vectoriel ? Si oui, exprimez-le comme combinaison linéaire des vecteurs de la base.

Exercice 10. Résoudre les systèmes linéaires $A.X = B$ suivants. Quelle est la nature de l'ensemble des solutions ? Vérifier les résultats en traçant dans $\mathbb{R}^3$ les plans correspondant aux équations (commande *implicitplot3d* de la librairie *plots*).

$$A = \begin{pmatrix} 3 & 2 & -1 \\ 2 & -3 & 4 \\ 1 & 1 & -3 \end{pmatrix}, B = \begin{pmatrix} 1 \\ 3 \\ 2 \end{pmatrix} \qquad A = \begin{pmatrix} 3 & 2 & 1 \\ 2 & -3 & 5 \\ 1 & 1 & 0 \end{pmatrix}, B = \begin{pmatrix} 1 \\ -8 \\ 1 \end{pmatrix}$$

$$A = \begin{pmatrix} 1 & 1 & 1 \\ 2 & 2 & 2 \\ -1 & -1 & -1 \end{pmatrix}, B = \begin{pmatrix} 0 \\ 0 \\ 0 \end{pmatrix} \qquad A = \begin{pmatrix} 2 & -5 & 4 \\ 1 & -2 & 1 \\ 1 & -4 & 5 \end{pmatrix}, B = \begin{pmatrix} -3 \\ 5 \\ 10 \end{pmatrix}$$

Exercice 11. Résoudre le système linéaire $A.X = B$ suivant à un paramètre m en utilisant le pivot de Gauss. On pourra distinguer plusieurs cas suivant les valeurs de m. Résoudre ensuite ce système en utilisant *linsolve*. Que constatez-vous ?

$$A = \begin{pmatrix} 1 & 1 & m \\ 1 & 1 & -1 \\ 1 & m & -1 \end{pmatrix}, B = \begin{pmatrix} m \\ 1 \\ 1 \end{pmatrix}$$

Exercice 12.

1) Déterminer une base du noyau et de l'image de chacune des matrices suivantes.

$$A = \begin{pmatrix} 3 & 2 & -1 \\ 2 & -3 & 4 \\ 1 & 1 & -3 \end{pmatrix}, B = \begin{pmatrix} 3 & 2 & 1 \\ 2 & -3 & 5 \\ 1 & 1 & 0 \end{pmatrix}, C = \begin{pmatrix} 1 & 1 \\ 2 & 2 \\ 4 & 5 \end{pmatrix}$$

2) On appelle dimension d'un sous-espace vectoriel de $\mathbb{R}^n$ le nombre de vecteurs d'une de ses bases. Sur les matrices précédentes, calculer la dimension du noyau, la dimension de l'image et la somme de ces dimensions. Que constatez-vous ?

Exercice 13. Soit $M = \begin{pmatrix} 2 & -2 & 1 \\ 2 & -3 & 2 \\ -1 & 2 & 0 \end{pmatrix}$. Déterminer les valeurs propres et les vecteurs propres de M. Démontrer que les vecteurs propres forment une base de $\mathbb{R}^3$. Ecrire la matrice de passage P de la base canonique à cette nouvelle base. Calculer $P^{-1}MP$. Pouvez-vous expliquer le résultat obtenu ?

Correction du TP n°8

Exercice 1

```
> restart;with(linalg):
Warning, the protected names norm and trace have been redefined and
unprotected
```

Après chaque *restart*, n'oubliez pas de faire un *with(linalg)* si vous souhaitez faire de l'algèbre linéaire !

```
> u:=randvector(3); v:=randvector(3); w:=crossprod(u,v);
  dotprod(u,w); dotprod(v,w);
```

$$u := [79, 56, 49]$$
$$v := [63, 57, -59]$$
$$w := [-6097, 7748, 975]$$
$$0$$
$$0$$

On remarque que w est orthogonal à u et v. C'est une propriété du produit vectoriel.

Exercice 2

```
> u:=vector([2,1]); v:=vector([2*sqrt(3)-1,sqrt(3)+2]);
```

$$u := [2, 1]$$
$$v := [2\sqrt{3}-1, \sqrt{3}+2]$$

On rappelle la propriété du produit scalaire : (u,v)=||u|| ||v|| cos(u,v). Dans Maple, la norme d'un vecteur s'obtient par la commande *norm(, 2)* de la librairie *linalg*. On peut aussi faire le calcul à la main.

```
> dotprod(u,v)/(norm(u,2)*norm(v,2));
```

$$\frac{\sqrt{3}\sqrt{5}}{\sqrt{(2\sqrt{3}-1)^2+(\sqrt{3}+2)^2}}$$

```
> simplify(%);
```

$$\frac{\sqrt{3}}{2}$$

```
> arccos(%);
```

$$\frac{\pi}{6}$$

Donc l'angle est $\frac{\pi}{6}$. Autre méthode : utiliser la commande *angle* de *linalg* qui calcule l'angle entre deux vecteurs par la même méthode que précédemment.

```
> ?linalg[angle]
> angle(u,v);
```

$$\arccos\left(\frac{\sqrt{3}\sqrt{5}}{\sqrt{(2\sqrt{3}-1)^2+(\sqrt{3}+2)^2}}\right)$$

Bien entendu, Maple n'a pas simplifié l'expression, et il faut le lui demander expressément.

Exercice 3

```
> A:=randmatrix(4,4): B:=randmatrix(4,4):
> evalm(matadd(A,B)-matadd(B,A));
```

$$\begin{bmatrix} 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \end{bmatrix}$$

Donc A+B=B+A.

```
> evalm(inverse(A&*B)-inverse(B)&*inverse(A));
```

$$\begin{bmatrix} 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \end{bmatrix}$$

Donc $AB^{(-1)} = B^{(-1)} A^{(-1)}$.

On calcule AB-BA par :

```
> matadd(multiply(A,B),-multiply(B,A));
```

$$\begin{bmatrix} -6683 & -13056 & -436 & -7078 \\ -25659 & -4688 & 7745 & -3096 \\ -8567 & -20601 & 5330 & -11399 \\ -2121 & 500 & -2714 & 6041 \end{bmatrix}$$

ou bien par :

```
> evalm(A&*B-B&*A);
```

$$\begin{bmatrix} -6683 & -13056 & -436 & -7078 \\ -25659 & -4688 & 7745 & -3096 \\ -8567 & -20601 & 5330 & -11399 \\ -2121 & 500 & -2714 & 6041 \end{bmatrix}$$

donc ici, $A\,B \neq B\,A$.

Exercice 4

```
> u:=vector([1,0]): v:=vector([3,5,2]): matadd(u,v);
Error, (in matadd) vector dimensions incompatible
```

On ne peut pas faire $u+v$ car u et v n'ont pas la même taille.

```
> A:=randmatrix(2,2): B:=randmatrix(3,3): matadd(A,B);
Error, (in matadd) matrix dimensions incompatible
```

On ne peut pas faire $A+B$ car A et B n'ont pas la même taille.

```
> multiply(A,v);
Error, (in multiply) non matching dimensions for vector/matrix product
```

On ne peut pas faire $A\,.\,v$ car le nombre de colonnes de A n'est pas égal au nombre de lignes de v.

```
> C:=matrix([[1,2,3],[4,5,6]]); multiply(C,B); multiply(B,C);
```

$$C := \begin{bmatrix} 1 & 2 & 3 \\ 4 & 5 & 6 \end{bmatrix}$$

```
Error, (in multiply) expecting a matrix or a vector

Error, (in multiply) expecting a matrix or a vector
```

On peut faire C . B car le nombre de colonnes de C est égal au nombre de lignes de B. On ne peut pas faire B . C car le nombre de colonnes de B n'est pas égal au nombre de lignes de C.

Exercice 5

Une façon de faire :

```
> ajoute:=proc(M)
  local n,N,k;
  n:=rowdim(M);
  for k from 1 to n do
      M[k,1]:=M[k,1]+1;
  od;
  RETURN(evalm(M));
  end:
> ajoute(C);
```

$$\begin{bmatrix} 2 & 2 & 3 \\ 5 & 5 & 6 \end{bmatrix}$$

Attention à bien mettre *evalm* dans le *RETURN*, sinon la procédure n'affiche pas le contenu de la matrice, mais seulement la lettre *M*.

Exercice 6

```
> M:=matrix(4,4,[0,1,0,0,0,0,1,0,0,0,0,1,1,0,0,0]);
```

$$M := \begin{bmatrix} 0 & 1 & 0 & 0 \\ 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & 1 \\ 1 & 0 & 0 & 0 \end{bmatrix}$$

```
> det(M);
```

$$-1$$

```
> evalm(M^5);
```

$$\begin{bmatrix} 0 & 1 & 0 & 0 \\ 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & 1 \\ 1 & 0 & 0 & 0 \end{bmatrix}$$

Donc $M^5 = M$. La matrice est inversible car son déterminant est non nul. En multipliant l'égalité précédente par $M^{(-1)}$, on en déduit $M^4 = M^{(-1)} M = I$. En multipliant à nouveau par $M^{(-1)}$, on en déduit $M^3 = M^{(-1)}$. On vérifie les calculs à l'aide de Maple :

```
> evalm(M^4);
```

$$\begin{bmatrix} 1 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 \\ 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & 1 \end{bmatrix}$$

```
> evalm(M^3-inverse(M));
```

$$\begin{bmatrix} 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \end{bmatrix}$$

$M^{125} = M^{(5.5\,.\,5)} = M^{5.5} = M^5 = M$ (en utilisant $M^5 = M$). On le vérifie :

```
> evalm(M^125-M);
```

$$\begin{bmatrix} 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \end{bmatrix}$$

Exercice 7

1) La famille de vecteurs $x_1,...,x_n$ est libre si la seule solution de $\lambda_1\, x_1 + ... + \lambda_n\, x_n = 0$ (où les λ sont des réels) est $\lambda_1 = ... = \lambda_n = 0$.

Première famille : cela revient à voir les solutions du système :

| $3\,x + 5\,y = 0$
| $4\,x + y = 0$
| $-2\,x + 7\,y = 0$

```
> solve({3*x+5*y=0,4*x+y=0,-2*x+7*y=0},{x,y});
```

$$\{y = 0, x = 0\}$$

Donc la famille est libre.

Deuxième famille :

| $x + 2\,y + 8\,z = 0$
| $-x - 2\,z = 0$
| $5\,y + 15\,z = 0$

```
> solve({x+2*y+8*z=0,-x-2*z=0,5*y+15*z=0},{x,y,z});
```

$$\{x = -2\,z, y = -3\,z, z = z\}$$

Donc la famille n'est pas libre.

2) La famille de vecteurs $x_1,...,x_n$ est génératrice de R^3 si : pour tout vecteur $x \in R^3$ il existe des réels $\lambda_1,...,\lambda_n$ tels que $x = \lambda_1\, x_1 + ... + \lambda_n\, x_n$.

Première famille : soit X le vecteur (x,y,z). Cela revient à voir les solutions du système (d'inconnues $l1,l2,l3$)

| $l1 + l2 = x$
| $-l1 + l3 = y$
| $l3 = z$

```
> solve({l1+l2=x,-l1+l3=y,l3=z},{l1,l2,l3});
```

$$\{l3 = z, l1 = z - y, l2 = -z + y + x\}$$

Donc la famille est génératrice.

Deuxième famille : soit X le vecteur (x,y,z). Cela revient à voir les solutions du système (d'inconnues $l1,l2,l3,l4$)

| $2\,l1 + l2 + l3 + 4\,l4 = x$
| $4\,l1 + 3\,l2 + l3 + 5\,l4 = y$
| $6\,l1 + 5\,l2 + l3 + 7\,l4 = z$

```
> solve({2*l1+l2+l3+4*l4 = x, 4*l1+3*l2+l3+5*l4 =
  y,6*l1+5*l2+l3+7*l4 = z},{l1,l2,l3,l4});
```

$$\{l4 = x + z - 2\,y,\ l1 = \frac{13\,y}{2} - l3 - 2\,x - \frac{7\,z}{2},\ l2 = -5\,y + l3 + x + 3\,z,\ l3 = l3\}$$

Le coefficient $l3$ n'est pas déterminé de façon unique. Mais en prenant n'importe quelle valeur pour $l3$, cela donne $l1$,$l2$ et $l4$. Donc la famille est génératrice.

Troisième famille : Soit X le vecteur (x,y,z). Cela revient à voir les solutions du système (d'inconnues $l1$,$l2$,$l3$)

| $l1 + l2 = x$
| $-l1 - l3 = y$
| $0 = z$

```
> solve({l1+l2=x,-l1-l3=y,0=z},{l1,l2,l3});
```

Pas de solution, donc la famille n'est pas génératrice.

Exercice 8

On utilise la commande *basis* (voir document de cours).

```
> u:=vector([1,-1,0]);v:=vector([1,0,0]);w:=vector([0,1,1]);basis
  ({u,v,w});
```

$$u := [1, -1, 0]$$
$$v := [1, 0, 0]$$
$$w := [0, 1, 1]$$
$$\{u, v, w\}$$

```
> u:=vector([2,4,6]);v:=vector([1,3,5]);w:=vector([1,1,1]);z:=vec
  tor([4,5,7]);basis({u,v,w,z});
```

$$u := [2, 4, 6]$$
$$v := [1, 3, 5]$$
$$w := [1, 1, 1]$$
$$z := [4, 5, 7]$$
$$\{u, v, z\}$$

```
> u:=vector([1,-1,0]);v:=vector([1,0,0]);w:=vector([0,-1,0]);basi
  s({u,v,w});
```

$$u := [1, -1, 0]$$
$$v := [1, 0, 0]$$
$$w := [0, -1, 0]$$
$$\{u, v\}$$

Exercice 9

```
> u1:=vector([2,4,5,6]);u2:=vector([1,2,5,3]);u3:=vector([3,1,-1,
  0]);u4:=vector([4, 3, 4, 3]);
```

$$u1 := [2, 4, 5, 6]$$
$$u2 := [1, 2, 5, 3]$$
$$u3 := [3, 1, -1, 0]$$
$$u4 := [4, 3, 4, 3]$$

```
> basis({u1,u2,u3,u4});
```

$$\{u1, u4, u2\}$$

```
> w:=vector([4,3,-1,3]);
```

$$w := [4, 3, -1, 3]$$

On voit si w s'écrit comme combinaison linéaire de la forme $x\,u_1 + y\,u_2 + z\,u_4$ en résolvant le système (d'inconnues x, y, z) :

| $2\,x + y + 4\,z = 4$
| $4\,x + 2\,y + 3\,z = 3$
| $5\,x + 5\,y + 4\,z = -1$
| $6\,x + 3\,y + 3\,z = 3$

```
> solve({2*x+y+4*z = 4,4*x+2*y+3*z = 3,5*x+5*y+4*z =
  -1,6*x+3*y+3*z = 3},{x,y,z});
```

$$\{z = 1, x = 1, y = -2\}$$

Donc $w = u_1 - 2\,u_2 + u_4$.

Exercice 10

<u>Premier système :</u>

```
> A:=matrix(3,3,[3,2,-1,2,-3,4,1,1,-3]);B:=vector([1,3,2]);
  v:=linsolve(A,B);
```

$$A := \begin{bmatrix} 3 & 2 & -1 \\ 2 & -3 & 4 \\ 1 & 1 & -3 \end{bmatrix}$$
$$B := [1, 3, 2]$$
$$v := \left[1, \frac{-7}{5}, \frac{-4}{5}\right]$$

Le système a une unique solution : $\left[1, -\frac{7}{5}, -\frac{4}{5}\right]$.

```
> with(plots):
Warning, the name changecoords has been redefined
```

```
> ?plots[implicitplot3d]
> eq:=3*x+2*y-z=1,2*x-3*y+4*z=3,x+y-3*z=2;
  implicitplot3d([eq],x=-5..5,y=-5..5,z=-5..5,color=[red,blue,gre
  en]);
```

$$eq := 3\,x + 2\,y - z = 1,\ 2\,x - 3\,y + 4\,z = 3,\ x + y - 3\,z = 2$$



On voit que les trois plans se coupent en un unique point.

<u>Deuxième système :</u>

```
> A:=matrix(3,3,[3,2,1,2,-3,5,1,1,0]); B:=vector([1,-8,1]);
  linsolve(A,B);
```

$$A := \begin{bmatrix} 3 & 2 & 1 \\ 2 & -3 & 5 \\ 1 & 1 & 0 \end{bmatrix}$$
$$B := [1, -8, 1]$$

$$[\_t_1, -\_t_1+1, -\_t_1-1]$$

Il y a une famille de solutions : les vecteurs $(t, -t+1, -t-1)$ pour t dans R.

```
> eq:=3*x+2*y+z=1,2*x-3*y+5*z=-8,x+y=1;
  implicitplot3d([eq],x=-5..5,y=-5..5,z=-5..5,color=[red,blue,gre
  en]);
```

$$eq := 3\,x+2\,y+z=1,\ 2\,x-3\,y+5\,z=-8,\ x+y=1$$



On voit que les trois plans se coupent en une droite.

<u>Troisième système :</u>

```
> A:=matrix(3,3,[1,1,1,2,2,2,-1,-1,-1]); B:=vector([0,0,0]);
  linsolve(A,B);
```

$$A := \begin{bmatrix} 1 & 1 & 1 \\ 2 & 2 & 2 \\ -1 & -1 & -1 \end{bmatrix}$$

$$B := [0, 0, 0]$$

$$[-\_t_1-\_t_2, \_t_1, \_t_2]$$

Il y a une famille de solutions : les vecteurs $(-t-u, t, u)$ pour t et u dans R. L'espace des solutions est un plan de R^3.

```
> eq:=x+y+z=0,2*x+2*y+2*z,-x-y-z=0;
  implicitplot3d([eq],x=-5..5,y=-5..5,z=-5..5,color=[red,blue,gre
  en]);
```

$$eq := x+y+z=0,\ 2\,x+2\,y+2\,z,\ -x-y-z=0$$



Les trois plans sont confondus. L'ensemble des solutions est ce plan, d'équation est : $x+y+z=0$.

<u>Quatrième système :</u>

```
> A:=matrix(3,3,[2,-5,4,1,-2,1,1,-4,5]); B:=vector([-3,5,10]);
  linsolve(A,B);
```

$$A := \begin{bmatrix} 2 & -5 & 4 \\ 1 & -2 & 1 \\ 1 & -4 & 5 \end{bmatrix}$$

$$B := [-3, 5, 10]$$

L'ensemble des solutions est vide.

```
> eq:=2*x-5*y+4*z=-3,x-2*y+z=5,x-4*y+5*z=10;
  implicitplot3d([eq],x=-10..10,y=-10..10,z=-10..10,color=[red,bl
  ue,green]);
```

$$eq := 2\,x-5\,y+4\,z=-3,\ x-2\,y+z=5,\ x-4\,y+5\,z=10$$



Les plans ne s'intersectent pas, il n'y a pas de solution.

<u>Exercice 11</u>

```
> A:=matrix(3,3,[1,1,m,1,1,-1,1,m,-1]); B:=vector([m,1,1]);
  C:=augment(A,B);
```

$$A := \begin{bmatrix} 1 & 1 & m \\ 1 & 1 & -1 \\ 1 & m & -1 \end{bmatrix}$$

$$B := [m, 1, 1]$$

$$C := \begin{bmatrix} 1 & 1 & m & m \\ 1 & 1 & -1 & 1 \\ 1 & m & -1 & 1 \end{bmatrix}$$

```
> gausselim(C);
```

$$\begin{bmatrix} 1 & 1 & m & m \\ 0 & m-1 & -1-m & 1-m \\ 0 & 0 & -1-m & 1-m \end{bmatrix}$$

Si $m=-1$, alors la dernière équation donne $0\,z=2$, donc le système n'a pas de solution.
Si $m=1$, le système est équivalent à $x+y+z=1$, $-2\,z=0$, $-2\,z=0$. Donc les solutions sont les (x, $-x$, 0) pour tout x dans R. C'est une droite de R^3, qui est contenue dans le plan $x+y+z=1$.

Si $m\neq -1$ et $m\neq 1$, alors la dernière équation donne $z=\frac{m-1}{m+1}$. Puis en reportant :

```
> subs(z=(m-1)/(m+1), (m-1)*y+(-1-m)*z=1-m);
```

$$(m-1)\,y+\frac{(-1-m)\,(m-1)}{m+1}=1-m$$

```
> solve(%,y);
```

$$0$$

Donc $y=0$. Et enfin :

```
> subs(z=(m-1)/(m+1),y=0,x+y+m*z=m);
```

$$x+\frac{m\,(m-1)}{m+1}=m$$

```
> solve(%,x);
```

$$\frac{2\,m}{m+1}$$

Donc $x=\frac{2\,m}{m+1}$. Pour $m\neq -1$ et $m\neq 1$, la solution est $\left[\frac{2\,m}{1+m}, 0, \frac{-1+m}{1+m}\right]$.
Résolution par *linsolve* :

```
> linsolve(A,B);
```

$$\left[\frac{2\,m}{m+1}, 0, \frac{m-1}{m+1}\right]$$

Attention, si $m=-1$, le résultat donné par Maple n'a pas de sens !!! De plus, on ne peut pas distinguer sur la solution de Maple le cas particulier $m=1$ qui conduit à une infinité de solutions.
Conclusion : faire très attention quand on utilise *linsolve* pour résoudre un système linéaire à paramètre.

Exercice 12

```
> A:=matrix(3,3,[3,2,-1,2,-3,4,1,1,-3]);
  kernel(A); # base du noyau
  n:=nops(%); # dimension du noyau
  colspace(A); # base de l'image
  m:=nops(%); # dimension de l'image
  m+n; # somme des deux dimensions
```

$$A:=\begin{bmatrix}3 & 2 & -1\\ 2 & -3 & 4\\ 1 & 1 & -3\end{bmatrix}$$

$$\{\ \}$$

$$n:=0$$

$$\{[1, 0, 0], [0, 1, 0], [0, 0, 1]\}$$

$$m:=3$$

$$3$$

(le noyau est {0})

```
> B:=matrix(3,3,[3,2,1,2,-3,5,1,1,0]);
  kernel(B); n:=nops(%);
  colspace(B); m:=nops(%);
  m+n;
```

$$B:=\begin{bmatrix}3 & 2 & 1\\ 2 & -3 & 5\\ 1 & 1 & 0\end{bmatrix}$$

$$\{[-1, 1, 1]\}$$

$$n:=1$$

$$\left\{\left[1, 0, \frac{5}{13}\right], \left[0, 1, \frac{-1}{13}\right]\right\}$$

$$m:=2$$

$$3$$

```
> C:=matrix(3,2,[1,1,2,2,4,5]);
  kernel(C); n:=nops(%);
  colspace(C); m:=nops(%);
  m+n;
```

$$C:=\begin{bmatrix}1 & 1\\ 2 & 2\\ 4 & 5\end{bmatrix}$$

$$\{\ \}$$

$$n:=0$$

$$\{[0, 0, 1], [1, 2, 0]\}$$

$$m:=2$$

$$2$$

On constate que la somme des dimensions est égale au nombre de colonnes de la matrice. C'est le "théorème du rang", ou "théorème noyau-image".
Remarque : la dimension de l'image, aussi appelée rang, peut etre calculé par la commande *rank*.

Exercice 13

```
> M:=matrix(3,3,[2,-2,1,2,-3,2,-1,2,0]);
```

$$M:=\begin{bmatrix}2 & -2 & 1\\ 2 & -3 & 2\\ -1 & 2 & 0\end{bmatrix}$$

```
> eigenvals(M);
```

$$-3, 1, 1$$

Deux valeurs propres : -3 (valeur propre simple) et 1 (valeur propre double).

```
> V:=eigenvects(M);
```

$$V:=[1, 2, \{[2, 1, 0], [-1, 0, 1]\}], [-3, 1, \{[-1, -2, 1]\}]$$

```
> v1:=vector([2,1,0]); # vecteur propre pour 1
  v2:=vector([-1,0,1]); # vecteur propre pour 1
  v3:=vector([-1,-2,1]); # vecteur propre pour -3
```

$$v1:=[2, 1, 0]$$

$$v2:=[-1, 0, 1]$$

$$v3:=[-1, -2, 1]$$

```
> basis({v1,v2,v3});
```

$$\{v1, v2, v3\}$$

La famille v_1, v_2, v_3 est donc libre dans R^3, et comme elle est de cardinal 3 = dimension de R^3, c'est une base de R^3.

La matrice de passage de la base canonique à la nouvelle base est formée de la manière suivante :

```
> P:=augment(v1,v2,v3);
```

$$P := \begin{bmatrix} 2 & -1 & -1 \\ 1 & 0 & -2 \\ 0 & 1 & 1 \end{bmatrix}$$

(chaque vecteur colonne est l'expression d'un vecteur dans la nouvelle base comme combinéaison linéaire des vecteurs de l'ancienne base)

Alors $P^{(-1)} M P$ vaut :

```
> evalm(inverse(P)&*M&*P);
```

$$\begin{bmatrix} 1 & 0 & 0 \\ 0 & 1 & 0 \\ 0 & 0 & -3 \end{bmatrix}$$

On obtient une matrice diagonale. C'est normal car, par définition des vecteurs propres, $M v_1 = v_1$, $M v_2 = v_2$ et $M v_3 = -3\ v_3$, ce qui donne la matrice de l'application linéaire dans la nouvelle base.

Nom :

Prénom :

Groupe :

MK1 "Calcul formel" Maple

Devoir sur table

Durée : 30 minutes

Documents autorisés : feuilles de TPs et corrections, résumés de cours personnels - Matériel interdit : livres, téléphones portables, documents informatiques,...

Rappel des commandes pour claviers Mac :
pour { : Alt (pour [: Alt Shift (
pour } : Alt)   pour] : Alt Shift)

Pour chaque exercice, donnez les commandes Maple qui permettent de le résoudre. Une attention particulière sera apportée à la syntaxe de Maple et à la rédaction !

Exercice 1 : Corriger les erreurs de syntaxe dans les commandes suivantes :

```
> f=x->exp(x)*lnx ;
> PLOT(f(x),x);
```

Exercice 2 : Construire à l'aide de la commande *seq* les séquences suivantes :

$$t-1, t^2-2, t^3-3, t^4-4, t^5-5$$

$$x+2\,y, 2\,x+3\,y, 3\,x+4\,y, 4\,x+5\,y, 5\,x+6\,y$$

Exercice 3 : Soient A, B, C les points du plan d'affixes respectives $a = \sqrt{3} + i$, $b = -1 - i\sqrt{3}$ et $c = -2$. Quelle est la nature du triangle ABC ? (donnez les commandes Maple et expliquez votre démarche)

Exercice 4 : On considère le polynome suivant : $x^3 + (1 - 3\,a)\,x^2 + (3\,a^2 - 2\,a)\,x + a^2 - a^3$

Commencer par définir la fonction P qui à x associe le polynome ci-dessus (il est inutile de recopier la commande sur la feuille).

1) Calculer les racines x de P.

2) En utilisant les commandes *expand* et *collect* (consultez l'aide au besoin), développer *P(x)* puis regrouper ses termes suivant les puissances de a.

3) On suppose désormais $a = -1$.

a) Calculer l'image de $1+\sqrt{3}$ par P et donner un résultat simplifié.

b) Calculer l'intégrale de P entre -1 et 1.

Correction du contrôle du 18 octobre

Exercice 1

```
> f:=x->exp(x)*ln(x);
```

$$f := x \to e^x \ln(x)$$

```
> plot(f(x),x);
```



Remarque : La commande *plot* a plusieurs syntaxes possibles. Avec la syntaxe ci-dessus, Maple choisit par défaut l'intervalle -10..10 en abscisse. On peut en imposer un en remplaçant x par x=a..b. On peut aussi imposer un intervalle en ordonnée par y=c..d Voir l'aide de la fonction *plot* pour plus d'informations.

Exercice 2

Il y a un exercice similaire dans la feuille d'exercices n°2. Il s'agit de déterminer la "formule" qui se cache derrière les séquences, puis utiliser la commande *seq* avec cette formule.

```
> seq(t^n-n,n=1..5);
```

$$t-1,\ t^2-2,\ t^3-3,\ t^4-4,\ t^5-5$$

```
> seq(n*x+(n+1)*y,n=1..5);
```

$$x+2\,y,\ 2\,x+3\,y,\ 3\,x+4\,y,\ 4\,x+5\,y,\ 5\,x+6\,y$$

Exercice 3

Commençons par définir les trois nombres complexes.

```
> restart ; a:=sqrt(3)+I ; b:=-1-I*sqrt(3) ; c:=-2;
```

$$a := \sqrt{3}+I$$
$$b := -1-\sqrt{3}\,I$$
$$c := -2$$

La longueur AB est égale au module |b-a| du nombre complexe b-a. De même, BC=|c-b| et CA=|a-c|. Calculons ces modules à l'aide de Maple.

```
> abs(b-a); abs(c-b); abs(a-c);
```

$$\sqrt{2}\,(1+\sqrt{3})$$
$$2$$
$$\sqrt{(\sqrt{3}+2)^2+1}$$

Simplifions pour voir si certains sont égaux.

```
> simplify(abs(b-a)) ; simplify(abs(c-b)) ; simplify(abs(a-c));
```

$$\sqrt{2}\,(1+\sqrt{3})$$
$$2$$
$$\sqrt{2}\,\sqrt{3}+\sqrt{2}$$

Donc les longueurs AB et AC valent $\sqrt{2}\,\sqrt{3}+\sqrt{2}$. Par contre, le troisième côté n'a pas la même longueur. Donc le triangle ABC est isocèle en A.

Remarque 1 : on pouvait également essayer de déterminer la nature du triangle en calculant ses angles.
En plaçant les points sur le plan, et en orientant les angles dans le sens trigonométrique, on voit que les angles des sommets du triangle sont : (BA,BC), (CB,CA) et (AC,AB). Leurs mesures sont données respectivement par :
argument((c-b)/(a-b)), argument ((a-c)/(b-c)) et argument ((b-a)/(c-a)).

```
> s:=[ argument((c-b)/(a-b)) , argument((a-c)/(b-c)) ,
  argument((b-a)/(c-a)) ];
```

$$s := \left[\arctan\left(\frac{1+\sqrt{3}}{\sqrt{3}-1}\right), \frac{5\,\pi}{12}, \arctan\left(\frac{\sqrt{3}+2}{2\,\sqrt{3}+3}\right)\right]$$

```
> simplify(s);
```

$$\left[\arctan\left(\frac{1+\sqrt{3}}{\sqrt{3}-1}\right), \frac{5\,\pi}{12}, \arctan\left(\frac{\sqrt{3}+2}{2\,\sqrt{3}+3}\right)\right]$$

On n'arrive pas à les simplifier avec la commande *simplify*. En fait, il faut d'abord simplifier la fraction à l'intérieur de l'arctan (en enlevant les racines carrées au dénominateur à l'aide de *rationalize*) . Par exemple, pour le premier angle :

```
> rationalize(1/(3^(1/2)-1)*(1+3^(1/2)));
```

$$\frac{(1+\sqrt{3})^2}{2}$$

```
> expand(%);
```

$$\sqrt{3}+2$$

puis appliquer l'arctan :

```
> arctan(%);
```

$$\frac{5\,\pi}{12}$$

Donc le premier angle vaut en fait 5*Pi/12. On fait la meme manipulation pour le troisième angle :

```
> rationalize(1/(2*3^(1/2)+3)*(3^(1/2)+2));
```

$$\frac{(\sqrt{3}+2)(-3+2\,\sqrt{3})}{3}$$

```
> expand(%);
  arctan(%);
```

$$\frac{\sqrt{3}}{3}$$
$$\frac{\pi}{6}$$

Le troisième angle vaut Pi/6. Les angles (BA,BC) et (CB,CA) sont égaux. Le triangle est isocèle en A.
Cette démonstration est plus délicate, notamment pour simplifier les expressions des angles obtenus. Elle n'était pas exigée, et il était plus facile de traiter l'exercice avec les modules !

Remarque 2 : certains se sont demandés si le triangle ABC était rectangle. Une façon de voir si c'est vrai, sans passer par les angles, est d'utiliser la réciproque du théorème de Pythagore.
Soient L1, L2 et L3 les longueurs des côtés.

```
> L1:=abs(b-a) ; L2:=abs(c-b) ; L3:=abs(a-c);
```

$$L1 := \sqrt{2}\,(1+\sqrt{3})$$
$$L2 := 2$$
$$L3 := \sqrt{(\sqrt{3}+2)^2+1}$$

Regardons si nous avons L1^2+L2^2=L3^2 ou bien L2^2+L3^2=L1^2 ou bien L1^2+L3^2=L2^2.

```
> simplify(L1^2+L2^2-L3^2);
  simplify(L2^2+L3^2-L1^2);
  simplify(L1^2+L3^2-L2^2);
```

$$4$$
$$4$$
$$12+8\sqrt{3}$$

Ces trois quantités sont non nulles, donc le triangle n'est pas rectangle.

Remarque 3 : Revenons aux longueurs des côtés :

```
> L:=[L1,L2,L3];
```

$$L := [\sqrt{2}\,(1+\sqrt{3}), 2, \sqrt{(\sqrt{3}+2)^2+1}\,]$$

On peut demander des valeurs *approchées* de ces longueurs (par défaut à 10 chiffres significatifs):

```
> evalf(L);
```

$$[3.863703305, 2., 3.863703305]$$

On voit que la première et la troisième ont la même valeur approchée. Cela peut donner une piste, mais **cela ne démontre en aucun cas** que le premier et le troisième côté ont même longueur. En effet, qu'est-ce qui vous dit que les 15èmes chiffres après la virgule ne vont pas etre différents ? Ce n'est pas la meme chose de faire du calcul approché et du calcul exact (dit aussi formel). Dans l'exercice, on vous demandait une démonstration, et non pas un calcul approché.

Exercice 4

```
> restart;
  P:=x->x^3+(1-3*a)*x^2+(3*a^2-2*a)*x+a^2-a^3;
```

$$P := x \to x^3+(1-3\,a)\,x^2+(3\,a^2-2\,a)\,x+a^2-a^3$$

Attention à bien définir une *fonction* comme demandé dans l'énoncé ! (par la flèche ou par *unapply*). Certains ont défini une expression en x, c'est-à-dire qu'ils ont tapé : P := x^3+(1-3*a)*x^2+....etc

```
> P;
  P(x);
```

$$P$$
$$x^3+(1-3\,a)\,x^2+(3\,a^2-2\,a)\,x+a^2-a^3$$

1) On demande de trouver les racines x, c'est-à-dire qu'on résout l'équation P(x)=0 par rapport à x (les solutions seront alors exprimées en fonction de a).

```
> solve(P(x)=0,x);
```

$$-1+a, a, a$$

La commande suivante ne répondait pas à la question : elle résout l'équation en fonction de x et de a.

```
> solve(P(x)=0);
```

$$\{x=x, a=x+1\}, \{x=x, a=x\}$$

2)

```
> expand(P(x));
```

$$x^3+x^2-3\,x^2\,a+3\,x\,a^2-2\,x\,a+a^2-a^3$$

```
> collect(P(x),a);
```

$$-a^3+(3\,x+1)\,a^2+(-3\,x^2-2\,x)\,a+x^3+x^2$$

3 a) Maintenant, on attribue à a la valeur -1.

```
> a:=-1; P(x);
```

$$a := -1$$
$$x^3+4\,x^2+5\,x+2$$

```
> P(1+sqrt(3));
```

$$(1+\sqrt{3})^3+4\,(1+\sqrt{3})^2+7+5\,\sqrt{3}$$

```
> simplify(%);
```

$$33+19\,\sqrt{3}$$

3b)

```
> int(P(x),x=-1..1);
```

$$\frac{20}{3}$$

Nom :

Prénom :

Groupe :

MK1 "Calcul formel" Maple

Devoir sur table

Durée : 30 minutes

Documents autorisés : feuilles de TPs et corrections, résumés de cours personnels - Matériel interdit : livres, téléphones portables, documents informatiques,...

Rappel des commandes pour claviers Mac :
pour { : Alt (pour [: Alt Shift (
pour } : Alt)   pour] : Alt Shift)

Pour chaque exercice, donnez les commandes Maple qui permettent de le résoudre. Une attention particulière sera apportée à la syntaxe de Maple et à la rédaction !

Exercice 1 : Corriger les erreurs de syntaxe dans les commandes suivantes :

```
> z:=1+isqrt3 ;
> argument(Z);
```

Exercice 2 : Construire à l'aide de la commande *seq* les séquences suivantes :

$$1, 3, 5, 7, 9, 11, 13, 15, 17$$

$$\frac{x}{y+1}, \frac{2\,x}{y+2}, \frac{3\,x}{y+3}, \frac{4\,x}{y+4}, \frac{5\,x}{y+5}$$

Exercice 3 : Soit $n = 262537412640768744$ et $a = e^{(\pi\sqrt{163})}$

1) Calculer a avec 30 chiffres significatifs.

2) Calculer une valeur approchée de $a - n$ (avec un nombre de chiffres significatifs adapté). Quelle conclusion pouvez-vous faire ?

Exercice 4 :

1) Définir la fonction f qui à x associe $x^3\,e^x$.

2) Calculer la dérivée seconde g de f.

3) Donner des valeurs approchées des racines de g.

4) Consulter l'aide sur *plot,options* rubrique *numpoints*. Quel est par défaut le nombre de points que calcule Maple pour tracer une courbe ?

5) Représenter graphiquement g en choisissant des intervalles adaptés pour x et y, et en fixant le nombre de points à 1000 (vous pourrez vous aider des exemples dans l'aide de *plot* pour la syntaxe).

Correction du contrôle du 21 octobre

Exercice 1

```
> z:=1+sqrt(3)*I;
```

$$z := 1+\sqrt{3}\,I$$

```
> argument(z);
```

$$\frac{\pi}{3}$$

Remarque : il existe une commande Maple qui se nomme *isqrt*. Dans la deuxième phrase à corriger, on calcule l'argument du nombre défini à la première ligne. Il me semble alors clair que la première ligne sert à définir un nombre complexe, et qu'il s'agissait de I*sqrt(3), et non pas de isqrt(3).

Exercice 2

Il y a un exercice similaire dans la feuille d'exercices n°2. Il s'agit de déterminer la "formule" qui se cache derrière les séquences, puis utiliser la commande *seq* avec cette formule.

```
> seq(2*n+1,n=0..8);
```

$$1, 3, 5, 7, 9, 11, 13, 15, 17$$

```
> seq(n*x/(y+n),n=1..5);
```

$$\frac{x}{y+1}, \frac{2\,x}{y+2}, \frac{3\,x}{y+3}, \frac{4\,x}{y+4}, \frac{5\,x}{y+5}$$

Exercice 3

1)

```
> restart;
> a:=exp(Pi*sqrt(163));
```

$$a := \mathbf{e}^{(\pi\sqrt{163})}$$

```
> evalf(a,30);
```

$$0.262537412640768744000000000024\ 10^{18}$$

2)

```
> n:=262537412640768744;
```

$$n := 262537412640768744$$

```
> evalf(a-n);
```

$$0.$$

En calculant une valeur approchée de a-n à 10 chiffres significatifs (c'est ce que fait par défaut *evalf*), on trouve 0. Il faut donc "pousser" plus loin.

```
> evalf(a-n,15);
```

$$0.$$

En continuant, par exemple avec 30, on trouve :

```
> evalf(a-n,30);
```

$$0.24\ 10^{-10}$$

Ceci montre que $|a-n| \le 10^{(-10)}$. Donc n est une approximation de a à 10^(-10) près.
Rappel. Soient deux nombres a et n. On dit que n est une approximation (ou valeur approchée) de a à ε près si on a $|a-n| \le \varepsilon$

Exercice 4

1) Attention à bien définir une *fonction* comme demandé (par exemple à l'aide de ->).

```
> restart;
  f:=x->x^3*exp(x);
```

$$f := x \to x^3\,\mathbf{e}^x$$

2) Comme f est une fonction, on s'attend à ce que g soit également une fonction.
On peut procéder en plusieurs étapes :

```
> h:=x->diff(f(x),x);
```

$$h := x \to \mathrm{diff}(\mathrm{f}(x), x)$$

```
> g:=x->diff(h(x),x);
```

$$g := x \to \mathrm{diff}(\mathrm{h}(x), x)$$

```
> g(x);
```

$$6\,x\,\mathbf{e}^x + 6\,x^2\,\mathbf{e}^x + x^3\,\mathbf{e}^x$$

ou en une seule :

```
> g:=x->diff(diff(f(x),x),x);
```

$$g := x \to \mathrm{diff}(\mathrm{diff}(\mathrm{f}(x), x), x)$$

```
> g(x);
```

$$6\,x\,\mathbf{e}^x + 6\,x^2\,\mathbf{e}^x + x^3\,\mathbf{e}^x$$

ou encore avec D, au choix !

```
> g:=D(D(f));
```

$$g := x \to 6\,x\,\mathbf{e}^x + 6\,x^2\,\mathbf{e}^x + x^3\,\mathbf{e}^x$$

```
> g(x);
```

$$6\,x\,\mathbf{e}^x + 6\,x^2\,\mathbf{e}^x + x^3\,\mathbf{e}^x$$

3) Pour calculer les valeurs approchées des racines :

```
> solve(g(x)=0,x);
```

$$0, -3+\sqrt{3}, -3-\sqrt{3}$$

```
> evalf(%);
```

$$0., -1.267949192, -4.732050808$$

Attention, si vous aviez défini g à la question 2 comme une expression en x (et non pas comme une fonction), la syntaxe pour *solve* n'est pas la même. Exemple :

```
> g:=diff(diff(f(x),x),x);
```

$$g := 6\,x\,\mathbf{e}^x + 6\,x^2\,\mathbf{e}^x + x^3\,\mathbf{e}^x$$

```
> solve(g(x),x);
```

$$\mathbf{proc}(x)\ 0\ \mathbf{end\ proc},\ \mathrm{RootOf}((\mathbf{e}^{-Z})(\_Z)),\ \mathbf{proc}(x)\ \mathrm{RootOf}(6+6*\_Z+\_Z\text{^}2)\ \mathbf{end\ proc}$$

Cela ne fonctionne pas parce que g(x) n'a pas de sens ! La preuve :

```
> g(x);
```

$$6\,\mathrm{x}(x)\,(\mathbf{e}^x)(x) + 6\,\mathrm{x}(x)^2\,(\mathbf{e}^x)(x) + \mathrm{x}(x)^3\,(\mathbf{e}^x)(x)$$

Il fallait alors utiliser *solve* comme ceci :

```
> solve(g,x);
```

$$0, -3+\sqrt{3}, -3-\sqrt{3}$$

Conclusion : attention à bien faire la différence entre **fonctions** et **expressions. Elles ne se définissent pas et ne s'utilisent pas de la même façon.**

Enfin, pour résoudre la question, on pouvait essayer *fsolve*. Malheureusement, on ne trouve qu'une racine :

```
> g:=D(D(f));
  fsolve(g(x)=0,x=-100..100);
```

$$g := x \to 6\,x\,\mathbf{e}^x + 6\,x^2\,\mathbf{e}^x + x^3\,\mathbf{e}^x$$

0.

En consultant l'aide, on apprend que pour une équation générale (non polynomiale), *fsolve* ne calcule qu'une racine réelle. Si on a déterminé les valeurs exactes des racines avec *solve*, on pouvait s'en sortir en choisissant des intervalles bien adaptés pour "attraper" les deux autres racines :

```
> fsolve(g(x)=0,x=-5..-2);
```

-4.732050808

```
> fsolve(g(x)=0,x=-2..-1);
```

-1.267949192

La solution recommandée était celle avec *solve*, puis *evalf*.

4) Pour afficher l'aide en question :

```
> ?plot,options
```

Cela amène sur la page spéciale consacrée aux options de la commande *plot* (elles sont très nombreuses). Au paragraphe consacré à l'option *numpoints*, on y lit : "Specifies the minimum number of points to be generated (the default is n = 50)". Donc Maple utilise 50 points par défaut pour tracer une courbe.

5) Comme intervalle adapté en abcisse, on choisit un intervalle faisant apparaître les racines de g, par exemple -10..5

```
> plot(g(x),x=-10..5);
```



Du coup, il faut choisir un intervalle en y adapté, par exemple :

```
> plot(g(x),x=-10..5,y=-3..5);
```



Enfin, on demande à Maple de tracer la courbe avec 1000 points :

```
> plot(g(x),x=-10..5,y=-3..5,numpoints=1000);
```



(Remarquer que la courbe obtenue est plus "lisse" que la précédente)

Devoir à la maison

Une copie, de préférence manuscrite, doit être remise par personne. Elle devra comporter les commandes Maple correspondant aux exercices (il n'est pas nécessaire d'indiquer les réponses de Maple). Dans l'exercice 3, les questions 1a et 2a sont uniquement mathématiques. Tout travail remis en retard aura une note réduite. Vous pouvez travailler à deux ou trois, mais chacun doit remettre une copie individuelle.

Les trois exercices sont indépendants. Les deux premiers peuvent être traités sans le logiciel (avec papier, crayon et documents de cours). Maple est disponible dans la salle A du Script, barre 65-66, deuxième étage. La salle n'est pas disponible tout le temps, aussi renseignez-vous sur ses horaires en libre-service !

Exercice 1. On considère le polynôme $P = z^{10} + z^2 + 1$ et ses racines complexes. Donner les commandes Maple qui permettent d'afficher la liste des valeurs approchées des modules de ces racines. Même question avec la liste des valeurs approchées des arguments.

Exercice 2. Ecrire une procédure qui prend en entrée un polynôme P de degré 2 et qui affiche en sortie ses racines réelles éventuelles.

Instructions à suivre : la procédure devra vérifier que P est de degré 2 et afficher un message d'erreur le cas échéant. Elle devra traiter tous les cas et ne devra pas utiliser les commandes de résolution d'équation comme *solve*, *fsolve*, ...

Commandes utiles : *degree*(P, X) donne le degré en X de P
coeff(P, X, k) donne le coefficient de X^k dans P

Exercice 3. Approximation d'intégrale

Soit f une fonction continue positive de $[a, b]$ dans $\mathbb{R}$. Le but de l'exercice est d'approcher l'intégrale $J = \int_a^b f(x)dx$ par deux méthodes classiques.

1) **Méthode des rectangles**

   On découpe $[a, b]$ en n intervalles égaux. On approche l'intégrale de f sur chacun de ces intervalles par l'aire du rectangle ayant pour base cet intervalle et pour hauteur la valeur de f en l'extrémité gauche de l'intervalle.

   (a) En vous aidant d'un dessin, exprimer l'aire totale $Jrect(f, a, b, n)$ calculée ainsi entre a et b, en fonction de f, a, b et n.

   (b) Dans Maple, écrire une fonction ou une procédure *rectangle*, qui prend en entrée f, a, b et n et calcule $Jrect(f, a, b, n)$. On utilisera la commande *sum* plutôt que *add*.

Pour les trois questions suivantes, on prendra $f(x) = x^2 \ln(x+1)$, $a = 0$, $b = 5$.

(c) Tester *rectangle* sur cet exemple en prenant différentes valeurs pour n. Comparer les valeurs approchées de vos résultats à ceux de la commande Maple qui calcule la méthode des rectangles : commande *leftsum* de la librairie *student*.

(d) Vérifier que

$$\lim_{n\to\infty} Jrect(f, a, b, n) = J.$$

(e) Expliquer pourquoi il existe un entier m tel que :

$$|J - Jrect(f, a, b, m)| \leq 0.3$$

A l'aide d'une boucle *while*, trouver le plus petit entier m vérifiant cette propriété.

2) **Méthode des trapèzes**

On découpe $[a, b]$ en n intervalles égaux. On approche l'intégrale de f sur chacun de ces intervalles par l'aire du trapèze ayant pour sommets les extrémités de l'intervalle et les images par f de ces extrémités.

(a) En vous aidant d'un dessin, exprimer l'aire totale $Jtrap(f, a, b, n)$ calculée ainsi entre a et b, en fonction de f, a, b et n. On rappelle que l'aire d'un trapèze de bases B_1, B_2 et de hauteur h est donnée par $(B_1 + B_2)h/2$.

(b) Dans Maple, écrire une fonction ou une procédure *trapeze*, qui prend en entrée f, a, b et n et calcule $Jtrap(f, a, b, n)$. On utilisera la commande *sum* plutôt que *add*.

Pour les trois questions suivantes, on prendra $f(x) = x^2 \ln(x+1)$, $a = 0$, $b = 5$.

(c) Tester *trapeze* sur cet exemple en prenant différentes valeurs pour n. Comparer les valeurs approchées de vos résultats à ceux de la commande Maple qui calcule la méthode des trapèzes : commande *trapezoid* de la librairie *student*.

(d) Vérifier que

$$\lim_{n\to\infty} Jtrap(f, a, b, n) = J.$$

(e) Expliquer pourquoi il existe un entier p tel que :

$$|J - Jtrap(f, a, b, p)| \leq 0.3$$

A l'aide d'une boucle *while*, trouver le plus petit entier p vérifiant cette propriété. Comparer ce résultat à celui obtenu à la question 1.e . Qu'en pensez-vous ?

Correction du devoir à la maison

Exercice 1 : une solution possible

```
> restart;
> P:=z^10+z^2+1;
```

$$P := z^{10} + z^2 + 1$$

```
> S:=[solve(P,z)];
```

$$S := \left[-\frac{1}{2}+\frac{1}{2} I\sqrt{3}, -\frac{1}{2}-\frac{1}{2} I\sqrt{3}, \frac{1}{2}+\frac{1}{2} I\sqrt{3}, \frac{1}{2}-\frac{1}{2} I\sqrt{3}, \frac{\frac{1}{6} I\sqrt{6}\sqrt{(100+12\sqrt{69})^{(1/3)}((100+12\sqrt{69})^{(2/3)}+4-2(100+12\sqrt{69})^{(1/3)})}}{(100+12\sqrt{69})^{(1/3)}}, \frac{\frac{-1}{6} I\sqrt{6}\sqrt{(100+12\sqrt{69})^{(1/3)}((100+12\sqrt{69})^{(2/3)}+4-2(100+12\sqrt{69})^{(1/3)})}}{(100+12\sqrt{69})^{(1/3)}}, (-3 (100+12\sqrt{69})^{(1/3)} (-(100+12\sqrt{69})^{(2/3)}-4-4(100+12\sqrt{69})^{(1/3)}+\sqrt{3}(100+12\sqrt{69})^{(2/3)} I - 4 I\sqrt{3}))^{(1/2)} \Big/ (6(100+12\sqrt{69})^{(1/3)}), -(-3(100+12\sqrt{69})^{(1/3)} (-(100+12\sqrt{69})^{(2/3)}-4-4(100+12\sqrt{69})^{(1/3)}+\sqrt{3}(100+12\sqrt{69})^{(2/3)} I - 4 I\sqrt{3}))^{(1/2)} \Big/ (6(100+12\sqrt{69})^{(1/3)}), \sqrt{3}((100+12\sqrt{69})^{(1/3)} ((100+12\sqrt{69})^{(2/3)}+4+4(100+12\sqrt{69})^{(1/3)}+\sqrt{3}(100+12\sqrt{69})^{(2/3)} I - 4 I\sqrt{3}))^{(1/2)} \Big/ (6(100+12\sqrt{69})^{(1/3)}), -\sqrt{3}((100+12\sqrt{69})^{(1/3)} ((100+12\sqrt{69})^{(2/3)}+4+4(100+12\sqrt{69})^{(1/3)}+\sqrt{3}(100+12\sqrt{69})^{(2/3)} I - 4 I\sqrt{3}))^{(1/2)} \Big/ (6(100+12\sqrt{69})^{(1/3)}) \right]$$

Liste des valeurs approchées des modules des racines :

```
> [ seq( evalf(abs(S[i])) ,i=1..nops(S) ) ];
```

[1., 1., 1., 1., 0.8688369618, 0.8688369618, 1.072829869, 1.072829869, 1.072829869, 1.072829869]

Liste des valeurs approchées des arguments des racines :

```
> [ seq( evalf(argument(S[i])) ,i=1..nops(S) ) ];
```

[2.094395103, -2.094395103, 1.047197551, -1.047197551, 1.570796327, -1.570796327, -0.3519288607, 2.789663793, 0.3519288608, -2.789663793]

Attention, on demandait comme résultats des **listes** (et non pas des séquences, ou un affichage en colonne comme ce qu'on peut obtenir avec une boucle *for*).

Exercice 2 : une solution possible

```
> restart;
> racines:=proc(P)
  local d,a,b,c,delta;
  d:=degree(P,X);
  if d<>2 then RETURN("Erreur")
  else a:=coeff(P,X,2);
       b:=coeff(P,X,1);
       c:=coeff(P,X,0);
       delta:=b^2-4*a*c;
       if delta<0 then RETURN("Pas de racines réelles")
       elif delta=0 then RETURN(-b/(2*a))
       else
  RETURN((-b-sqrt(delta))/(2*a),(-b+sqrt(delta))/(2*a));
       fi;
  fi;
  end:
> racines(X^3+4);
```

"Erreur"

```
> racines(X^2+1);
```

"Pas de racines réelles"

```
> racines(X^2-2*X+1);
```

$$1$$

```
> racines(X^2+5*X+2);
```

$$-\frac{5}{2}-\frac{\sqrt{17}}{2}, -\frac{5}{2}+\frac{\sqrt{17}}{2}$$

Plusieurs remarques :
- bien penser à traiter tous les cas ($0 < \Delta$, $\Delta = 0$, $\Delta < 0$).
- attention à la règle de priorité entre la multiplication et la division :

```
> a:=3; b:=7;
  -b/2*a;
  -b/(2*a);
```

$$a := 3$$
$$b := 7$$
$$\frac{-21}{2}$$
$$\frac{-7}{6}$$

Dans le doute, mieux vaut mettre des parenthèses, même si elles sont superflues.
- enfin, notez bien que le test "si degré(P)<>2 alors erreur sinon suite d'instructions qui calculent les racines" se termine en fin de programme (*fi*). Il vaut penser que si degré(P)<>2, le programme n'est pas censé effectuer les instructions qui calculent les racines !

Exercice 3

1a) Un calcul et un dessin (que je ne reproduirai pas ici mais qui étaient exigés) montrent que :

$$\text{Jrect}(f, a, b, n) = \sum_{k=0}^{n-1} \frac{(b-a)\, f\left(a + \frac{k(b-a)}{n}\right)}{n}.$$

1b) Voici comment on peut écrire *rectangle* comme une fonction de plusieurs variables, en utilisant la commande *sum.*

```
> restart;
  rectangle:=(f,a,b,n)->sum((b-a)/n * f(a+k*(b-a)/n) , k=0..n-1);
```

$$rectangle := (f, a, b, n) \to \sum_{k=0}^{n-1} \frac{(b-a)\, f\left(a + \frac{k\,(b-a)}{n}\right)}{n}$$

Un exemple :

```
> rectangle(exp,0,5,15);
  evalf(%);
```

$$\frac{1}{3}+\frac{1}{3}\mathbf{e}^{(1/3)}+\frac{1}{3}\mathbf{e}^{(2/3)}+\frac{1}{3}\mathbf{e}+\frac{1}{3}\mathbf{e}^{(4/3)}+\frac{1}{3}\mathbf{e}^{(5/3)}+\frac{1}{3}\mathbf{e}^{2}+\frac{1}{3}\mathbf{e}^{(7/3)}+\frac{1}{3}\mathbf{e}^{(8/3)}+\frac{1}{3}\mathbf{e}^{3}+\frac{1}{3}\mathbf{e}^{(10/3)}$$
$$+\frac{1}{3}\mathbf{e}^{(11/3)}+\frac{1}{3}\mathbf{e}^{4}+\frac{1}{3}\mathbf{e}^{(13/3)}+\frac{1}{3}\mathbf{e}^{(14/3)}$$

124.2067149

1c)

```
> f:=x->x^2*ln(x+1);
  a:=0;
  b:=5;
```

$$f := x \to x^2 \ln(x+1)$$
$$a := 0$$
$$b := 5$$

Remarque : on rappelle qu'une *fonction* se définit dans Maple par la flèche (par exemple) : $x \to x^2 \ln(x+1)$. Si on souhaite lui donner un nom (c'est-à-dire la "stocker" dans une variable), on fait une affectation par $:=$. Par exemple, si on veut l'appeler *f*, on fait $f := x\text{-}> x^2 \ln(x+1)$.

```
> evalf(rectangle(f,a,b,10));
  evalf(rectangle(f,a,b,50));
  evalf(rectangle(f,a,b,100));
```

53.12706907
61.64371382
62.74976012

```
> ?student
> with(student);
```

[D, *Diff*, *Doubleint*, *Int*, *Limit*, *Lineint*, *Product*, *Sum*, *Tripleint*, *changevar*, *completesquare*, *distance*, *equate*, *integrand*, *intercept*, *intparts*, *leftbox*, *leftsum*, *makeproc*, *middlebox*, *middlesum*, *midpoint*, *powsubs*, *rightbox*, *rightsum*, *showtangent*, *simpson*, *slope*, *summand*, *trapezoid*]

```
> evalf(leftsum(f(x),x=a..b,10));
  evalf(leftsum(f(x),x=a..b,50));
  evalf(leftsum(f(x),x=a..b,100));
```

53.12706905
61.64371381
62.74976010

On obtient quasiment les mêmes valeurs (ouf !). Les petits écarts sont dûs à des erreurs d'approximations.

Pour info, l'intégrale $\int_a^b f(x)\,dx$ vaut (en valeur approchée) :

```
> evalf(int(f(x),x=a..b));
```

63.86500884

1d)

```
> limit(rectangle(f,a,b,n),n=infinity) - int(f(x),x=a..b);
```

$$-\frac{1}{3}\ln(1)+42\ln(6)-42\ln(2)-42\ln(3)$$

```
> simplify(%);
```

0

Remarque. On demande de montrer que la limite vaut *J*. On ne demande pas de montrer que la limite et *J* ont mêmes valeurs approchées (ce n'est pas suffisant). C'est pour cela qu'on garde des expressions **exactes** pour la limite et pour *J*, et qu'on n'utilise pas la commande *evalf*.

1e) Par définition de la limite, pour tout ε >0, il existe *N* tel que pour tout *n*>=*N* , | Jrect(f,a,b,n) - $\int_a^b f(x)\,dx$ | <= ε. Pour ε = 0.3, il existe *N* tel que pour tout *n* >= *N*, | Jrect(f,a,b,n) - $\int_a^b f(x)\,dx$ | <= 0.3. Il existe donc (au moins un) entier *m* tel que | Jrect(f,a,b,m) - $\int_a^b f(x)\,dx$ | <= 0.3

Pour trouver ce plus petit entier *m*, on procède avec une boucle *while*.

```
> J:=int(f(x),x=a..b):
> m:=1:
  while abs(evalf(J-rectangle(f,a,b,m)))>0.3 do m:=m+1 od:
> m;
```

373

La réponse est donc m=373.

Remarques :

- le contraire de | Jrect(f,a,b,m) - $\int_a^b f(x)\,dx$ | <= 0.3 est | Jrect(f,a,b,m) - $\int_a^b f(x)\,dx$ | >0.3 (et non pas >=). C'est cette condition qui doit figurer dans le *while*.

- pour que l'expression booléenne | Jrect(f,a,b,m) - $\int_a^b f(x)\,dx$ | >0.3 puisse être évaluée, on doit avoir recours à une expression approchée par *evalf*.

- enfin, je rappelle une dernière fois que la "virgule" des nombres décimaux (notation française) est un point dans Maple (notation anglo-saxone).

2a) Un calcul et un dessin (que je ne reproduirai pas ici mais qui étaient exigés) montrent que :

$$\text{Jtrap}(f, a, b, n) = \sum_{k=0}^{n-1} \frac{(b-a)\left(f\left(a+\frac{k\,(b-a)}{n}\right)+f\left(a+\frac{(k+1)\,(b-a)}{n}\right)\right)}{2\,n}.$$

2b) Comme une procédure (pour changer) :

```
> trapeze:=proc(f,a,b,n)
  local h;
  h:=(b-a)/n;
  RETURN(sum(h/2 * (f(a+k*h)+f(a+(k+1)*h)), k=0..n-1) );
  end:
```

```
> trapeze(exp,0,5,15);
  evalf(%);
```

$$\frac{1}{3}\mathbf{e}^{(2/3)}+\frac{1}{3}\mathbf{e}^{(1/3)}+\frac{1}{3}\mathbf{e}^{(4/3)}+\frac{1}{3}\mathbf{e}^{2}+\frac{1}{3}\mathbf{e}^{(5/3)}+\frac{1}{3}\mathbf{e}^{(8/3)}+\frac{1}{3}\mathbf{e}^{(7/3)}+\frac{1}{3}\mathbf{e}^{3}+\frac{1}{3}\mathbf{e}^{(10/3)}+\frac{1}{3}\mathbf{e}^{(13/3)}$$
$$+\frac{1}{3}\mathbf{e}^{4}+\frac{1}{3}\mathbf{e}^{(11/3)}+\frac{1}{3}\mathbf{e}^{(14/3)}+\frac{1}{6}\mathbf{e}^{5}+\frac{1}{6}+\frac{1}{3}\mathbf{e}$$

$$148.7755747$$

2c)

```
> f(x) ; a ; b ;
```

$$x^2\ln(x+1)$$
$$0$$
$$5$$

```
> evalf(trapeze(f,a,b,10));
  evalf(trapeze(f,a,b,50));
  evalf(trapeze(f,a,b,100));
```

$$64.32556575$$
$$63.88341317$$
$$63.86960978$$

```
> evalf( trapezoid(f(x),x=a..b,10) );
  evalf( trapezoid(f(x),x=a..b,50) );
  evalf( trapezoid(f(x),x=a..b,100) );
```

$$64.32556573$$
$$63.88341315$$
$$63.86960977$$

On obtient quasiment les mêmes valeurs (ouf !). Les petits écarts sont dûs à des erreurs d'approximations.

2d) Mêmes remarques qu'en 1d)

```
> limit(trapeze(f,a,b,n),n=infinity)-int(f(x),x=a..b);
```

$$42\ln(6)-42\ln(2)-42\ln(3)$$

```
> simplify(%);
```

$$0$$

2e) Le même raisonnement qu'à la question 1e donne l'existence de p.

```
> p:=1:
  while abs(evalf(J-trapeze(f,a,b,p)))>0.3 do p:=p+1 od:
> p;
```

$$13$$

La réponse est *p=13*. On remarque que ce nombre est bien meilleur que celui obtenu par la méthode des rectangles. Cela signifie que pour cette fonction f, pour obtenir une erreur <0.3 sur le calcul de l'intégrale, on doit utiliser moins de subdivisions n avec les trapèzes qu'avec les rectangles. Cela laisse penser que quand n-> ∞, la méthode des trapèzes converge plus vite vers l'intégrale $\int_a^b f(x)\,dx$ que la méthode des rectangles.

Nom :

Prénom :

Groupe :

MK1 "Calcul formel" Maple

Partiel

Durée : 55 minutes

Documents autorisés : feuilles de TPs et corrections, résumés de cours personnels - Matériel interdit : livres, téléphones portables, documents informatiques,...

Rappel des commandes pour claviers Mac :
pour { : Alt (pour [: Alt Shift (
pour } : Alt) pour] : Alt Shift)

Une attention particulière sera apportée à la syntaxe de Maple et à la rédaction !

Pour les exercices 1 et 2, donnez uniquement les *commandes* Maple qui permettent de répondre aux questions.

Exercice 1

Commencer par charger la librairie *numtheory* par la commande suivante :

```
> with(numtheory);
```

1) En consultant l'aide sur la librairie *numtheory*, trouver la commande qui permet de calculer les diviseurs d'un nombre entier et l'utiliser pour déterminer les diviseurs de $n = 478965400$.

2) Compter combien n a de diviseurs (en utilisant Maple).

3) Déterminer les diviseurs communs à n et 220925.

Exercice 2

En utilisant des boucles, trouver les solutions entières $[x, y]$ de l'équation $x^2 - y^2 = 3640$ avec x compris entre 0 et 200 et y compris entre 0 et 200.
(*print* pour faire afficher les solutions)

L'exercice 3 est à résoudre avec Maple. Lorsque vous effectuez des calculs qui permettent de répondre aux questions, donnez les *commandes* Maple correspondantes. Pour certaines questions, on demande des *explications* ou des *raisonnements mathématiques* . La rédaction sera prise en compte.

Exercice 3

Soit f la fonction qui à x (réel) associe $x - \frac{2\,\mathbf{e}^x}{\mathbf{e}^x - 1}$. Soit C la courbe représentative de f.

1) Définir f.

Expliquer mathématiquement quel est l'ensemble de définition Df de f.

Vérifier à l'aide de Maple que f est continue sur Df.

2) Calculer les limites de f aux bornes des intervalles qui composent Df.

La courbe C admet-elle une asymptote ? Si oui, précisez laquelle.

3) Démontrer que la courbe C admet pour asymptote en ∞ la droite Δ d'équation $y = x - 2$.

Etudier la position de C par rapport à cette asymptote.

4) Calculer la fonction dérivée de f.

Etudier son signe.

En déduire les variations de f.

5) Tracer sur un même graphe la courbe C et la droite Δ en choisissant de bons intervalles d'affichage en abscisse et ordonnée.

6) Sans utiliser *solve* ni *fsolve*, démontrer que f a un unique zéro compris entre -2 et -1.

7) En utilisant *fsolve*, donner une valeur approchée de ce zéro.

Correction du partiel du 8 novembre

Remarque générale : dans les copies, un certain nombre d'erreurs sont dûes à une mauvaise lecture de l'énoncé. Lisez attentivement les instructions et les questions avant d'y répondre ! Si on demande de calculer la *fonction* dérivée, donnez une fonction et non pas une expression. Dans l'exercice sur l'étude de fonction, tous les calculs qui peuvent etre faits *doivent etre faits avec Maple*, et vous devez donner les commandes correspondantes !

Enfin, attention pour ceux qui font encore l'erreur : l'exponentielle $\mathbf{e}^x$ s'obtient dans Maple en tapant `exp(x)` et non pas `e^x` !!!

Exercice 1

1)

```
> restart;
  with(numtheory);
Warning, the protected name order has been redefined and unprotected
```

[*GIgcd*, *bigomega*, *cfrac*, *cfracpol*, *cyclotomic*, *divisors*, *factorEQ*, *factorset*, *fermat*, *imagunit*, *index*, *integral_basis*, *invcfrac*, *invphi*, *issqrfree*, *jacobi*, *kronecker*, λ, *legendre*, *mcombine*, *mersenne*, *migcdex*, *minkowski*, *mipolys*, *mlog*, *mobius*, *mroot*, *msqrt*, *nearestp*, *nthconver*, *nthdenom*, *nthnumer*, *nthpow*, *order*, *pdexpand*, φ, π, *pprimroot*, *primroot*, *quadres*, *rootsunity*, *safeprime*, σ, *sq2factor*, *sum2sqr*, τ, *thue*]

```
> ?numtheory
```

La commande de la librairie *numtheory* qui permet de calculer les diviseurs d'un nombre entier est *divisors* (pour consulter son aide, taper *?numtheory[divisors]*)

```
> n:=478965400;
```

$$n := 478965400$$

```
> divisors(n);
```

{1, 2, 4, 5, 8, 10, 20, 25, 40, 50, 100, 200, 271, 542, 1084, 1355, 2168, 2710, 5420, 6775, 8837, 10840, 13550, 17674, 27100, 35348, 44185, 54200, 70696, 88370, 176740, 220925, 353480, 441850, 883700, 1767400, 2394827, 4789654, 9579308, 11974135, 19158616, 23948270, 47896540, 59870675, 95793080, 119741350, 239482700, 478965400}

2) On remarque que la commande précédente retourne un *ensemble* de diviseurs. Pour calculer le nombre d'éléments d'un ensemble (ou d'une liste), on dispose de la commande *nops.*

```
> nops(%);
```

$$48$$

Certains ont remarqué, à juste titre, que la commande *tau* de la librairie permet de calculer le nombre de diviseurs :

```
> tau(n);
```

$$48$$

3) Il suffit de faire l'intersection de l'ensemble des diviseurs de *n* avec l'ensemble des diviseurs de 220925.

```
> divisors(n) intersect divisors(220925);
```

$$\{1, 5, 25, 8837, 44185, 220925\}$$

Exercice 2

Pour x entre 0 et 200, et y entre 0 et 200, on teste si l'équation est vérifiée (par *if*). Si c'est le cas, alors on demande d'afficher la solution [*x*, *y*] à l'aide de *print*.

```
> restart;
> for x from 0 to 200 do
      for y from 0 to 200 do
          if x^2-y^2=3640 then print([x,y]) fi;
      od;
  od;
```

$$[61, 9]$$
$$[79, 51]$$
$$[83, 57]$$
$$[101, 81]$$
$$[137, 123]$$
$$[187, 177]$$

Remarque : *RETURN* sert à afficher un résultat dans une procédure. En dehors d'une procédure, pour des boucles *for*, on utilise *print* pour afficher un résultat.

Exercice 3

1) On nous demande de définir une <u>*fonction*</u> dans Maple (x -> ...). Attention à bien employer exp(x) et non pas e^x pour l'exponentielle.

```
> restart;
> f:=x->x-2*exp(x)/(exp(x)-1);
```

$$f := x \to x - \frac{2\,\mathbf{e}^x}{\mathbf{e}^x - 1}$$

La fonction *f* est définie pour tout *x*, sauf pour ceux tels que $\mathbf{e}^x - 1 = 0$, c'est-à-dire pour $x = 0$.
L'ensemble de définition de *f* est donc : Df =]-∞,0[U]0,+∞[
On vérifie que *f* est continue sur les deux intervalles qui composent Df.

```
> iscont(f(x),x=-infinity..0);
```

$$\mathit{true}$$

```
> iscont(f(x),x=0..infinity);
```

$$\mathit{true}$$

2)

```
> limit(f(x),x=-infinity);
```

$$-\infty$$

```
> limit(f(x),x=infinity);
```

$$\infty$$

```
> limit(f(x),x=0);
```

$$\mathit{undefined}$$

```
> limit(f(x),x=0,left);
```

$$\infty$$

```
> limit(f(x),x=0, right);
```

$$-\infty$$

A cause des deux dernières limites, la courbe C admet une asymptote verticale d'équation $x = 0$.

3) On vérifie que $x - 2$ est asymptote en +∞ en montrant que la limite de $f(x) - (x - 2)$ quand x tend vers ∞ est 0.

```
> limit(f(x)-(x-2),x=infinity);
```

$$0$$

Donc la droite d'équation $y = x - 2$ est asymptote.
Position de la courbe par rapport à la droite asymptote :

```
> solve(f(x)-(x-2)>0,x);
```

$$\mathrm{RealRange}(-\infty, \mathrm{Open}(0))$$

```
> solve(f(x)-(x-2)<0,x);
```

$$\mathrm{RealRange}(\mathrm{Open}(0), \infty)$$

Donc pour x > 0, la courbe est au-dessous de la droite, et pour x < 0, la courbe est au-dessus de la droite.

4) On calcule la *fonction* dérivée par :

```
> g:=D(f);
```

$$g := x \to 1 - \frac{2\,\mathbf{e}^x}{\mathbf{e}^x - 1} + \frac{2\,(\mathbf{e}^x)^2}{(\mathbf{e}^x - 1)^2}$$

```
> g(x);
```

$$1 - \frac{2\,\mathbf{e}^x}{\mathbf{e}^x - 1} + \frac{2\,(\mathbf{e}^x)^2}{(\mathbf{e}^x - 1)^2}$$

ou bien par :

```
> g:=x->diff(f(x),x);
```

$$g := x \to \mathrm{diff}(\mathrm{f}(x), x)$$

```
> g(x);
```

$$1 - \frac{2\,\mathbf{e}^x}{\mathbf{e}^x - 1} + \frac{2\,(\mathbf{e}^x)^2}{(\mathbf{e}^x - 1)^2}$$

Etude de son signe.

```
> normal(g(x));
```

$$\frac{(\mathbf{e}^x)^2 + 1}{(\mathbf{e}^x - 1)^2}$$

On voit que le numérateur et le dénominateur sont strictement positifs pour $x \neq 0$. On peut le vérifier :

```
> solve(g(x)>0,x);
```

$$\mathrm{RealRange}(-\infty, \mathrm{Open}(0)), \mathrm{RealRange}(\mathrm{Open}(0), \infty)$$

On en déduit que *f* est strictement croissante sur]-∞,0[U]0,+∞[.

5)

```
> plot([f(x),x-2],x);
```



On recadre en ordonnée, par exemple comme ceci, pour faire apparaitre la courbe et son asymptote.

```
> plot([f(x),x-2],x,y=-10..10);
```



Remarque : comme je l'ai dit lors de la correction du controle, évitez d'employer *infinity* comme borne de l'intervalle d'affichage dans *plot*. Cela donne des graphes qui sont "déformés" au voisinage de l'infini. Essayez `plot(sin(x),x=-infinity..infinity);` pour vous en convaincre.

6)

```
> f(2) ; f(-1) ;
```

$$2-\frac{2\,\mathbf{e}^2}{\mathbf{e}^2-1}$$

$$-1-\frac{2\,\mathbf{e}^{(-1)}}{\mathbf{e}^{(-1)}-1}$$

```
> evalf(f(-2)); evalf(f(-1));
```

$$-1.686964715$$

$$0.163953414$$

On a $f(-2)<0$ et $f(-1)>0$. De plus, *f* est continue et strictement croissante sur I=[-2,-1] donc c'est une bijection de I sur J=f([-2,-1]). Comme 0 est dans J, il existe un unique $a \in I$ tel que $f(a)=0$.

7)

```
> fsolve(f(x)=0,x);
```

$$2.238645835$$

Attention ! *fsolve* nous donne une valeur approchée, mais qui n'est pas entre -2 et -1. En regardant le graphique précédent, on voit que f a deux zéros: a et un autre qui est situé entre 2 et 4. *fsolve* nous a donné par défaut une valeur approchée de cet autre zéro. Pour obtenir une valeur approchée de a, on demande à *fsolve* de chercher une solution sur [-2,-1] (consultez l'aide sur fsolve si vous ne vous souvenez plus de la syntaxe)

```
> fsolve(f(x)=0,x=-2..-1);
```

$$-1.060090320$$

Nom :

Prénom :

Groupe :

MK1 "Calcul formel" Maple

Partiel

Durée : 55 minutes

Documents autorisés : feuilles de TPs et corrections, résumés de cours personnels - Matériel interdit : livres, téléphones portables, documents informatiques,...

Rappel des commandes pour claviers Mac :
pour { : Alt (pour [: Alt Shift (
pour } : Alt)   pour] : Alt Shift)

Une attention particulière sera apportée à la syntaxe de Maple et à la rédaction !

Pour les exercices 1 et 2, donnez uniquement les *commandes* Maple qui permettent de répondre aux questions.

<u>Exercice 1</u>

Commencer par charger la librairie *numtheory* par la commande suivante :

```
> with(numtheory);
```

1) En consultant l'aide sur la librairie *numtheory*, trouver la commande qui permet de calculer les diviseurs d'un nombre entier et l'utiliser pour déterminer les diviseurs de $n = 4568742320$.

2) Compter combien n a de diviseurs (en utilisant Maple).

3) Déterminer les diviseurs communs à n et 154400.

Exercice 2

Soit la suite récurrente d'ordre 1 définie par

$$u_1 = 2$$

$$u_{n+1} = \frac{u_n}{2} + \frac{1}{u_n} \text{ pour } 2 \le n$$

1) Définir cette suite dans une procédure qui prend en entrée n. Cette procédure devra retourner un message d'erreur si $n \le 0$.

2) Afficher les valeurs approchées des 10 premières valeurs de la suite. Que constatez-vous ?

L'exercice 3 est à résoudre avec Maple. Lorsque vous effectuez des calculs qui permettent de répondre aux questions, donnez les *commandes* Maple correspondantes. Pour certaines questions, on demande des *explications* ou des *raisonnements mathématiques* . La rédaction sera prise en compte.

Exercice 3

Soit f la fonction qui à x (réel) associe $\dfrac{(x^2 - x - 6)^2 - x + 3}{x^2 - 4x + 3}$. Soit C la courbe représentative de f.

1) Définir f.

Déterminer le domaine Df où f est continue.

Vérifier que f est continue sur Df.

2) Calculer les limites de f aux bornes des intervalles qui composent Df.

La courbe C admet-elle une asymptote ? Si oui, précisez laquelle.

3) La fonction f est-elle prolongable par continuité ? Si oui, précisez ce prolongement.

Simplifier l'expression de $f(x)$. Pouviez-vous prévoir un prolongement par continuité ?

4) En utilisant la commande D, calculer la fonction dérivée de f et l'appeler g.

Déterminer (de façon exacte) son unique racine réelle et l'appeler a.

Etudier le signe de g.

En déduire les variations de f.

5) Déterminer l'unique zéro réel z de f.

Donner l'équation de la tangente T à la courbe C au point $(z, 0)$.

6) Tracer sur un même graphe la courbe C et la droite T en choisissant de bons intervalles d'affichage en abscisse et ordonnée.

Correction du partiel du 4 novembre

Remarque générale : dans les copies, un certain nombre d'erreurs sont dûes à une mauvaise lecture de l'énoncé. Lisez attentivement la question avant d'y répondre ! Si on demande une valeur approchée, ne calculez pas une valeur exacte. Si on demande de définir une fonction, ne définissez pas une expression. Si on demande d'utiliser *D*, n'utilisez pas *diff* ! Etc...

Exercice 1

1)

```
> restart;
  with(numtheory);
Warning, the protected name order has been redefined and unprotected
```

[*GIgcd*, *bigomega*, *cfrac*, *cfracpol*, *cyclotomic*, *divisors*, *factorEQ*, *factorset*, *fermat*, *imagunit*, *index*, *integral_basis*, *invcfrac*, *invphi*, *issqrfree*, *jacobi*, *kronecker*, λ, *legendre*, *mcombine*, *mersenne*, *migcdex*, *minkowski*, *mipolys*, *mlog*, *mobius*, *mroot*, *msqrt*, *nearestp*, *nthconver*, *nthdenom*, *nthnumer*, *nthpow*, *order*, *pdexpand*, φ, π, *pprimroot*, *primroot*, *quadres*, *rootsunity*, *safeprime*, σ, *sq2factor*, *sum2sqr*, τ, *thue*]

```
> ?numtheory
```

La commande de la librairie *numtheory* qui permet de calculer les diviseurs d'un nombre entier est *divisors* (pour consulter son aide, taper *?numtheory[divisors]*)

```
> n:=4568742320;
```

$$n := 4568742320$$

```
> divisors(n);
```

{1, 2, 4, 5, 8, 10, 16, 20, 40, 80, 193, 386, 772, 965, 1544, 1930, 3088, 3860, 7720, 15440, 295903, 591806, 1183612, 1479515, 2367224, 2959030, 4734448, 5918060, 4568742320, 1142185580, 2284371160, 11836120, 23672240, 57109279, 114218558, 228437116, 285546395, 456874232, 571092790, 913748464}

2) La commande précédente nous a donné un *ensemble* de diviseurs. Pour calculer le nombre d'éléments d'un ensemble (ou d'une liste), on dispose de la commande *nops*.

```
> nops(%);
```

$$40$$

Certains ont remarqué, à juste titre, que la commande *tau* de la librairie permet de calculer le nombre de diviseurs :

```
> tau(n);
```

$$40$$

3) Les diviseurs communs à *n* et 154400 sont les diviseurs de *n* qui sont aussi diviseurs de 154400. Pour les déterminer, il suffit de faire l'intersection de l'ensemble des diviseurs de *n* avec l'ensemble des diviseurs de 154400.

```
> divisors(n) intersect divisors(154400);
```

{1, 2, 4, 5, 8, 10, 16, 20, 40, 80, 193, 386, 772, 965, 1544, 1930, 3088, 3860, 7720, 15440}

Exercice 2

1) On fait une procédure récursive, dans laquelle on fait un test avec *if* pour distinguer les différents cas.

```
> restart;
> u:=proc(n)
  if n<=0 then RETURN("Erreur")
  elif n=1 then RETURN(2)
  else RETURN(u(n-1)/2+1/u(n-1));
  fi;
  end;
```

u := **proc**(*n*)
  if $n \leq 0$ **then** RETURN("Erreur")
  elif $n = 1$ **then** RETURN(2)
  else RETURN(1 / 2*u($n-1$) + 1 / u($n-1$))
  end if
end proc

Remarques :

- pour afficher une chaîne de caractères, on utilise " "
- **la formule de récurrence donnée exprime u(n+1) en fonction de u(n). Pour définir la suite dans Maple, on a besoin de la ramener à la forme : u(n) en fonction de u(n-1).**

2) On utilise *seq*.

```
> seq(evalf(u(k)),k=1..10);
```

2., 1.500000000, 1.416666667, 1.414215686, 1.414213562, 1.414213562, 1.414213562, 1.414213562, 1.414213562, 1.414213562

```
> evalf(sqrt(2));
```

$$1.414213562$$

On constate que la suite semble converger vers $\sqrt{2}$.

Exercice 3

1) On nous demande de définir une <u>*fonction*</u>.

```
> restart;
> f:=x->((x^2-x-6)^2-x+3)/(x^2-4*x+3);
```

$$f := x \to \frac{(x^2-x-6)^2-x+3}{x^2-4\,x+3}$$

```
> discont(f(x),x);
```

$$\{1, 3\}$$

Donc f est continue sur Df=]-∞,1[U]1,3[U [3, ∞[.

```
> iscont(f(x),x=-infinity..1);
```

$$true$$

```
> iscont(f(x),x=1..3);
```

$$true$$

```
> iscont(f(x),x=3..infinity);
```

$$true$$

2)

```
> limit(f(x),x=-infinity);
```

$$\infty$$

```
> limit(f(x),x=infinity);
```

$$\infty$$

```
> limit(f(x),x=1);
```

$$undefined$$

```
> limit(f(x),x=1,left);
```

$$\infty$$

```
> limit(f(x),x=1,right);
```

$$-\infty$$

```
> limit(f(x),x=3);
```

$$\frac{-1}{2}$$

Comme la limite de $f(x)$ en 1 vaut ∞ à gauche et $-\infty$ à droite, la courbe C admet une asymptote verticale d'équation $x=1$.

3) Comme la limite de $f(x)$ en 3 est finie et vaut $-\frac{1}{2}$, la fonction f peut être prolongée par continuité en $x=3$ en posant $f(3)=-\frac{1}{2}$.

```
> simplify(f(x));
```

$$\frac{x^3+x^2-8x-13}{x-1}$$

Sur l'expression simplifiée, on voit que 3 n'apparaît plus comme zéro du dénominateur. Cette expression est continue en 3. On retrouve le fait que f est prolongeable par continuité en 3.

4) Pour calculer la dérivée avec D :

```
> g:=D(f);
```

$$g:=x\to\frac{2(x^2-x-6)(2x-1)-1}{x^2-4x+3}-\frac{((x^2-x-6)^2-x+3)(2x-4)}{(x^2-4x+3)^2}$$

Remarque : on peut lui donner meilleure allure en mettant tout au même dénominateur :

```
> normal(g(x));
```

$$\frac{2x^3-2x^2-2x+21}{(x-1)^2}$$

Calculer les racines de g.

```
> S:=solve(g(x),x);
```

$$S:=-\frac{(1090+6\sqrt{32889})^{(1/3)}}{6}-\frac{8}{3(1090+6\sqrt{32889})^{(1/3)}}+\frac{1}{3},\frac{(1090+6\sqrt{32889})^{(1/3)}}{12}$$
$$+\frac{4}{3(1090+6\sqrt{32889})^{(1/3)}}+\frac{1}{3}$$
$$+\frac{1}{2}I\sqrt{3}\left(-\frac{(1090+6\sqrt{32889})^{(1/3)}}{6}+\frac{8}{3(1090+6\sqrt{32889})^{(1/3)}}\right),\frac{(1090+6\sqrt{32889})^{(1/3)}}{12}$$
$$+\frac{4}{3(1090+6\sqrt{32889})^{(1/3)}}+\frac{1}{3}$$
$$-\frac{1}{2}I\sqrt{3}\left(-\frac{(1090+6\sqrt{32889})^{(1/3)}}{6}+\frac{8}{3(1090+6\sqrt{32889})^{(1/3)}}\right)$$

```
> evalf(S);
```

$$-2.032828057,\ 1.516414027-1.692839762\,I,\ 1.516414027+1.692839762\,I$$

On voit qu'il y a une racine réelle (la première) et deux racines complexes. On nous demande la racine réelle , donc on ne garde que la première.

```
> a:=S[1];
```

$$a:=-\frac{(1090+6\sqrt{32889})^{(1/3)}}{6}-\frac{8}{3(1090+6\sqrt{32889})^{(1/3)}}+\frac{1}{3}$$

Etude du signe de g.

```
> solve(g(x)>0,x);
```

$$\mathrm{RealRange}\left(\mathrm{Open}\left(-\frac{(1090+6\sqrt{32889})^{(1/3)}}{6}-\frac{8}{3(1090+6\sqrt{32889})^{(1/3)}}+\frac{1}{3}\right),\mathrm{Open}(1)\right),$$
$$\mathrm{RealRange}(\mathrm{Open}(1),\infty)$$

```
> solve(g(x)<0,x);
```

$$\mathrm{RealRange}\left(-\infty,\mathrm{Open}\left(-\frac{(1090+6\sqrt{32889})^{(1/3)}}{6}-\frac{8}{3(1090+6\sqrt{32889})^{(1/3)}}+\frac{1}{3}\right)\right)$$

Donc g est strictement négative sur]- ∞,a[, puis strictement positive sur]a,1[U]1, +∞[. On en déduit que f est strictement décroissante sur]- ∞,a[, puis strictement croissante sur]a,1[U]1, +∞ [.

5)

```
> solve(f(x),x);
```

$$\frac{(1108+12\sqrt{1581})^{(1/3)}}{6}+\frac{50}{3(1108+12\sqrt{1581})^{(1/3)}}-\frac{1}{3},-\frac{(1108+12\sqrt{1581})^{(1/3)}}{12}$$
$$-\frac{25}{3(1108+12\sqrt{1581})^{(1/3)}}-\frac{1}{3}$$
$$+\frac{1}{2}I\sqrt{3}\left(\frac{(1108+12\sqrt{1581})^{(1/3)}}{6}-\frac{50}{3(1108+12\sqrt{1581})^{(1/3)}}\right),-\frac{(1108+12\sqrt{1581})^{(1/3)}}{12}$$
$$-\frac{25}{3(1108+12\sqrt{1581})^{(1/3)}}-\frac{1}{3}$$
$$-\frac{1}{2}I\sqrt{3}\left(\frac{(1108+12\sqrt{1581})^{(1/3)}}{6}-\frac{50}{3(1108+12\sqrt{1581})^{(1/3)}}\right)$$

On voit qu'il y a une racine réelle (la première) et deux racines complexes. On nous demande la racine réelle , donc on ne garde que la première.

```
> z:=solve(f(x),x)[1];
```

$$z := \frac{(1108+12\sqrt{1581})^{(1/3)}}{6} + \frac{50}{3\,(1108+12\sqrt{1581})^{(1/3)}} - \frac{1}{3}$$

La tangente à C en $(z, 0)$ a pour équation : y = f(z) + (x-z) f'(z) c'est-à-dire y = (x-z) g(z).

```
> T:=(x-z)*g(z);
```

$$\begin{aligned}
T := \Bigg(& x - \frac{(1108+12\sqrt{1581})^{(1/3)}}{6} - \frac{50}{3\,(1108+12\sqrt{1581})^{(1/3)}} + \frac{1}{3}\Bigg)\Bigg(\Bigg(2\Bigg(\\
&\Bigg(\frac{(1108+12\sqrt{1581})^{(1/3)}}{6} + \frac{50}{3\,(1108+12\sqrt{1581})^{(1/3)}} - \frac{1}{3}\Bigg)^2 - \frac{(1108+12\sqrt{1581})^{(1/3)}}{6} \\
&- \frac{50}{3\,(1108+12\sqrt{1581})^{(1/3)}} - \frac{17}{3}\Bigg)\Bigg(\frac{(1108+12\sqrt{1581})^{(1/3)}}{3} + \frac{100}{3\,(1108+12\sqrt{1581})^{(1/3)}} - \frac{5}{3}\Bigg) \\
&- 1\Bigg) \Bigg/ \Bigg(\Bigg(\Bigg(\frac{(1108+12\sqrt{1581})^{(1/3)}}{6} + \frac{50}{3\,(1108+12\sqrt{1581})^{(1/3)}} - \frac{1}{3}\Bigg)^2 \\
&- \frac{2\,(1108+12\sqrt{1581})^{(1/3)}}{3} - \frac{200}{3\,(1108+12\sqrt{1581})^{(1/3)}} + \frac{13}{3}\Bigg) - \Bigg(\Bigg(\\
&\Bigg(\frac{(1108+12\sqrt{1581})^{(1/3)}}{6} + \frac{50}{3\,(1108+12\sqrt{1581})^{(1/3)}} - \frac{1}{3}\Bigg)^2 - \frac{(1108+12\sqrt{1581})^{(1/3)}}{6} \\
&- \frac{50}{3\,(1108+12\sqrt{1581})^{(1/3)}} - \frac{17}{3}\Bigg)^2 - \frac{(1108+12\sqrt{1581})^{(1/3)}}{6} - \frac{50}{3\,(1108+12\sqrt{1581})^{(1/3)}} \\
&+ \frac{10}{3}\Bigg)\Bigg(\frac{(1108+12\sqrt{1581})^{(1/3)}}{3} + \frac{100}{3\,(1108+12\sqrt{1581})^{(1/3)}} - \frac{14}{3}\Bigg) \Bigg/ \Bigg(\\
&\Bigg(\frac{(1108+12\sqrt{1581})^{(1/3)}}{6} + \frac{50}{3\,(1108+12\sqrt{1581})^{(1/3)}} - \frac{1}{3}\Bigg)^2 - \frac{2\,(1108+12\sqrt{1581})^{(1/3)}}{3} \\
&- \frac{200}{3\,(1108+12\sqrt{1581})^{(1/3)}} + \frac{13}{3}\Bigg)^2\Bigg)\Bigg)
\end{aligned}$$

6)

```
> plot([f(x),T],x);
```



On recadre en abscisse et en ordonnée pour voir clairement la courbe et sa tangente. Par exemple ainsi :

```
> plot([f(x),T],x=-5..5,y=-30..30);
```



Université Paris 7 - Denis Diderot
L1
MK1 - MAPLE
Mardi 10 Janvier

Année 2005-2006
Fabrice Mathurin
mathurin@math.jussieu.fr

Examen MK1

L'examen est composé de 6 exercices indépendants. Il suffit de faire entièrement 4 exercices pour obtenir la note maximale. Il ne faut pas en entamer plus que 4, seuls les 4 premiers exercices de votre copie seront notés. Toutes les réponses sont à reporter sur les feuilles d'examen Les justifications comporteront essentiellement les lignes de codes Maple et des explications succintes pour expliquer précisément ce qui vous amène aux conclusions.
Nous tiendrons grandement compte de la clarté de la rédaction et de la syntaxe.
Aucun document n'est autorisé. Les portables doivent être éteints (sous peine d'annulation de la copie), les calculatrices sont interdites.

Exercice 1 (Analyse (5 points)) *Vous répondrez aux questions, et donnerez les lignes de codes nécessaires à l'affichage des résultats dans les situations classiques d'analyse suivante :*

1. *Quelles sont les limites de :*
   $ln(arctan(\frac{3x+1}{4x+\sqrt{3}}))$ *en* 0.
   $\frac{(\sin(x))^{\sin(x)}-1}{(\tan(x))^{\tan(x)}-1}$ *en* 0 *par valeurs positives*
2. *Donnez les développements limités des fonctions suivantes*
   $e^{\cos(x)}$ *en* 0 *et à l'ordre* 7.
   $sin(\sqrt{x^2-3\pi^2})$ *en* 2π *et à l'ordre* 3
3. *Calculez les intégrales suivantes :*
   $\int_0^{\frac{\pi}{2}} \frac{\cos(x)}{1+\sin^3(x)}dx$
   $\int_0^{\frac{\pi}{2}} \frac{\sin(x)}{\sqrt{1+\sin(x)\cos(x)}}dx$
4. *Donnez une primitive de :*
   $arctan(\sqrt{\frac{x+1}{x+3}})$
   $\frac{\sqrt[3]{x^3+1}}{x^2}$
5. *Donnez la dérivée de :*
   $-\frac{1}{3}\ln(1+\cos(x))+\frac{1}{6}\ln(\cos^2(x)-\cos(x)+1)-\frac{1}{\sqrt{3}}arctan(\frac{2\cos(x)-1}{\sqrt{3}})$
   $\frac{1}{\sqrt{x}+\sqrt[3]{x}}$

Exercice 2 (Géométrie, Complexe (5 points)) *Prenons trois points* A, B *et* C *du plan d'affixes complexes respectives* a, b *et* c. *Donnez une condition algébrique sur les complexes pour qu'ils soient alignés.*
En utilisant votre formule, donnez un programme align *qui prend en entrée trois complexes et qui renvoie en sortie* true *si les trois points du plan correspondants sont alignés et* false *dans le cas contraire.*

Exercice 3 (Tracé (5 points)) *Nous allons étudier la fonction* $f : x \mapsto arccos(4x^3-3x)$.
Bien entendu, tout cet exercice est à résoudre avec Maple.

1. *Donnez le domaine de définition de la fonction.*
2. *Déterminez la valeur de la dérivée de cette fonction.*
3. *Calculez les limites aux bornes du domaine. Donnez les différentes branches asymptotiques.*
4. *Établissez le tableau de variation de* f. *Donnez un tableau de valeurs de* f.
5. *Réalisez le tracé du graphe de* f. *Pour cela vous réfléchirez à l'échelle la plus appropriée. Outre le tracé approximatif de* f, *vous préciserez la (ou les) commande(s) Maple utilisées.*

Exercice 4 (Boucles for (5 points)) *Écrivez un programme calculant la double somme des puissances $k^{ème}$ et $r^{ème}$ des entiers de 1 à n, c'est à dire :*

$$\sum_{i=1}^{n}\sum_{j=1}^{n} i^k j^r$$

Ce programme prendra donc 3 *entiers en entrée,* n, k, r*, utilisera des boucles* for *et renverra la double somme* - i.e. *un entier.*

Exercice 5 (Résolution d'équations (5 points)) *Vous utiliserez Maple pour résoudre ces équations provenant de différentes situations mathématiques (pour les équations différentielles vous pouvez utiliser la fonction* dsolve*) :*

1. *Donnez les racines de* $X^4 + 3X^2 + 1$.
2. *Donnez les valeurs approchées des racines réelles de* $X^6 - 13X^3 + 2X^2 - 1$ *par une méthode de votre choix. (à* 10^{-5} *près)*
3. *Résoudre l'équation différentielle :*
$$(1 + x^2)y' - 2xy = 0$$
4. *Résoudre le système linéaire :*
$$\left\{\begin{array}{rcr} 3x + 2y - 4z & = & 2 \\ 5y + 2z & = & -3 \\ x - y + 8z & = & 1 \end{array}\right.$$
5. *Résoudre l'équation différentielle :*
$$y^{(4)} - y = \cos(x)$$

Exercice 6 (Boucles "si" et "tant que" (5 points)) *On appelle* ρ *la fonction définie ainsi :*

$$x \in \mathbb{R}^+ \mapsto card\{n \in \mathbb{N}, n \leq x, n \quad \textit{premier}\}$$

en terme clair, $\rho(x)$ *est le nombre d'entiers premiers plus petit que* x.
En utilisant des boucles if ... then ... else ... fi *et* while ... do ... od *évoquez d'abord dans les grandes lignes le principe d'une procédure prenant en entrée un réel positif* x *et renvoyant* $\rho(x)$ *puis finalement donnez le code Maple.*
*Tracez sur un même graphe (*display*) et sur une grande échelle les fonctions* ρ *et* $x \mapsto \frac{x}{ln(x)}$. *Que pouvez-vous en déduire ?*

Université Paris 7 - Denis Diderot
L1
MK1 - MAPLE
Mardi 10 Janvier

Année 2005-2006
Fabrice Mathurin
mathurin@math.jussieu.fr

Examen MK1

L'examen est composé de 6 exercices indépendants. Il suffit de faire entièrement 4 exercices pour obtenir la note maximale. Il ne faut pas en entamer plus que 4, seuls les 4 premiers exercices de votre copie seront notés. Toutes les réponses sont à reporter sur les feuilles d'examen Les justifications comporteront essentiellement les lignes de codes Maple et des explications succintes pour expliquer précisément ce qui vous amène aux conclusions.
Nous tiendrons grandement compte de la clarté de la rédaction et de la syntaxe.
Aucun document n'est autorisé. Les portables doivent être éteints (sous peine d'annulation de la copie), les calculatrices sont interdites.

Exercice 1 (Analyse (5 points)) *Vous répondrez aux questions, et donnerez les lignes de codes nécessaires à l'affichage des résultats dans les situations classiques d'analyse suivante :*

1. *Quelles sont les limites de :*
   $ln(arctan(\frac{3x+1}{4x+\sqrt{3}}))$ *en* 0.
   $\frac{(\sin(x))^{\sin(x)}-1}{(\tan(x))^{\tan(x)}-1}$ *en* 0 *par valeurs positives*
2. *Donnez les développements limités des fonctions suivantes*
   $e^{\cos(x)}$ *en* 0 *et à l'ordre* 7.
   $sin(\sqrt{x^2-3\pi^2})$ *en* 2π *et à l'ordre* 3
3. *Calculez les intégrales suivantes :*
   $\int_0^{\frac{\pi}{2}} \frac{\cos(x)}{1+\sin^3(x)}dx$
   $\int_0^{\frac{\pi}{2}} \frac{\sin(x)}{\sqrt{1+\sin(x)\cos(x)}}dx$
4. *Donnez une primitive de :*
   $arctan(\sqrt{\frac{x+1}{x+3}})$
   $\frac{\sqrt[3]{x^3+1}}{x^2}$
5. *Donnez la dérivée de :*
   $-\frac{1}{3}\ln(1+\cos(x))+\frac{1}{6}\ln(\cos^2(x)-\cos(x)+1)-\frac{1}{\sqrt{3}}arctan(\frac{2\cos(x)-1}{\sqrt{3}})$
   $\frac{1}{\sqrt{x}+\sqrt[3]{x}}$

Exercice 2 (Géométrie, Courbes (5 points)) *Dans cet exercice vous préciserez les commandes tapées :*

a *Étudiez et tracez les courbes données par :*
   $\rho=\sin(\theta/4)$
   $\rho=\theta\cdot\sin(\theta)$
   $\rho=1+4\cdot\cos(3\theta)$

b *Tracez la courbe définie par :*

$$\left\{\begin{array}{lcl} x &=& \sin(t) \\ y &=& \dfrac{\cos^2(t)}{2-\cos(t)} \end{array}\right.$$

Exercice 3 (Tracé (5 points)) *Nous allons étudier la fonction* $f\colon x \mapsto arccos(4x^3 - 3x)$.

Bien entendu, tout cet exercice est à résoudre avec Maple.

1. *Donnez le domaine de définition de la fonction.*
2. *Déterminez la valeur de la dérivée de cette fonction.*
3. *Calculez les limites aux bornes du domaine. Donnez les différentes branches asymptotiques.*
4. *Établissez le tableau de variation de* f*. Donnez un tableau de valeurs de* f.
5. *Réalisez le tracé du graphe de* f*. Pour cela vous réfléchirez à l'échelle la plus appropriée. Outre le tracé approximatif de* f*, vous préciserez la (ou les) commande(s) Maple utilisées.*

Exercice 4 (Boucles for (5 points)) *Écrivez un programme calculant la double somme des puissances* $k^{ème}$ *et* $r^{ème}$ *des entiers de* 1 *à* n*, c'est à dire :*

$$\sum_{i=1}^{n}\sum_{j=1}^{n} i^k j^r$$

Ce programme prendra donc 3 *entiers en entrée,* n, k, r*, utilisera des boucles* for *et renverra la double somme* - i.e. *un entier.*

Exercice 5 (Résolution d'équations (5 points)) *Vous utiliserez Maple pour résoudre ces équations provenant de différentes situations mathématiques (pour les équations différentielles vous pouvez utiliser la fonction* dsolve*) :*

1. *Donnez les racines de* $X^4 + 3X^2 + 1$.
2. *Donnez les valeurs approchées des racines réelles de* $X^6 - 13X^3 + 2X^2 - 1$ *par une méthode de votre choix. (à* 10^{-5} *près)*
3. *Résoudre l'équation différentielle :*
$$(1 + x^2)y' - 2xy = 0$$
4. *Résoudre le système linéaire :*
$$\left\{\begin{array}{lcr} 3x + 2y - 4z & = & 2 \\ 5y + 2z & = & -3 \\ x - y + 8z & = & 1 \end{array}\right.$$
5. *Résoudre l'équation différentielle :*
$$y^{(4)} - y = \cos(x)$$

Exercice 6 (Boucles "si" et "tant que" (8 points)) *Nous allons chercher dans cet exercice à simuler le jeu "plus grand, plus petit". Une personne cherche à deviner un nombre. Pour cela, elle peut interroger un oracle en lui proposant une valeur, et celui-ci répond simplement par l'une des trois réponses suivantes : "gagné", "plus grand" et "plus petit". On considère une variable globale* a *contenant le nombre à deviner. On rappelle que pour faire des tirages aléatoires, on utilise généralement la fonction* rand.

1. *Écrire une procédure qui prend en argument un entier naturel* $N \geq a$*, puis applique la méthode de recherche précédente afin de déterminer la valeur de* a *: on commence par choisir un nombre* b *entre* 1 *et* N *et on le compare à* a*. S'il est égal à* a*, on retourne le résultat, sinon on modifie le domaine de recherche en conséquence (dans* $[1, b-1]$ *si* $b > a$*, et dans* $[b+1, N]$ *si* $b < a$*), et ainsi de suite. La procédure pourra être itérative, ou récursive.*
2. *Modifiez la procédure précédente de manière à retourner également le nombre de tests réalisés.*
3. *Écrire enfin une procédure prenant en entrée un entier* N*, qui tire d'abord un nombre* a *compris entre* 1 *et* N*, puis applique la méthode précédente et retourne le nombre d'itérations nécessaires.*
4. *En évaluant la valeur moyenne du nombre d'itérations nécessaires pour une valeur donnée de* N*, donnez un lien entre ces deux valeurs.*

Université Paris 7 - Denis Diderot
L1
MK1 - MAPLE
Mercredi 11 Janvier

Année 2005-2006
Fabrice Mathurin
mathurin@math.jussieu.fr

Examen MK1

L'examen est composé de 6 exercices indépendants. Il suffit de faire entièrement 4 exercices pour obtenir la note maximale. Il ne faut pas en entamer plus que 4, seuls les 4 premiers exercices de votre copie seront notés. Toutes les réponses sont à reporter sur les feuilles d'examen Les justifications comporteront essentiellement les lignes de codes Maple et des explications succintes pour expliquer précisément ce qui vous amène aux conclusions.
Nous tiendrons grandement compte de la clarté de la rédaction et de la syntaxe.
Aucun document n'est autorisé. Les portables doivent être éteints (sous peine d'annulation de la copie), les calculatrices sont interdites.

Exercice 1 (Analyse (5 points)) *Donnez les limites et les DLs des fonctions suivantes à l'ordre et au point précisé : (en fournissant les commandes Maple)*

- $\tan(2t)\cdot\ln(\tan(t))$ *(à l'ordre 7 et en $\frac{\pi}{4}$)*
- $\cos(t)\cdot e^{\frac{1}{t-\sin(t)}}$ *(à l'ordre 2 et en $\frac{\pi}{2}$)*
- $\cos(t)^{\tan(t^2)}$ *(à l'ordre 11 et en 0)*
- $(e^t+t)^{\frac{1}{t}}$ *(à l'ordre 5 et en 0)*
- $\cos(\arcsin(t))$ *(à l'ordre 9 et en 0)*

Exercice 2 (Suite (6 points)) *Soit $a\in\mathbb{R}$. On considère la suite $(u_n)_{n\in\mathbb{N}}$ suivante :*

$$u_0=a; \qquad \forall\, n\geq 0 \qquad u_{n+1}=\frac{5u_n-3}{u_n+1}$$

1. *Écrivez une procédure Maple prenant en entrée un nombre réel $a\in\mathbb{R}$ et un entier naturel n et renvoyant le n-ème terme de la suite.*
2. *Indiquez la valeur de u_{10} en fonction de a.*
3. *Étudiez la convergence de la suite en fonction de la valeur de a.*

Soient a et b dans $\mathbb{R}$. On considère la suite définie par :

$$u_0=a; \qquad u_1=b; \qquad \forall\, n\geq 0 \qquad u_{n+2}=|u_{n+1}|-u_n$$

1. *Écrivez une procédure Maple prenant en entrée un nombre réel $a\in\mathbb{R}$, un nombre réel $b\in\mathbb{R}$ et un entier naturel n et renvoyant le n-ème terme de la suite.*
2. *En prenant quelques exemples faites des simulations numériques. Que constatez-vous ?*

Exercice 3 (Tracé (5 points)) *Nous allons étudier la fonction* $f\colon x\mapsto 1+\frac{1}{x-1}\displaystyle\int_1^x \frac{t^2}{\sqrt{1+t^8}}dt$

Bien entendu, tout cet exercice est à résoudre avec Maple.

1. *Donnez le domaine de définition de la fonction.*
2. *Déterminez la valeur de la dérivée de cette fonction.*
3. *Calculez les limites aux bornes du domaine. Donnez les différentes branches asymptotiques.*
4. *(**Difficile**) Établissez le tableau de variation de f. Donnez un tableau de valeurs de f.*
5. *Réalisez le tracé du graphe de f. Pour cela vous réfléchirez à l'échelle la plus appropriée. Outre le tracé approximatif de f, vous préciserez la (ou les) commande(s) Maple utilisées.*

Exercice 4 (Boucles for (5 points)) *Écrivez un programme calculant la double somme des puissances $k^{\text{ème}}$ et $r^{\text{ème}}$ des entiers de 1 à n, c'est à dire :*

$$\sum_{i=1}^{n}\sum_{j=1}^{n} i^k j^r$$

Ce programme prendra donc 3 entiers en entrée, n,k,r, utilisera des boucles for *et renverra la double somme* - i.e. *un entier.*

Exercice 5 (Résolution d'équations (5 points)) *Vous utiliserez Maple pour résoudre ces équations provenant de différentes situations mathématiques (pour les équations différentielles vous pouvez utiliser la fonction* dsolve*) :*

1. *Donnez les racines de* $X^4 + 3X^2 + 1$.
2. *Donnez les valeurs approchées des racines réelles de* $X^6 - 13X^3 + 2X^2 - 1$ *par une méthode de votre choix. (à* 10^{-5} *près)*
3. *Résoudre l'équation différentielle :*
$$(1 + x^2)y' - 2xy = 0$$
4. *Résoudre le système linéaire :*
$$\left\{ \begin{array}{rcl} 3x + 2y - 4z & = & 2 \\ 5y + 2z & = & -3 \\ x - y + 8z & = & 1 \end{array} \right.$$
5. *Résoudre l'équation différentielle :*
$$y^{(4)} - y = \cos(x)$$

Exercice 6 (Divers (5 points))

1. *(Boucle While) Donnez le plus petit entier* $n \geq 1$ *tel que* $E(2005 \cdot (\frac{3}{2})^n))$ *soit premier. Expliquez votre méthode. (*$E(x)$ *désigne la partie entière de* x, i.e. *le plus petit entier précédant* x.*)*
2. *(Matrices) On considère la matrice* $A = (a_{i,j})_{1 \leq i,j \leq 4}$ *définie par* $\forall\, 1 \leq i, j \leq 4, a_{i,j} = \frac{i-j}{i+j}$.
   (a) *Proposez une méthode permettant de définir directement la matrice (c'est-à-dire sans rentrer un à un les 16 coefficients).*
   (b) *Donnez le rang de* M.
   (c) *Calculez, si elle existe, l'inverse de la matrice* A.